Automation .AU Programs for DISR91
(including Inverse, Solids, and Selective/Shaped Programs)

Table of Contents

I.Magnetic Resonance Facility–Specific Programs

FT.AU

```
; Automatically EM;FT;PK a Set of FID's
;  C. FRY - 22 SEPT 1993

1 ZE
2 RE #1
3 EM
4 FT
5 PK
6 WR #2
7 IF #1
8 IF #2
9 IN=1
10 EXIT

; USE AI=1 AND RE LARGEST INTENSITY FID FOR ABSOLUTE INTENSITY
; EM;FT FID, THEN PHASE IN EP
; NOW CAN RUN FT.AU WITH NE=# EXPERIMENTS
```

SLOW.AUR

```
; Acquisition With Very Long Delays Between Files (e.g., for Drift Measurements)
;  C. Fry - 25 April 1994

1 ZE
10 GO=10         ;ACQUIRE DATA AFTER 90 DEG PULSE, LOOP TO 7
11 WR #1        ;STORE CURRENT FID
12 IF #1        ;INCREMENT FILE EXTENSION
13 D1
14 IN=1         ;REPEAT CYCLE THROUGH DELAY LIST
15 EXIT
```

T1NULL.AUR

```
; Quick T1 Determination With Inversion-Recovery
;   D1 - 180 - VD - 90 - FID
;  C. FRY  -  10 DEC 1993

1 ZE
2 P2            ;180 DEG PULSE
3 VD            ;VARIABLE DELAY (TAKEN FROM CURRENT 'VD' LIST)
4 GO=2          ;ACQUIRE DATA AFTER 90 DEG PULSE
5 EXIT

;D1 = 3*T1
;P2 = 180 PULSE (CONSTANT PHASE)
;RD=0, PW= 90 PULSE
;NS = 1, DS=0
;NE = 1
;CHANGE VD VALUE MANUALLY USING VD COMMAND UNTIL PROTON OF
;INTEREST IS NULLED
;USE AI=1 TO HELP VISUALIZE CHANGE IN INTENSITIES
```

TONET.AU

```
; Takes Data And Automatically Sends To Novell Network
; 21 MAR 94 C.G.FRY
```

```
1 RF #1
2 ZE
3 GO=3
4 WR #1
5 RE #1
6 PASC CURFILE
7 PASC TOPC
8 EXIT
```

WRTOPC.AU

; Writes Data And Automatically Sends To Novell Network

```
; 21 MAR 94 C.G.FRY

1 RF #1
2 WR #1
3 RE #1
4 PASC CURFILE
5 PASC TOPC
6 EXIT
```

II. Bruker Supplied Programs

ADC.AUR

; Sample Program For Data Acquisition Using 'ADC' Command And Normal Dwell Clock.

```
1 ZE
2 D1            ;RELAXATION DELAY
3 P1 PH1        ;TRANSM. PULSE (RECEIVER BLANKED)
4 D5            ;DE/2 (RECEIVER STILL OFF)
5 D5 PH2        ;SET REFERENCE PHASE FOR DETECTION AND
                ; OPEN RECEIVER GATE
6 D6 ADC        ;D6=>AQ, 'ADC' OPENS REC. AND STARTS DIGITIZER
                ; TAKE TD DATA POINTS USING DWELL TIME DW
7 RCYC=2 PH3    ;LOOP FOR NS SCANS
  EXIT

PH1=0 0 2 2             ;TRANSM. PHASES (:A COULD ALSO BE USED)
PH2=0 0 0 0 3 3 3 3     ;REFERENCE PHASE FOR DETECTION
                                    ;  IS NORMALLY '0'.
PH3=R0 R0 R2 R2 R1 R1 R3 R3   ;RECEIVER PHASE

;PERFORMS DATA ACQUISITION IN A MANNER IDENTICAL TO 'GO'
; D1 IS EQUIVALENT TO 'RD'
; P1 IS EQUIVALENT TO PW
; 2*D5 IS EQUIVALENT TO DE
; D6 IS EQUIVALENT TO AQ

;NB: WHEN SHIFTING DETECTOR PHASE (PH2) INSTEAD OF PULSE PHASE
; (PH1), PHASE SHIFT MUST BE IN OPPOSITE SENSE SO THAT QP,
; FOR EXAMPLE, FUNCTIONS IN THE USUAL WAY.
```

ADCMLEV.AUR

; Sample Program For Data Acquisition Using Mlev-4 Decoupling Using 'ADC' Command And Normal Dwell Clock.

```
1 ZE
2 D1 BB S2       ;RELAXATION DELAY WITH DEC. FOR NOE
3 P3:A           ;TRANSM. PULSE (RECEIVER BLANKED)
4 D5             ;DE/2 (RECEIVER STILL OFF)
5 D5 PH9 CW      ;SET REFERENCE PHASE FOR DETECTION AND
                 ; OPEN RECEIVER GATE, SET CW MODE
6 D6 ADC         ;D6=2 USEC, 'ADC' OPENS REC. AND STARTS DIGITIZER
                 ; TAKE TD DATA POINTS USING DWELL TIME DW
7 (P1 PH0 P2 PH1 P1 PH0):D      ;'R' ELEMENT OF MLEV-4
  (P1 PH0 P2 PH1 P1 PH0):D      ;'R'
  (P1 PH2 P2 PH3 P1 PH2):D      ;'R-BAR'
  (P1 PH2 P2 PH3 P1 PH2):D      ;'R-BAR'
8 L1 TO 7 TIMES UPR             ;HARDWARE LOOP USING COUNTER L1
9 RCYC=2 PH8                    ;LOOP FOR NS SCANS
  EXIT

PH0=0   ;DECOUPLER PHASES
PH1=1
PH2=2
PH3=3
PH8=R0 R0 R2 R2 R1 R1 R3 R3     ;RECEIVER PHASE
PH9=0 0 0 0 3 3 3 3  ;REFERENCE PHASE FOR DETECTION

;PERFORMS DATA ACQUISITION IN A MANNER IDENTICAL TO 'GO' (QP)
; D1 IS EQUIVALENT TO 'RD'
; S2 DEFINES DECOUPLER POWER
; P3 IS EQUIVALENT TO PW
; 2*D5 IS EQUIVALENT TO DE
; D6=2 USEC FOR ADC COMMAND
; P1=90 DEG 1H DEC. PULSE AT POWER SETTING S2
; P2=240 DEG DEC. PULSE GIVES QUITE GOOD RESULTS FOR SIMPLE
;  MLEV-4 SEQUENCE.
;L1 = LOOP COUNTER, SET SO THAT L1*4*(P1+P2+P1) => AQ.
```

ANGLE.AUR

; To Be Used For Mas Angle Setting With Either Normal Transmitter Or AR.

```
100 ZE
200 D1                          ;D1 = .01
    (P4):C8                     ;P4 = 40 USEC FOR UNBLANKING
    (P1 PH1):T (P1):C8
    D3                          ;RINGDOWN
    GO=200
    IN=100
EXIT

PH1= 0 0 2 2

;P1= 2 USEC
;PW, RD = 0
;DE => 5 USEC
;D3 = 20 USEC
```

ANTIRING.AU

; Sequence To Reduce Acoustic Ringing

```
;P.S.Belton, I.J.Cox & R.K.Harris, J. Chem. Soc. Faraday Trans. 2,
;   81, 63-75 (1985)
```

```
1 ZE
  D3 S2 CPD              ;power switching delay 3msec
2 D1                     ;relaxation delay
  P3 PH1                 ;90 deg X pulse
  GO=2 PH3
3 D1                     ;relaxation delay
  P3 PH2                 ;90 deg X pulse
  GO=3 PH4
4 D1                     ;relaxation delay
  P4 PH1                 ;180 deg X pulse
  D2                     ;0.5 usec
  P3 PH2                 ;90 deg X pulse
  GO=4 PH3
5 D1                     ;relaxation delay
  P4 PH1                 ;180 deg X pulse
  D2                     ;0.5 usec
  P3 PH1                 ;90 deg X pulse
  GO=5 PH4
  LO TO 2 TIMES L0       ;hardware loop to repeat sequence
  WR #1                  ;store FID
  IF #1                  ;increment filename extension
  IN=2                   ;loop for long term cycling NE times
  D3 DO                  ;exit with decoupler gated off
EXIT

PH1= 0 0 0 0 1 1 1 1 2 2 2 2 3 3 3 3
PH2= 2 2 2 2 3 3 3 3 0 0 0 0 1 1 1 1
PH3=R0 R0 R0 R0 R1 R1 R1 R1 R2 R2 R2 R2 R3 R3 R3 R3
PH4=R2 R2 R2 R2 R3 R3 R3 R3 R0 R0 R0 R0 R1 R1 R1 R1


;D1 : 1-5 T1
;P3,P4 : 90, 180 deg X pulse
;D2 = 0.5 usec
;D3 = 3 msec
;S2 : decoupler power level for CPD
;P9 : 90 deg H-1 pulse at power level S2
;NS : 16 * n
;DS : 2 or 4
```

ASATREC.AUR

```
; Aperiodic Saturation-Recovery For T1
;  HETERONUC. APPLICATION WITH POWER GATED DECOUPLING
;  (90 - T)*N - VD - 90 - FID
;  USES A TRAIN OF 90 DEG PULSES WITH DECREASING DELAYS TO
;  SATURATE ALL RESONANCES.

1 ZE
2 WR #1          ;PREPARE A SET OF ZEROED DISK FILES
3 IF #1
4 LO TO 2 TIMES C        ;FIRST C = NO. OF FILES=NE
  VC             ;SELECT SECOND 'C' VALUE
5 RF #1.001      ;RESET FILE EXTENSION, BEGIN CYCLE
6 RE #1          ;READ CURRENT FID
7 D1 BB S1       ;DELAY FOR TEMP. EQUILIBRIUM, HOLD NOE
8 SP             ;SAT. PULSE TRAIN
9 VD             ;VARIABLE DELAY (TAKEN FROM CURRENT 'VD' LIST)
10 D2 S2         ;SWITCH DEC. POWER FOR GOOD DECOUPLING
11 GO=7          ;ACQUIRE DATA AFTER 90 DEG PULSE, LOOP TO 7
12 WR #1         ;STORE CURRENT FID
13 IF #1         ;INCREMENT FILE EXTENSION
14 IN=6          ;LOOP TO 6 AND INCREMENT VD LIST POINTER
15 LO TO 5 TIMES C       ;REPEAT CYCLE THROUGH DELAY LIST
16 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES THE NUMBER OF TAU VALUES IN THE VD LIST, THE NUMBER
;OF FIDS STORED.
;THE CURRENT VD LIST MUST CONTAIN THE SET OF RECOVERY DELAYS
;TO BE USED (IN ANY ORDER).
;D1 = A TIME LONG ENOUGH TO AVOID EXCESSIVE HEATING OF SAMPLE
;  BY DEC. POWER S2 WHEN VD IS SHORT (E.G. 2-4*AQ)
;S1 = LOW POWER FOR NOE (E.G. 0.5 WATT)
;D2 = 2 MSEC TO SWITCH POWER (RECOVERY TAU=VD+D2)
;S2 = HIGHER DEC. POWER FOR GOOD DECOUPLING
;SP: ENTER A TIME IN SEC WHICH IS THE DELAY BETWEEN THE FIRST
;  TWO PULSES IN THE SAT. PULSE TRAIN. THIS DELAY WILL BE REDUCED
;  BY CA. FACTOR 2 FOR EACH SUBSEQUENT PULSE UNTIL A MIN. OF
;  1 MSEC IS REACHED.
;P0 = THE PULSE LENGTH USED BY THE SP PULSE TRAIN (=90 DEG).
;RD=0, PW= 90 PULSE
;NS = MULTIPLE OF 8, DS=0
;CURRENT VC LIST MUST CONTAIN AN ENTRY WHICH DEFINES THE NUMBER
;OF FILES (=NE) AND A SECOND ENTRY FOR THE NUMBER
;OF CYCLES TO BE MADE THROUGH THE VD LIST FOR LONG-TERM AVERAGING.
;TOTAL TRANSIENTS PER FILE = C*NS.
```

AUTOSAMA.AU

; Manual Sample Change (see also MCHSAMA.AU)

```
1 SXM
2 RJX
3 LOPO
4 ROT
5 LOCK
6 RJXS
7 AU @
  JOUA
8 IJX
9 LO TO 1 TIMES 10000
10 EXIT
```

AUTOSAMP.AU

; Manual Sample Change (see also MCHSAMP.AU)

```
1 JOUP
2 RJX
3 RJXS
4 AU @
5 LO TO 1 TIMES 1000
6 EXIT
```

BBCHECK.AUR

; Test Program To Check Quality Of BB Decoupling As A Function Of O2 Offset.

```
1 ZE
2 D1 BB S1 O2   ;RELAXATION WITH MIN. DECOUPLING FOR NOE,
                ;SET OFFSET O2 FROM FL LIST
3 D2 BB S2      ;INCREASE POWER FOR GOOD DECOUPLING
4 GO=4          ;ACQUIRE FID
  D3 BB S1      ;TIME TO OBSERVE FID
5 IN=1          ;LOOP TO 1 FOR NEXT O2 VALUE
6 D2 BB S1
7 EXIT          ;EXIT WITH DEC. ON FOR NOE

;USE SUITABLE TEST SAMPLE SUCH AS C6H6
;DEFINE FL LIST CONTAINING O2 VALUES FOR 1H DECOUPLING,
;  TYPICALLY WITH VALUES FOR 0,1,2,3,4,5,6, AND 0 PPM
;  OFF-RESONANCE FROM 1H SIGNAL (IE. NE=8).
;D1=CA. 1 SEC, D2=.01, D3= CA. 3 SEC
;S1=CA. 0.4 WATT
;S2=ADJUST AS NEEDED FOR GOOD DEC.,
;NS=1, RD=0, AQ=CA. 1 SEC, PW=CA. 20 DEG FLIP.
;RG: ADJUST TO FILL DIGITIZER WITH FID
```

BIRD.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using BIRD Sequence In Inverse Mode. No Decoupling During Acquisition (see also INVH1.AU)

```
;A.Bax and S. Subramanian, J. Magn. Reson. 67, 565-569 (1986)

  II
1 ZE
2 D1 S1 DO                        ;relaxation delay
  P1:D PH1                        ;90 deg H-1 pulse
  D2                              ;1/(2J)XH
  (P2 PH1):D (P4 PH7)             ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                              ;recovery delay
  P3 PH10                         ;90 deg X pulse
3 P1:D PH1                        ;90 deg H-1 pulse
4 D2                              ;1/(2J)XH
5 P3 PH3                          ;90 deg X pulse
  D0                              ;t1/2
6 P2:D PH2                        ;180 deg H-1 pulse
  D0
7 P3 PH4
9 GO=2 PH5
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH7=0
PH9=2
PH10=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH2=0
PH3=0 1 2 3
PH4=0 0 0 0 2 2 2 2
PH5=R0 R3 R2 R1 R2 R1 R0 R3

;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0 = 2
```

BIRDD9.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using BIRD Sequence In Inverse Mode. Decoupling During Acquisition Using GARP1

```
;this requires BSV7 with fast TLO (RCP7) and additional
;   amplifier (e.g. BFX5, gated with RCP3)
;(see also INVH1.AU)
;A.Bax and S. Subramanian, J. Magn. Reson. 67, 565-569 (1986)


  II
1 ZE
2 D1 S1 DO                        ;relaxation delay
  P1:D PH1                        ;90 deg H-1 pulse
  D2                              ;1/(2J)XH
  (P2 PH1):D (P4 PH7)             ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                              ;recovery delay
  P3 PH11                         ;90 deg X pulse
  P1:D PH1                        ;90 deg H-1 pulse
  D2                              ;1/(2J)XH
  P3 PH3                          ;90 deg X pulse
  D0                              ;t1/2
  P2:D PH2                        ;180 deg H-1 pulse
  D0
  P3 PH4
  D2                              ;1/(2J)XH  refocussing delay
  D5                              ;DE/2
3 D5:D PH8
4 D6 ADC                          ;D6 = 2 usec
  P0      :C7
5 P8*0.339:C7:T:C3 PH10           ;GARP1 decoupling
  P8*0.613:C7:T:C3 PH12
  P8*2.864:C7:T:C3 PH10
  P8*2.981:C7:T:C3 PH12
  P8*0.770:C7:T:C3 PH10
  P8*0.691:C7:T:C3 PH12
  P8*0.944:C7:T:C3 PH10
  P8*1.020:C7:T:C3 PH12
  P8*1.494:C7:T:C3 PH10
  P8*2.846:C7:T:C3 PH12
  P8*0.738:C7:T:C3 PH10
  P8*0.510:C7:T:C3 PH12
  P8*0.283:C7:T:C3 PH10
  P8*0.808:C7:T:C3 PH12
  P8*1.328:C7:T:C3 PH10
  P8*1.536:C7:T:C3 PH12
  P8*2.871:C7:T:C3 PH10
  P8*0.721:C7:T:C3 PH12
  P8*0.788:C7:T:C3 PH10
  P8*0.858:C7:T:C3 PH12
  P8*1.091:C7:T:C3 PH10
  P8*1.484:C7:T:C3 PH12
  P8*2.843:C7:T:C3 PH10
  P8*0.729:C7:T:C3 PH12
  P8*0.593:C7:T:C3 PH10
  L0 TO 5 TIMES 2
6 P8*0.339:C7:T:C3 PH12
  P8*0.613:C7:T:C3 PH10
  P8*2.864:C7:T:C3 PH12
  P8*2.981:C7:T:C3 PH10
  P8*0.770:C7:T:C3 PH12
  P8*0.691:C7:T:C3 PH10
  P8*0.944:C7:T:C3 PH12
```

```
  P8*1.020:C7:T:C3 PH10
  P8*1.494:C7:T:C3 PH12
  P8*2.846:C7:T:C3 PH10
  P8*0.738:C7:T:C3 PH12
  P8*0.510:C7:T:C3 PH10
  P8*0.283:C7:T:C3 PH12
  P8*0.808:C7:T:C3 PH10
  P8*1.328:C7:T:C3 PH12
  P8*1.536:C7:T:C3 PH10
  P8*2.871:C7:T:C3 PH12
  P8*0.721:C7:T:C3 PH10
  P8*0.788:C7:T:C3 PH12
  P8*0.858:C7:T:C3 PH10
  P8*1.091:C7:T:C3 PH12
  P8*1.484:C7:T:C3 PH10
  P8*2.843:C7:T:C3 PH12
  P8*0.729:C7:T:C3 PH10
  P8*0.593:C7:T:C3 PH12
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
  P0      :C7
8 RCYC=2 PH5
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH7=0
PH9=2
PH11=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH2=0
PH3=0 1 2 3
PH4=0 0 0 0 2 2 2 2
PH5=R0 R3 R2 R1 R2 R1 R0 R3

PH8=0

PH10=0
PH12=2



;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;D5 : DE/2
;D6 = 2 usec
;P0 = 5 usec
;P8 : 90 deg pulse for X decoupling
;L2 : L2 * 31.75 * 4 * P9 => AQ
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0 = 2
```

BIRDDP3.AUR

```
; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence
  Using BIRD Sequence In Inverse Mode.  Phase Sensitive Using TPPI Decoupling
  During Acquisition Using GARP1
;this requires BSV7 with fast TLO (RCP7) and additional
;   amplifier (e.g. BFX5, gated with RCP3)
;(see also INVH1.AU)
;A.Bax and S. Subramanian, J. Magn. Reson. 67, 565-569 (1986)


  II
1 ZE
2 D1 S1 DO                          ;relaxation delay
  P1:D PH1                          ;90 deg H-1 pulse
  D2                                ;1/(2J)XH
  (P2 PH1):D (P4 PH7)               ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                                ;recovery delay
  P3 PH11                           ;90 deg X pulse
  P1:D PH1                          ;90 deg H-1 pulse
  D2                                ;1/(2J)XH
  P3 PH3                            ;90 deg X pulse
  D0                                ;t1/2
  P2:D PH2                          ;180 deg H-1 pulse
  D0
  P3 PH4
  D2                                ;1/(2J)XH  refocussing delay
  D5                                ;DE/2
3 D5:D PH8
4 D6 ADC                            ;D6 = 2 usec
  P0 :C7
5 P7 :C7:T:C3 PH10                  ;GARP1 decoupling
  P10:C7:T:C3 PH12
  P28:C7:T:C3 PH10
  P30:C7:T:C3 PH12
  P15:C7:T:C3 PH10
  P11:C7:T:C3 PH12
  P19:C7:T:C3 PH10
  P20:C7:T:C3 PH12
  P24:C7:T:C3 PH10
  P27:C7:T:C3 PH12
  P14:C7:T:C3 PH10
  P8 :C7:T:C3 PH12
  P6 :C7:T:C3 PH10
  P17:C7:T:C3 PH12
  P22:C7:T:C3 PH10
  P25:C7:T:C3 PH12
  P29:C7:T:C3 PH10
  P12:C7:T:C3 PH12
  P16:C7:T:C3 PH10
  P18:C7:T:C3 PH12
  P21:C7:T:C3 PH10
  P23:C7:T:C3 PH12
  P26:C7:T:C3 PH10
  P13:C7:T:C3 PH12
  P9 :C7:T:C3 PH10
  L0 TO 5 TIMES 2
6 P7 :C7:T:C3 PH12
  P10:C7:T:C3 PH10
  P28:C7:T:C3 PH12
  P30:C7:T:C3 PH10
  P15:C7:T:C3 PH12
  P11:C7:T:C3 PH10
  P19:C7:T:C3 PH12
```

```
  P20:C7:T:C3 PH10
  P24:C7:T:C3 PH12
  P27:C7:T:C3 PH10
  P14:C7:T:C3 PH12
  P8 :C7:T:C3 PH10
  P6 :C7:T:C3 PH12
  P17:C7:T:C3 PH10
  P22:C7:T:C3 PH12
  P25:C7:T:C3 PH10
  P29:C7:T:C3 PH12
  P12:C7:T:C3 PH10
  P16:C7:T:C3 PH12
  P18:C7:T:C3 PH10
  P21:C7:T:C3 PH12
  P23:C7:T:C3 PH10
  P26:C7:T:C3 PH12
  P13:C7:T:C3 PH10
  P9 :C7:T:C3 PH12
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
  P0 :C7
8 RCYC=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH7=0
PH9=2
PH11=0 0 0 0 2 2 2 2
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0

PH8=0

PH10=0
PH12=2




;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;D5 : DE/2
;D6 = 2 usec
;P0 = 5 usec
;P9 : 90 deg pulse for X decoupling
;L2 : L2 * 31.75 * 4 * P9 => AQ
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
;MC2 = W
```

BIRDDP9.AUR

```
; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence
  Using BIRD Sequence In Inverse Mode.  Phase Sensitive Using TPPI.
  Decoupling During Acquisition Using GARP1
;this requires BSV7 with fast TLO (RCP7) and additional
;   amplifier (e.g. BFX5, gated with RCP3)
;(see also INVH1.AU)
;A.Bax and S. Subramanian, J. Magn. Reson. 67, 565-569 (1986)


  II
1 ZE
2 D1 S1 DO                        ;relaxation delay
  P1:D PH1                        ;90 deg H-1 pulse
  D2                              ;1/(2J)XH
  (P2 PH1):D (P4 PH7)             ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                              ;recovery delay
  P3 PH11                         ;90 deg X pulse
  P1:D PH1                        ;90 deg H-1 pulse
  D2                              ;1/(2J)XH
  P3 PH3                          ;90 deg X pulse
  D0                              ;t1/2
  P2:D PH2                        ;180 deg H-1 pulse
  D0
  P3 PH4
  D2                              ;1/(2J)XH  refocussing delay
  D5                              ;DE/2
3 D5:D PH8
4 D6 ADC                          ;D6 = 2 usec
  P0      :C7
5 P8*0.339:C7:T:C3 PH10           ;GARP1 decoupling
  P8*0.613:C7:T:C3 PH12
  P8*2.864:C7:T:C3 PH10
  P8*2.981:C7:T:C3 PH12
  P8*0.770:C7:T:C3 PH10
  P8*0.691:C7:T:C3 PH12
  P8*0.944:C7:T:C3 PH10
  P8*1.020:C7:T:C3 PH12
  P8*1.494:C7:T:C3 PH10
  P8*2.846:C7:T:C3 PH12
  P8*0.738:C7:T:C3 PH10
  P8*0.510:C7:T:C3 PH12
  P8*0.283:C7:T:C3 PH10
  P8*0.808:C7:T:C3 PH12
  P8*1.328:C7:T:C3 PH10
  P8*1.536:C7:T:C3 PH12
  P8*2.871:C7:T:C3 PH10
  P8*0.721:C7:T:C3 PH12
  P8*0.788:C7:T:C3 PH10
  P8*0.858:C7:T:C3 PH12
  P8*1.091:C7:T:C3 PH10
  P8*1.484:C7:T:C3 PH12
  P8*2.843:C7:T:C3 PH10
  P8*0.729:C7:T:C3 PH12
  P8*0.593:C7:T:C3 PH10
  L0 TO 5 TIMES 2
6 P8*0.339:C7:T:C3 PH12
  P8*0.613:C7:T:C3 PH10
  P8*2.864:C7:T:C3 PH12
  P8*2.981:C7:T:C3 PH10
  P8*0.770:C7:T:C3 PH12
  P8*0.691:C7:T:C3 PH10
  P8*0.944:C7:T:C3 PH12
```

```
  P8*1.020:C7:T:C3 PH10
  P8*1.494:C7:T:C3 PH12
  P8*2.846:C7:T:C3 PH10
  P8*0.738:C7:T:C3 PH12
  P8*0.510:C7:T:C3 PH10
  P8*0.283:C7:T:C3 PH12
  P8*0.808:C7:T:C3 PH10
  P8*1.328:C7:T:C3 PH12
  P8*1.536:C7:T:C3 PH10
  P8*2.871:C7:T:C3 PH12
  P8*0.721:C7:T:C3 PH10
  P8*0.788:C7:T:C3 PH12
  P8*0.858:C7:T:C3 PH10
  P8*1.091:C7:T:C3 PH12
  P8*1.484:C7:T:C3 PH10
  P8*2.843:C7:T:C3 PH12
  P8*0.729:C7:T:C3 PH10
  P8*0.593:C7:T:C3 PH12
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
  P0      :C7
8 RCYC=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH7=0
PH9=2
PH11=0 0 0 0 2 2 2 2
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0

PH8=0

PH10=0
PH12=2



;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;D5 : DE/2
;D6 = 2 usec
;P0 = 5 usec
;P8 : 90 deg pulse for X decoupling
;L2 : L2 * 31.75 * 4 * P9 => AQ
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
;MC2 = W
```

BIRDDPS.AUR

```
; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence
  Using BIRD Sequence In Inverse Mode.  Phase Sensitive Using TPPI.
  Decoupling During Acquisition Using GARP1
;this requires BSV7 with fast TLO (RCP7) and additional
;   amplifier (e.g. BFX5, gated with RCP3)
;with water suppression using presaturation
;this requires fast switching decoupler
;(see also INVH1.AU)
;A.Bax and S. Subramanian, J. Magn. Reson. 67, 565-569 (1986)


  II
1 ZE
2 D8:D PH8
  D8 S3 DO
  D1 HG                          ;relaxation delay
  D8 DO S1
  P1:D PH1                       ;90 deg H-1 pulse
  D2                             ;1/(2J)XH
  (P2 PH1):D (P4 PH7)            ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                             ;recovery delay
  P3 PH11                        ;90 deg X pulse
  P1:D PH1                       ;90 deg H-1 pulse
  D2                             ;1/(2J)XH
  P3 PH3                         ;90 deg X pulse
  D0                             ;t1/2
  P2:D PH2                       ;180 deg H-1 pulse
  D0
  P3 PH4
  D2                             ;1/(2J)XH  refocussing delay
  D5                             ;DE/2
3 D5:D PH8
4 D6 ADC                         ;D6 = 2 usec
  P0      :C7
5 P8*0.339:C7:T:C3 PH10          ;GARP1 decoupling
  P8*0.613:C7:T:C3 PH12
  P8*2.864:C7:T:C3 PH10
  P8*2.981:C7:T:C3 PH12
  P8*0.770:C7:T:C3 PH10
  P8*0.691:C7:T:C3 PH12
  P8*0.944:C7:T:C3 PH10
  P8*1.020:C7:T:C3 PH12
  P8*1.494:C7:T:C3 PH10
  P8*2.846:C7:T:C3 PH12
  P8*0.738:C7:T:C3 PH10
  P8*0.510:C7:T:C3 PH12
  P8*0.283:C7:T:C3 PH10
  P8*0.808:C7:T:C3 PH12
  P8*1.328:C7:T:C3 PH10
  P8*1.536:C7:T:C3 PH12
  P8*2.871:C7:T:C3 PH10
  P8*0.721:C7:T:C3 PH12
  P8*0.788:C7:T:C3 PH10
  P8*0.858:C7:T:C3 PH12
  P8*1.091:C7:T:C3 PH10
  P8*1.484:C7:T:C3 PH12
  P8*2.843:C7:T:C3 PH10
  P8*0.729:C7:T:C3 PH12
  P8*0.593:C7:T:C3 PH10
  L0 TO 5 TIMES 2
6 P8*0.339:C7:T:C3 PH12
  P8*0.613:C7:T:C3 PH10
```

```
  P8*2.864:C7:T:C3 PH12
  P8*2.981:C7:T:C3 PH10
  P8*0.770:C7:T:C3 PH12
  P8*0.691:C7:T:C3 PH10
  P8*0.944:C7:T:C3 PH12
  P8*1.020:C7:T:C3 PH10
  P8*1.494:C7:T:C3 PH12
  P8*2.846:C7:T:C3 PH10
  P8*0.738:C7:T:C3 PH12
  P8*0.510:C7:T:C3 PH10
  P8*0.283:C7:T:C3 PH12
  P8*0.808:C7:T:C3 PH10
  P8*1.328:C7:T:C3 PH12
  P8*1.536:C7:T:C3 PH10
  P8*2.871:C7:T:C3 PH12
  P8*0.721:C7:T:C3 PH10
  P8*0.788:C7:T:C3 PH12
  P8*0.858:C7:T:C3 PH10
  P8*1.091:C7:T:C3 PH12
  P8*1.484:C7:T:C3 PH10
  P8*2.843:C7:T:C3 PH12
  P8*0.729:C7:T:C3 PH10
  P8*0.593:C7:T:C3 PH12
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
  P0      :C7
8 RCYC=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH7=0
PH9=2
PH11=0 0 0 0 2 2 2 2
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0

PH8=0

PH10=0
PH12=2



;D8 : > 400 usec
;S3 : power level for presaturation
;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;D5 : DE/2
;D6 = 2 usec
;P0 = 5 usec
;P8 : 90 deg pulse for X decoupling
;L2 : L2 * 31.75 * 4 * P9 => AQ
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
```

```
;MC2 = W
```

BIRDPH.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using BIRD Sequence In Inverse Mode. Phase Sensitive Using TPPI. No Decoupling During Acquisition

```
;(see also INVH1.AU)
;A.Bax and S. Subramanian, J. Magn. Reson. 67, 565-569 (1986)

  II
1 ZE
2 D1 S1 DO                      ;relaxation delay
  P1:D PH1                      ;90 deg H-1 pulse
  D2                            ;1/(2J)XH
  (P2 PH1):D (P4 PH7)           ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                            ;recovery delay
  P3 PH10                       ;90 deg X pulse
3 P1:D PH1                      ;90 deg H-1 pulse
4 D2                            ;1/(2J)XH
5 P3 PH3                        ;90 deg X pulse
  D0                            ;t1/2
6 P2:D PH2                      ;180 deg H-1 pulse
  D0
7 P3 PH4
  D2                            ;1/(2J)XH  refocussing delay
9 GO=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH7=0
PH9=2
PH10=0 0 0 0 2 2 2 2
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0



;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
;MC2 = W
```

COLOC.AUR

; X-H Shift Correlation By Long-Range Coupling And With H-H Decoupling.(See Also COSYDEC.AU). Particularly Good For Quaternary Carbons

```
; KESSLER ET AL, J.MAGN.RES. 57, 331 (84)
; F2=X-NUCLEUS SHIFTS, F1=1H SHIFTS
```

```
; 1H: D1-90-D0-180- (D2-D0) -90-   BB
;  X:        -180-          -90-D3-FID

1 ZE
2 D1 S1 DO      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG H PULSE
4 (D2)  (D0 P2 PH6):D  (D0 P4 PH4)    ;EVOLUTION OF H SHIFTS
                                      ;ONLY, D2 IS POLARIZATION
                                      ;DELAY FOR LONG-RANGE J
5 (P1 PH2):D  (P3 PH3)  ;POLARIZATION TRANSFER
6 D3 S2                 ;REFOCUSSING DELAY
7 GO=2 PH5 CPD          ;ACQUIRE WITH DEC.
8 D3 DO                 ;GATE DEC. OFF
9 WR #1                 ;STORE FID
10 IF #1                ;INCREMENT FILE NUMBER
11 IN=1                 ;LOOP FOR NEXT EXPERIMENT
   EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2
    A1 A1 A1 A1 A3 A3 A3 A3
PH5=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH6=B0 B0 B0 B0 B2 B2 B2 B2

;D2 MUST BE >=2*D0 + (P2 OR P4) + 2(NE-1)*IN
; OR EXACTLY =2*D0 + (P2 OR P4) +  (NE-1)*IN (SEE COSYDEC.AU)
; NB: D2,D3 SHOULD HAVE TWICE THE VALUES DETERMINED IN AN
; OPTIMIZED INEPTRD EXPERIMENT THAT GIVES GOOD INTENSITY FOR
; THE DESIRED SIGNALS. FOR A SINGLE LONG-RANGE COUPLING
; THE OPTIMUM D2 = 1/(2J) = 2*D3.

;P1,P2=90,180 FOR 1H
;P3,P4=90,180 FOR X
;S1=0H FOR PULSING
;S2 = OPTIMAL POWER FOR CPD DEC. (P9).
;ND0=2 BECAUSE OF ECHO TECHNIQUE
;NS=4*N
;RD=PW=0
;D0=3 USEC
;NB: USE REV=Y FOR TRANSFORM, SINCE PULSE SEQUENCE REVERSES
;    SENSE OF THE T1 TIME DOMAIN.
```

CONOEPH.AUR

; Combined COSY And NOESY Shift-Correlated 2-D (TPPI Mode)

```
; COSY DATA ACQUIRED DURING THE MIXING TIME
; TWO DATA BLOCKS ARE USED WITH THE 'ADC','STA' COMMANDS
; SEE HAASNOOT ET AL, J.MAGN.RES. 56, 343 (84)
;     GUVENICH ET AL, J.MAGN.RES. 56, 471 (84)

;  D1-90-D0-D2-90-D2-FID(1)-D9-90-FID(2)

1 ST0   ;ZERO MEMORY BLOCK 1
  ZE
  ST    ;ZERO MEMORY BLOCK 2
```

```
  ZE
2 STA0           ;BLOCK 1
  D1             ;RELAXATION DELAY
  P1 PH1         ;90 DEG PREPARATION PULSE
  D0             ;EVOLUTION
3 D2             ;FIXED DELAY WHEN DESIRED TO EMPHASIZE COSY
                 ;  EFFECTS OF SMALL J
  P1 PH2         ;90 DEG MIXING PULSE
  D2             ;FIXED DELAY
4 D5
  D5 PH5         ;SET DETECTOR PHASE FOR COSY (2*D5=DE)
5 D6 ADC         ;ACQUIRE COSY DATA (1 TRANSIENT, D6>=AQ)
                 ; USING RECEIVER PHASE PROGRAM FROM 'GO'
6 EOSC           ;END OF SCAN, PROCEED WITH SEQUENCE WITHOUT
                 ; INCREMENTING SCAN COUNTER
7 STA            ;BLOCK 2
  D9             ;RANDOMIZED MIXING TIME
  P1 PH3         ;90 DEG DETECTION PULSE
8 GO=2 PH4       ;ACQUIRE NOESY DATA (1 TRANSIENT), INCREMENT
                 ; SCAN COUNTER, LOOP
9 ST0            ;BLOCK 1, STORE COSY FID
  WR #1          /DEFINE COSY FID
  IF #1
  ST             ;BLOCK 2, STORE NOESY FID
  WR #2          /DEFINE NOESY FID
  IF #2
  IP1            ;INCREMENT PH1 FOR TPPI
  IN=1           ;LOOP FOR NEXT EXPERIMENT INCREMENTING D0
  EXIT

;PHASE PROGRAM POINTERS FOLLOW SCAN COUNTER

PH1=0 0 0 0 1 1 1 1     ;SCANS 1-2 SUPPRESS AXIAL PEAKS
    2 2 2 2 3 3 3 3     ;3-4 SUPPRESS NOESY DBL-QUANTUM
                        ;FURTHER 90 DEG INCREMENTS FOR CYCLOPS

PH2=0 2 0 2 1 3 1 3
    2 0 2 0 3 1 3 1

PH3=0 0 2 2 1 1 3 3
    2 2 0 0 3 3 1 1

PH4= R0 R2 R2 R0 R1 R3 R3 R1
     R2 R0 R0 R2 R3 R1 R1 R3

PH5=0 2 2 0             ;DETECTOR PHASES FOR ADC COMMAND NEEDED
                        ; BECAUSE COSY REQUIRES DIFFERENT PHASES
                        ; COMPARED TO NOESY.


;NS=4*N, DS=2 OR 4
;NE = NO. EXPERIMENTS = TD1
;NBL=2 , NBL*SI MEMORY RESERVED FOR 'ST'
;RD=PW=0
;D1 = RELAXATION DELAY (>T1)
;D2 = CA. 0.25/J TO MAXIMIZE COSY CROSS-PEAKS FOR SMALL J
;  OR USE 0.05-0.1 TO MODERATELY ENHANCE EFFECTS OF SMALL J.
;P1=90 DEG
;D0=3E-6 INITIAL DELAY
;D5=DE/2 USED TO DEFINE DETECTOR PHASE FOR 'ADC'
;D6 MUST BE >= AQ FOR DIGITIZATION WITH 'ADC' COMMAND
;D9  SET SO THAT D2+D6+D9 = DESIRED MIXING TIME
;V9: THE MIXING TIME D9 WILL BE VARIED RANDOMLY OVER A RANGE
;  +/- V9% OF ITS VALUE; CHOOSE V9 SO THAT D9 IS VARIED BY
;  CA. +/- 20 MSEC TO CANCEL J CROSS PEAKS BETWEEN SPINS WITH
;  SHIFT DIFFERENCES > 1/0.020=50 HZ.
```

```
;MC2=W, REV=Y;  ND0=2 SO THAT IN=DW
;FOR I2D=0.5 SET NE=SI/4, SW1=SW/2, SI=TD, SI1=2*TD1
```

CONOESY.AUR

```
; Combined COSY And NOESY Shift-Correlated 2-D (Magnitude Mode)
; COSY DATA ACQUIRED DURING THE MIXING TIME
; TWO DATA BLOCKS ARE USED WITH THE 'ADC','STA' COMMANDS
; SEE HAASNOOT ET AL, J.MAGN.RES. 56, 343 (84)
;     GUVENICH ET AL, J.MAGN.RES. 56, 471 (84)

;  D1-90-D0-D2-90-D2-FID(1)-D9-90-FID(2)

1 ST0   ;ZERO MEMORY BLOCK 1
  ZE
  ST    ;ZERO MEMORY BLOCK 2
  ZE
2 STA0          ;BLOCK 1
  D1            ;RELAXATION DELAY
  P1 PH1        ;90 DEG PREPARATION PULSE
  D0            ;EVOLUTION
3 D2            ;FIXED DELAY WHEN DESIRED TO EMPHASIZE COSY
                ;  EFFECTS OF SMALL J
  P1 PH2        ;90 DEG MIXING PULSE
  D2            ;FIXED DELAY
4 D5
  D5 PH5        ;SET DETECTOR PHASE FOR COSY (2*D5=DE)
5 D6 ADC        ;ACQUIRE COSY DATA (1 TRANSIENT, D6>=AQ)
                ; USING RECEIVER PHASE PROGRAM FROM 'GO'
6 EOSC          ;END OF SCAN, PROCEED WITH SEQUENCE WITHOUT
                ; INCREMENTING SCAN COUNTER
7 STA           ;BLOCK 2
  D9            ;RANDOMIZED MIXING TIME
  P1 PH3        ;90 DEG DETECTION PULSE
8 GO=2 PH4      ;ACQUIRE NOESY DATA (1 TRANSIENT), INCREMENT
                ; SCAN COUNTER, LOOP
9 ST0           ;BLOCK 1, STORE COSY FID
  WR #1         /DEFINE COSY FID
  IF #1
  ST            ;BLOCK 2, STORE NOESY FID
  WR #2         /DEFINE NOESY FID
  IF #2
  IN=1          ;LOOP FOR NEXT EXPERIMENT INCREMENTING D0
  EXIT

;PHASE PROGRAM POINTERS FOLLOW SCAN COUNTER

PH1=0 0 0 0 0 0 0 0     ;SCANS 1-2 SUPPRESS AXIAL PEAKS
    1 1 1 1 1 1 1 1     ;3-4 FOR F1-QUAD (N-TYPE)
    2 2 2 2 2 2 2 2     ;5-8 SUPPRESS NOESY DBL-QUANTUM
    3 3 3 3 3 3 3 3     ;FURTHER 90 DEG INCREMENTS FOR CYCLOPS

PH2=0 2 1 3 0 2 1 3
    1 3 2 0 1 3 2 0
    2 0 3 1 2 0 3 1
    3 1 0 2 3 1 0 2

PH3=0 0 1 1 2 2 3 3
    1 1 2 2 3 3 0 0
    2 2 3 3 0 0 1 1
    3 3 0 0 1 1 2 2

PH4= R0 R2 R2 R0 R2 R0 R0 R2
     R1 R3 R3 R1 R3 R1 R1 R3
     R2 R0 R0 R2 R0 R2 R2 R0
     R3 R1 R1 R3 R1 R3 R3 R1

PH5=0 2 0 2 2 0 2 0     ;DETECTOR PHASES FOR ADC COMMAND NEEDED
                        ; BECAUSE COSY REQUIRES DIFFERENT PHASES
```

```
                        ; COMPARED TO NOESY.

;NS=8*N, DS=2 OR 4
;NE = NO. EXPERIMENTS = TD1
;NBL=2 , NBL*SI MEMORY RESERVED FOR 'ST'
;RD=PW=0
;D1 = RELAXATION DELAY (>T1)
;D2 = CA. 0.25/J TO MAXIMIZE COSY CROSS-PEAKS FOR SMALL J
;  OR USE 0.05-0.1 TO MODERATELY ENHANCE EFFECTS OF SMALL J.
;P1=90 DEG
;D0=3E-6 INITIAL DELAY
;D5=DE/2 USED TO DEFINE DETECTOR PHASE FOR 'ADC'
;D6 MUST BE >= AQ FOR DIGITIZATION WITH 'ADC' COMMAND
;D9  SET SO THAT D2+D6+D9 = DESIRED MIXING TIME
;V9: THE MIXING TIME D9 WILL BE VARIED RANDOMLY OVER A RANGE
;  +/- V9% OF ITS VALUE; CHOOSE V9 SO THAT D9 IS VARIED BY
;  CA. +/- 20 MSEC TO CANCEL J CROSS PEAKS BETWEEN SPINS WITH
;  SHIFT DIFFERENCES > 1/0.020=50 HZ.
;
;FILE #1 FOR COSY, FILE #2 FOR NOESY
;ND0=1, I2D=1, SW1=SW/2, SI=TD, SI1=2*TD1
```

COSY.AUR

```
; Homonuclear Shift-Correlated 2-D NMR (Jeener)
; W.P.AUE, E.BARTHOLDI, R.R.ERNST, J.CHEM.PHYS. 64, 2229 (1976)
; K.NAGAYAMA ET AL, J.MAGN.RES. 40, 321 (1980)

;  D1 - 90 - D0 - 90 OR 45 - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; SCALAR COUPLING J.

1 ZE
2 D1              ;RELAXATION
3 P1 PH1          ;90 DEG EXCITATION PULSE
4 D0              ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P2 PH2          ;MIXING PULSE, 90 OR 45 DEG
6 GO=2 PH3        ;ACQUIRE FID
7 WR #1           ;STORE FID
8 IF #1           ;INCREMENT FILE NUMBER
9 IN=1            ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
10 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;PHASE PROGRAMS CANCEL AXIAL
    A2 A2 A2 A2 A3 A3 A3 A3     ;PEAKS (SCANS 1-2), SELECT N-TYPE
                                ;PEAKS (SCANS 3-4), SUPPRESS F2
PH2=A0 A2 A1 A3 A1 A3 A2 A0     ;QUAD IMAGES (SCANS 5-8), AND CANCEL
    A1 A3 A2 A0 A2 A0 A3 A1     ;ARTEFACTS FROM P1 (SCANS 9-16).

PH3=R0 R0 R2 R2 R1 R1 R3 R3

PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG
;P2 = 90 DEG FOR MAX. SENSITIVITY
;   = 45 DEG FOR MINIMAL DIAGONAL (GOOD FOR TIGHT AB SYSTEMS)
;        AND 'TILTED' CORREL. PEAKS (SIGNS OF COUPLINGS).
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;CHOOSE SW AND SI SO THAT HZ/PT = CA. 2-6 HZ
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL
```

COSY11.AU

```
; Using 1-1bar Water Suppression Homonuclear Shift-Correlated 2-D NMR (Jeener)
; W.P.AUE, E.BARTHOLDI, R.R.ERNST, J.CHEM.PHYS. 64, 2229 (1976)
; K.NAGAYAMA ET AL, J.MAGN.RES. 40, 321 (1980)

;  D1 - 90(1 - 1BAR) - D0 - 90(1 - 1BAR) - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; SCALAR COUPLING J.

1 ZE
2 D1              ;RELAXATION
3 P1 PH1          ;45 DEG PULSE
  D2              ;1/D2 GIVES DISTANCE TO NEXT NULL
  P1 PH3          ;45 DEG PULSE
4 D0              ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P1 PH2          ;MIXING PULSE (1 - 1BAR)
  D2
  P1 PH4
6 GO=2 PH5        ;ACQUIRE FID
7 WR #1           ;STORE FID
8 IF #1           ;INCREMENT FILE NUMBER
9 IN=1            ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
10 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;PHASE PROGRAMS CANCEL AXIAL
    A2 A2 A2 A2 A3 A3 A3 A3     ;PEAKS (SCANS 1-2), SELECT N-TYPE
                                ;PEAKS (SCANS 3-4), SUPPRESS F2
PH2=A0 A2 A1 A3 A1 A3 A2 A0     ;QUAD IMAGES (SCANS 5-8), AND CANCEL
    A1 A3 A2 A0 A2 A0 A3 A1     ;ARTEFACTS FROM P1 (SCANS 9-16).

PH3=A2 A2 A2 A2 A3 A3 A3 A3
    A0 A0 A0 A0 A1 A1 A1 A1

PH4=A2 A0 A3 A1 A3 A1 A0 A2
    A3 A1 A0 A2 A0 A2 A1 A3

PH5=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;USE  NS = 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 45 DEG
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2
;SET O1 TO BE ON RESONANCE ON WATER SIGNAL

;CHOOSE SW AND SI SO THAT HZ/PT = CA. 2-6 HZ
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL
```

COSYDEC.AUR

```
; Cosy With F1 Decoupling
; A.BAX & R.FREEMAN,  J.MAGN.RES. 44,542 (81)

;  D1-90-T1/2-180-T1/2-(D3-T1)-(90 OR 45)-D2-AQ(FID)
;        D3--------------------

1 ZE
2 D1                ;RELAXATION
3 P1 PH1            ;90 DEG
4 (D3) (D0 P2 PH2)  ;EVOLUTION OF CHEM. SHIFTS VARIES WITH
                    ;TIMING OF 180 PULSE
5 P3 PH3        ;90 OR 45 DEG PULSE FOR DETECTION
  D2            ; EXTRA DELAY
                ;IF DESIRED TO EMPHASIZE EFFECTS FROM SMALL J.
6 GO=2 PH4      ;ACQUIRE
7 WR #1
8 IF #1
9 IN=1
  EXIT

PH1=A0 A0 A0 A0 A0 A0 A0 A0     ;PHASE PROGRAMS ALTERNATE 180
    A1 A1 A1 A1 A1 A1 A1 A1     ;PULSE PHASE (SCANS 5-8), SELECT
PH2=A1 A1 A1 A1 A3 A3 A3 A3     ;COHERENCE TRANSFER ECHO (N-TYPE)
    A2 A2 A2 A2 A0 A0 A0 A0     ;(SCANS 3-4), SUPPRESS AXIAL PEAKS
PH3=A0 A2 A1 A3 A0 A2 A1 A3     ;(SCANS 1-2), SUPPRESS F2 QUAD
    A1 A3 A2 A0 A1 A3 A2 A0     ;IMAGES (SCANS 9-16)
PH4=R0 R0 R2 R2 R0 R0 R2 R2
    R1 R1 R3 R3 R1 R1 R3 R3

;NS=N*4(N=1,2,3....)
;P1=90 DEG   P2=180 DEG
;P3=45 DEG ACCORDING TO FREEMAN BUT 90 DEG GIVES BETTER
;  COHERENCE TRANSFER AND STRONGER CROSS PEAKS.
;THE MIN. D3 MUST= (NE-1)*IN+P2+2*D0, WHEN P2 IS STEPPED
;THROUGH ENTIRE D3 RANGE (MAX. NE=TD/2). EACH DISCREET T1 EVOL.
; PERIOD WILL OCCUR TWICE DURING THE SEQUENCE. IN THIS CASE
; ADJUST D0 SO THAT D3 CAN BE EXPRESSED EXACTLY BY 12 BITS
;(IE. DECIMAL  DIGITS 1-4095).

;SIMPLER IS TO USE NE=TD/4, D3 CAN BE >= P2+2*D0+2(NE-1)*IN; EACH
;DISCREET T1 PERIOD OCCURS ONLY ONCE.

;NORMALLY D2 IS SMALL (EG. 2 USEC).
;TO ENHANCE LONG-RANGE J EFFECTS (AS IN COSYLR.AU)
; USE A LONGER D2. NOTE THAT
; OPTIMUM CROSS PEAKS OCCUR WHEN D2=(2N+1)/(4J) AND CERTAIN
; CHOICES OF D2 MAY CAUSE LOSS OF CERTAIN CROSS PEAKS.

;FREEMAN SUGGESTS  ADJUSTING D3 SO MIDDLE OF AQ OCCURS AT TIME
;D3 AFTER P3 (PSEUDO ECHO); PERHAPS BETTER IS D3=D2-2*(NE-1)*IN
; TO GIVE A SYMMETRICAL MIXING PERIOD FOLLOWING EVOLUTION.

;ND0=2 BECAUSE OF ECHO AFTER 180 PULSE, CAUSING T1 PERIODS TO
; INCREMENT BY 2*IN.

;180 PULSE MUST BE EXACT; ARTEFACTS OCCUR IN F1 MIDWAY BETWEEN
;COUPLED SPINS.

;NB: REV=Y MUST BE USED SINCE T1-DOMAIN IS REVERSED BY
;  THE ECHO SEQUENCE USED.

;NB: MATRIX MAY BE SYMMETRIZED!!(WITH SOME CHANGES IN MULTIPLET
;  STRUCTURE IN F2 DOMAIN).
```

COSYDQF.AUR

```
; Cosy-90 With Double Quantum Filter
;WOKAUN & ERNST, CHEM. PHYS. LETT. 52, 407(77)
;PIANTINI ET AL., JACS 104, 6800(82)
;SHAKA & FREEMAN, J.MAGN.RES. 51, 169(83)

;  D1-90-D0-D2-90-D3-90-D2-FID

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG EXCITATION
4 D0            ;EVOLUTION
  D2            ;FIXED DELAY TO ENHANCE MQC
5 P1 PH2        ;GENERATE MULTIPLE QUANTUM COHERENCE
6 D3            ;3 USEC FOR PHASE SWITCHING
7 P1 PH3        ;SELECTION PULSE
  D2
8 GO=2 PH4      ;ACQUIRE
9 WR #1
10 IF #1
11 IN=1
   EXIT

PH1=A0 A2 A1 A3 A1 A3 A0 A2
    A1 A3 A2 A0 A2 A0 A1 A3
PH2=A0 A2 A1 A3 A2 A0 A1 A3
    A1 A3 A2 A0 A3 A1 A2 A0
PH3=A0 A0 A0 A0 A1 A1 A1 A1
    A1 A1 A1 A1 A2 A2 A2 A2
PH4=R0 R0 R2 R2 R0 R0 R2 R2
    R1 R1 R3 R3 R1 R1 R3 R3
;PHASE PROGRAMS SELECT FOR QUANTA OF ORDER 2,6,10,...(SCANS 1-4),
;  WITH N-TYPE SELECTION (SCANS 5-8).
;NB:  NS=8*N, MINIMUM OF 8 TRANS.

;RD=PW=0
;D1=1-5*T1
;P1=90 DEG PULSE
;D0=5E-6
;D3=3E-6
;D2=0.05-0.3 SEC TO ENHANCE INTENSITY OF MQC

;REQUIRES TWICE AS MANY TRANSIENTS FOR SAME S/N AS
;  FOR NORMAL COSY.
```

COSYHG.AUR

```
; Homonuclear Shift-Correlated 2-D NMR (Jeener) Using  Pre-Saturation Of
  Solvent With Two Power Levels.
; ALSO ALLOWS FOR FIXED DELAY AS IN COSYLR.AU

;  HG(S1)-(S2)-------------------------- DO
;  D1 -   D3 - 90 - D0 -D2-90 OR 45-D2 - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; SCALAR COUPLING J.

1 ZE
2 D1 HG S1      ;RELAXATION, PRE-SATURATION WITH POWER S1
  D3 S2         ;SWITCH TO MIN. POWER S2 FOR EVOLUTION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
  D2            ;FIXED DELAY TO ENHANCE EFFECTS FROM SMALL J
5 P2 PH2        ;MIXING PULSE, 90 OR 45 DEG
  D2
6 GO=2 PH3 DO   ;ACQUIRE FID WITH DEC. GATED OFF
7 WR #1         ;STORE FID
8 IF #1         ;INCREMENT FILE NUMBER
9 IN=1          ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
10 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;D2 = FIXED DELAY FOR ENHANCEMENT OF SMALL J EFFECTS (COSYLR)
;D3 = 2 MSEC TO SWITCH DEC. POWER
;S1 DEFINES DEC. POWER FOR PRE-SATURATION, SHOULD BE AS LOW
;  AS POSSIBLE TO AVOID SATURATING NEIGHBORING GROUPS (EG. 30L),
; S2 SHOULD BE EVEN LOWER TO AVOID BLOCH-SIEGERT EFFECTS DURING D0.
;P1 = 90 DEG
;P2 = 90 DEG FOR MAX. SENSITIVITY
;   = 45 DEG FOR MINIMAL DIAGONAL (GOOD FOR TIGHT AB SYSTEMS)
;        AND 'TILTED' CORREL. PEAKS (SIGNS OF COWPLINGS).
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;N-TYPE PEAK SELECTION
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
```

COSYLR.AUR

; Homonuclear Shift-Correlated 2-D NMR (Jeener) With Delay Period To Emphasize Long-Range Or Small Couplings

```
; A.BAX & R.FREEMAN, J.MAGN.RES. 44, 542 (1981)

;  D1 - 90 - D0 - D2 - 90 OR 45 - D2 - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; SCALAR COUPLING J.

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
  D2            ;FIXED DELAY TO EMPHASIZE EFFECT OF SMALL J
5 P2 PH2        ;MIXING PULSE, 90 OR 45 DEG
  D2            ;WAIT FOR EFFECT OF SMALL J
6 GO=2 PH3      ;ACQUIRE FID
7 WR #1         ;STORE FID
8 IF #1         ;INCREMENT FILE NUMBER
9 IN=1          ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
10 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;PHASES FOR N-TYPE SELECTION
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;USE QP, NS = 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG
;P2 = 90 DEG FOR MAX. SENSITIVITY
;   = 45 DEG FOR MINIMAL DIAGONAL (GOOD FOR TIGHT AB SYSTEMS)
;        AND 'TILTED' CORREL. PEAKS (SIGNS OF COUPLINGS).
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;D2 = CA. 0.25/J TO MAXIMIZE CORREL. PEAKS FROM SMALL J
; (INTENSITY FOLLOWS FUNCTION SIN(PI*J*D0)).
; TYPICALLY D2=0.08 SEC IS SUFFICIENT TO ENHANCE EFFECTS FROM
; SMALL J SO THAT WEAK CROSS PEAKS ARE OBSERVABLE EVEN IF
; HZ/PT IS 4-6. NB: T2 RELAXATION DURING D2 WILL REDUCE ALL
; INTENSITIES.
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL
```

COSYPDHG.AUR

; Cosy 2-D Data Acquisition For Phase-Sensitive Mode With DQF And Presaturation Of Solvent. With Pure Absorption/Dispersion Lineshapes Without Phase-Twist, And Quad Detection In F1, Using Time-Proportional Phase Increments (TPPI).

```
; MARION & WUETHRICH, BBRC 113, 967 (83)
; RANCE ET AL., BBRC 117, 479 (83)
; REQUIRES NEW 2-D TRANSFORM AND DISPLAY ROUTINES (MC2=W).

; D1 - 90 - D0 - 90 - D3 - 90 - FID

1 ZE
2 D1 HG S1      ;RELAXATION, PRESATURATION WITH POWER S1
  D4 S2         ;REDUCE DEC. POWER TO A MINIMUM
3 P1 PH1        ;90 DEG PREPARATION PULSE
4 D0            ;EVOLUTION
5 P1 PH2        ;90 DEG MIXING PULSE, GENERATE DQ
  D3            ;FIXED DELAY FOR PHASE SWITCHING
  P1 PH3        ;DQF SELECTION
6 GO=2 PH4 DO   ;DEC. OFF, ACQUIRE DATA, LOOP TO 2
7 WR #1         ;STORE FID
8 IF #1         ;INCREMENT FILE NUMBER
9 IP1           ;INCREMENT PHASE PROGRAM PH1 (TPPI)
10 IN=1         ;INCREMENT D0 AND LOOP TO 1 FOR NEXT EXPERIMENT
   EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1 A2 A2 A2 A2 A3 A3 A3 A3
PH2=A0 A0 A0 A0 A1 A1 A1 A1 A2 A2 A2 A2 A3 A3 A3 A3
PH3=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2
PH4=R0 R3 R2 R1 R1 R0 R3 R2 R2 R1 R0 R3 R3 R2 R1 R0

;TRANSFORM REQUIRES REDF=N AND REV=Y

;RECOMMEND SI=2*TD  FOR DESIRED RESOLUTION OF J COUPLINGS IN F2.
;THE PHASE SENSITIVE MATRIX (SI X SI1) WILL HAVE 4 QUADRANTS
; AND THE RR QUADRANT WILL BE USED FOR DISPLAY AND PLOTTING.
; EQUAL DIGITAL RES. IN F1 AND F2 REQUIRES SI1=SI, MAX. NE=TD.
; IT MAY BE DESIRABLE TO USE LESS DIGITAL RES. IN F1, BUT 'SYM'
; IS NO LONGER POSSIBLE.

;IN=DW (ONE-HALF ITS NORMAL VALUE, E.G.SET ND0=2, SW1=SW/2)

;D1=1-5*T1, S1 = DEC. POWER TO PRESATURATE
;D4=2 MSEC, S2 = MINIMUM DEC. POWER TO HOLD SATURATION AND
;                MINIMIZE BLOCH-SIEGERT SHIFTS.
;P1=90 DEG
;D0=3E-6
;D3=3E-6
;RD=PW=0
;NS=4, 8 OR 16*N (FULL PHASE CYCLE),  DS=2 OR 4

;TO PERFORM PHASE CORRECTION: ONE A 1-D SPECTRUM USING
; PW=90 DEG, RD=D1, AND PHASE CORRECT IN EP. NOTE PHASE
; CONSTANTS (USE 'TY' COMMAND). SUBTRACT 90 DEG FROM
; ZERO-ORDER PHASE AND ENTER AS PC0, ENTER FIRST-ORDER
; PHASE AS PC1, TYPE 'PZ' TO FIX THESE IN MEMORY (CHECK
; WITH 'TY'). PERFORM XFB WHICH THEN USES PC0,PC1 FOR F2
; DOMAIN AND ZERO PHASE CORRECTION FOR F1.
```

COSYPDQ2.AUR

;MODIFIED PHASECYCLE

; COSY 2-D Data Acquisition For Phase-Sensitive Mode With DQF With Pure Absorption/Dispersion Lineshapes Without Phase-Twist, And Quad Detection In F1, Using Time-Proportional Phase Increments (TPPI).

```
; MARION & WUETHRICH, BBRC 113, 967 (83)
; RANCE ET AL., BBRC 117, 479 (83)
; REQUIRES NEW 2-D TRANSFORM AND DISPLAY ROUTINES (MC2=W).

; D1 - 90 - D0 - 90 - D3 - 90 - FID
```

```
1 ZE
2 D1               ;RELAXATION
3 P1 PH1           ;90 DEG PREPARATION PULSE
4 D0               ;EVOLUTION
5 P1 PH2           ;90 DEG MIXING PULSE, GENERATE DQ
  D3               ;FIXED DELAY FOR PHASE SWITCHING
  P1 PH3           ;DQF SELECTION
6 GO=2 PH4         ;ACQUIRE DATA, LOOP TO 2
7 WR #1            ;STORE FID
8 IF #1            ;INCREMENT FILE NUMBER
9 IP1              ;INCREMENT PHASE PROGRAM PH1 (TPPI)
10 IN=1            ;INCREMENT D0 AND LOOP TO 1 FOR NEXT EXPERIMENT
   EXIT


PH1=0 2 1 3
PH2=0 0 1 1 2 2 3 3
PH3=0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1
    2 2 2 2 2 2 2 2 3 3 3 3 3 3 3 3
PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R3 R1 R1 R3 R1 R3 R3 R1
    R2 R0 R0 R2 R0 R2 R2 R0
    R1 R3 R3 R1 R3 R1 R1 R3

;TRANSFORM REQUIRES REDF=N AND REV=Y

;RECOMMEND SI=2*TD  FOR DESIRED RESOLUTION OF J COUPLINGS IN F2.
;THE PHASE SENSITIVE MATRIX (SI X SI1) WILL HAVE 4 QUADRANTS
; AND THE RR QUADRANT WILL BE USED FOR DISPLAY AND PLOTTING.
; EQUAL DIGITAL RES. IN F1 AND F2 REQUIRES SI1=SI, MAX. NE=TD.
; IT MAY BE DESIRABLE TO USE LESS DIGITAL RES. IN F1, BUT 'SYM'
; IS NO LONGER POSSIBLE.

;IN=DW (ONE-HALF ITS NORMAL VALUE, E.G.SET ND0=2, SW1=SW/2)

;D1=1-5*T1
;P1=90 DEG
;D0=3E-6
;D3=3E-6
;RD=PW=0
;NS=16*N
;DS=2 OR 4

;TO PERFORM PHASE CORRECTION: ONE A 1-D SPECTRUM USING
; PW=90 DEG, RD=D1, AND PHASE CORRECT IN EP. NOTE PHASE
; CONSTANTS (USE 'TY' COMMAND). SUBTRACT 90 DEG FROM
; ZERO-ORDER PHASE AND ENTER AS PC0, ENTER FIRST-ORDER
; PHASE AS PC1, TYPE 'PZ' TO FIX THESE IN MEMORY (CHECK
; WITH 'TY'). PERFORM XFB WHICH THEN USES PC0,PC1 FOR F2
; DOMAIN AND ZERO PHASE CORRECTION FOR F1.
```

COSYPH.AUR

```
; COSY 2-D Data Acquisition For Phase-Sensitive Mode With Pure
  Absorption/Dispersion Lineshapes Without Phase-Twist, And Quad Detection In
  F1, Using Time-Proportional Phase Increments (TPPI).
; MARION & WUETHRICH, BBRC 113, 967 (1983)
; G.BODENHAUSEN, H.KOGLER, R.R.ERNST, J.MAGN.RES. 58, 370 (1984)

; REQUIRES NEW 2-D TRANSFORM AND DISPLAY ROUTINES (MC2=W).

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG PREPARATION PULSE
4 D0            ;EVOLUTION
5 P2 PH2        ;90 DEG MIXING PULSE
6 GO=2 PH3      ;ACQUIRE DATA, LOOP TO 2
7 WR #1         ;STORE FID
8 IF #1         ;INCREMENT FILE NUMBER
9 IP1           ;INCREMENT PHASE PROGRAM PH1 (TPPI)
10 IN=1         ;INCREMENT D0 AND LOOP TO 1 FOR NEXT EXPERIMENT
   EXIT

;PHASE PROGRAMS FOR AXIAL PEAK (F1=0) SUPPRESSION

PH1=A0 A2 A2 A0 A1 A3 A3 A1
PH2=A0 A2 A0 A2 A1 A3 A1 A3
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;TRANSFORM REQUIRES REDF=N AND REV=Y

;RECOMMEND SI=2*TD  FOR DESIRED RESOLUTION OF J COUPLINGS IN F2.
;THE PHASE SENSITIVE MATRIX (SI X SI1) WILL HAVE 4 QUADRANTS
; AND THE RR QUADRANT WILL BE USED FOR DISPLAY AND PLOTTING.
; EQUAL DIGITAL RES. IN F1 AND F2 REQUIRES SI1=SI, MAX. NE=TD.
; IT MAY BE DESIRABLE TO USE LESS DIGITAL RES. IN F1, BUT 'SYM'
; IS NO LONGER POSSIBLE.

;IN=DW (ONE-HALF ITS NORMAL VALUE, E.G.SET ND0=2, SW1=SW/2)

;D1=1-5*T1
;P1=90 DEG
;D0=3E-6
;RD=PW=0
;NS=4 OR 8*N (FULL PHASE CYCLE),  DS=2 OR 4

;TO DEFINE PHASE CORRECTION: SET PW=90 DEG, RD=D1 AND RUN 1-D
; SPECTRUM USING SAME SW, SI ETC. PERFORM PHASE CORRECTION IN
; EP AND EXAMINE PHASE CONSTANTS WITH 'TY'. PERFORM XFB AND PHASE
; VALUES SHOWN BY 'TY' WILL BE USED FOR F2 DOMAIN, ZERO PHASE
; CORRECTION IS USED IN F1 DOMAIN.
```

COSYPHDQ.AUR

; COSY 2-D Data Acquisition For Phase-Sensitive Mode With DQF With Pure Absorption/Dispersion Lineshapes Without Phase-Twist, And Quad Detection In F1, Using Time-Proportional Phase Increments (TPPI).

```
; MARION & WUETHRICH, BBRC 113, 967 (83)
; RANCE ET AL., BBRC 117, 479 (83)
; REQUIRES NEW 2-D TRANSFORM AND DISPLAY ROUTINES (MC2=W).

; D1 - 90 - D0 - 90 - D3 - 90 - FID

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG PREPARATION PULSE
4 D0            ;EVOLUTION
5 P1 PH2        ;90 DEG MIXING PULSE, GENERATE DQ
  D3            ;FIXED DELAY FOR PHASE SWITCHING
  P1 PH3        ;DQF SELECTION
6 GO=2 PH4         ;ACQUIRE DATA, LOOP TO 2
7 WR #1         ;STORE FID
8 IF #1         ;INCREMENT FILE NUMBER
9 IP1           ;INCREMENT PHASE PROGRAM PH1 (TPPI)
10 IN=1         ;INCREMENT D0 AND LOOP TO 1 FOR NEXT EXPERIMENT
   EXIT


PH1=A0 A0 A0 A0 A1 A1 A1 A1 A2 A2 A2 A2 A3 A3 A3 A3
PH2=A0 A0 A0 A0 A1 A1 A1 A1 A2 A2 A2 A2 A3 A3 A3 A3
PH3=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2
PH4=R0 R3 R2 R1 R1 R0 R3 R2 R2 R1 R0 R3 R3 R2 R1 R0

;TRANSFORM REQUIRES REDF=N AND REV=Y

;RECOMMEND SI=2*TD  FOR DESIRED RESOLUTION OF J COUPLINGS IN F2.
;THE PHASE SENSITIVE MATRIX (SI X SI1) WILL HAVE 4 QUADRANTS
; AND THE RR QUADRANT WILL BE USED FOR DISPLAY AND PLOTTING.
; EQUAL DIGITAL RES. IN F1 AND F2 REQUIRES SI1=SI, MAX. NE=TD.
; IT MAY BE DESIRABLE TO USE LESS DIGITAL RES. IN F1, BUT 'SYM'
; IS NO LONGER POSSIBLE.

;IN=DW (ONE-HALF ITS NORMAL VALUE, E.G.SET ND0=2, SW1=SW/2)

;D1=1-5*T1
;P1=90 DEG
;D0=3E-6
;D3=3E-6
;RD=PW=0
;NS=4, 8 OR 16*N (FULL PHASE CYCLE),  DS=2 OR 4

;TO PERFORM PHASE CORRECTION: ONE A 1-D SPECTRUM USING
; PW=90 DEG, RD=D1, AND PHASE CORRECT IN EP. NOTE PHASE
; CONSTANTS (USE 'TY' COMMAND). SUBTRACT 90 DEG FROM
; ZERO-ORDER PHASE AND ENTER AS PC0, ENTER FIRST-ORDER
; PHASE AS PC1, TYPE 'PZ' TO FIX THESE IN MEMORY (CHECK
; WITH 'TY'). PERFORM XFB WHICH THEN USES PC0,PC1 FOR F2
; DOMAIN AND ZERO PHASE CORRECTION FOR F1.
```

COSYPHHG.AUR

; COSY 2-D Data Acquisition For Phase-Sensitive Mode With Presaturation Of Solvent, For Pure Absorption/Dispersion Lineshapes Without Phase-Twist, And Quad Detection In F1, Using Time-Proportional Phase Increments (TPPI).

```
; MARION & WUETHRICH, BBRC 113, 967 (1983)
; REQUIRES NEW 2-D TRANSFORM AND DISPLAY ROUTINES (MC2=W).
```

```
1 ZE
2 D1 HG S1       ;RELAXATION, PRESATURATION WITH POWER S1
  D3 S2          ;REDUCE DEC. POWER TO A MINIMUM
3 P1 PH1         ;90 DEG PREPARATION PULSE
4 D0             ;EVOLUTION
5 P1 PH2         ;90 DEG MIXING PULSE
6 GO=2 PH3 DO    ;DEC. OFF, ACQUIRE DATA, LOOP TO 2
7 WR #1          ;STORE FID
8 IF #1          ;INCREMENT FILE NUMBER
9 IP1            ;INCREMENT PHASE PROGRAM PH1 (TPPI)
10 IN=1          ;INCREMENT D0 AND LOOP TO 1 FOR NEXT EXPERIMENT
   EXIT

;PHASE PROGRAMS FOR AXIAL PEAK (F1=0) SUPPRESSION AND QP

PH1=0 2 2 0 1 3 3 1
PH2=0 2 0 2 1 3 1 3
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;TRANSFORM REQUIRES REDF=N AND REV=Y

;RECOMMEND SI=2*TD  FOR DESIRED RESOLUTION OF J COUPLINGS IN F2.
;THE PHASE SENSITIVE MATRIX (SI X SI1) WILL HAVE 4 QUADRANTS
; AND THE RR QUADRANT WILL BE USED FOR DISPLAY AND PLOTTING.
; EQUAL DIGITAL RES. IN F1 AND F2 REQUIRES SI1=SI, MAX. NE=TD.
; IT MAY BE DESIRABLE TO USE LESS DIGITAL RES. IN F1, BUT 'SYM'
; IS NO LONGER POSSIBLE.

;IN=DW (ONE-HALF ITS NORMAL VALUE, E.G.SET ND0=2, SW1=SW/2)

;D1=1-5*T1, S1 = DEC. POWER TO PRESATURATE
;D3=2 MSEC, S2 = MINIMUM POWER TO HOLD SATURATION AND MINIMIZE
;                BLOCH-SIEGERT SHIFTS DURING EVOLUTION.
;P1=90 DEG
;D0=3E-6
;RD=PW=0
;NS=4, OR 8*N (FULL PHASE CYCLE),  DS=2 OR 4

;TO DEFINE PHASE CORRECTION: SET PW=90 DEG, RD=D1 AND RUN 1-D
; SPECTRUM USING SAME SW, SI ETC. PERFORM PHASE CORRECTION IN
; EP AND EXAMINE PHASE CONSTANTS WITH 'TY'. PERFORM XFB AND PHASE
; VALUES SHOWN BY 'TY' WILL BE USED FOR F2 DOMAIN, ZERO PHASE
; CORRECTION IS USED IN F1 DOMAIN.
```

COSYPHX.AUR

; Homonuclear X-Nucleus (1H-Decoupled) COSY 2-D Data Acquisition For Phase-Sensitive Mode With Pure Absorption/Dispersion Lineshapes Without Phase-Twist, And Quad Detection In F1, Using Time-Proportional Phase Increments (TPPI).

```
; MARION & WUETHRICH, BBRC 113, 967 (1983)
; REQUIRES NEW 2-D TRANSFORM AND DISPLAY ROUTINES (MC2=W).

1 ZE
2 D1 CPD S1     ;RELAXATION, BB DEC. WITH MIN. POWER S1
  D3 S2         ;INCREASE POWER FOR OPTIMAL DEC.
3 P1 PH1        ;90 DEG PREPARATION PULSE
4 D0            ;EVOLUTION
5 P1 PH2        ;90 DEG MIXING PULSE
6 GO=2 PH3      ;ACQUIRE DATA, LOOP TO 2
  D3 CPD S1     ;REDUCE DEC. POWER
7 WR #1         ;STORE FID
8 IF #1         ;INCREMENT FILE NUMBER
9 IP1           ;INCREMENT PHASE PROGRAM PH1 (TPPI)
10 IN=1         ;INCREMENT D0 AND LOOP TO 1 FOR NEXT EXPERIMENT
   EXIT

;PHASE PROGRAMS FOR AXIAL PEAK (F1=0) SUPPRESSION

PH1=0 2 2 0 1 3 3 1
PH2=0 2 0 2 1 3 1 3
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;TRANSFORM REQUIRES REDF=N AND REV=Y

;RECOMMEND SI=2*TD  FOR DESIRED RESOLUTION OF J COUPLINGS IN F2.
;THE PHASE SENSITIVE MATRIX (SI X SI1) WILL HAVE 4 QUADRANTS
; AND THE RR QUADRANT WILL BE USED FOR DISPLAY AND PLOTTING.
; EQUAL DIGITAL RES. IN F1 AND F2 REQUIRES SI1=SI, MAX. NE=TD.
; IT MAY BE DESIRABLE TO USE LESS DIGITAL RES. IN F1, BUT 'SYM'
; IS NO LONGER POSSIBLE.

;IN=DW (ONE-HALF ITS NORMAL VALUE, E.G.SET ND0=2, SW1=SW/2)

;D1=1-5*T1, S1 = CA. 0.5 WATT TO HOLD NOE
;D3= 2 MSEC, S2 = OPTIMUM CPD DEC. POWER (P9) TO MINIMIZE
;   LOSS OF TRANSVERSE MAGNETIZATION DURING EVOLUTION.
;P1=90 DEG
;D0=3E-6
;RD=PW=0
;NS=4, OR 8*N (FULL PHASE CYCLE),  DS=2 OR 4

;TO DEFINE PHASE CORRECTION: SET PW=90 DEG, RD=D1 AND RUN 1-D
; SPECTRUM USING SAME SW, SI ETC. PERFORM PHASE CORRECTION IN
; EP AND EXAMINE PHASE CONSTANTS WITH 'TY'. PERFORM XFB AND PHASE
; VALUES SHOWN BY 'TY' WILL BE USED FOR F2 DOMAIN, ZERO PHASE
; CORRECTION IS USED IN F1 DOMAIN.
```

COSYRCT.AUR

; COSY With 1-Step Relayed Coherence Transfer (Magnitude Mode), See Also RECOSY.AUR

```
; G.WAGNER, JMR 55, 151 (83).
; A.BAX & G.DROBNY, JMR 61,306 (85)

;  90-D0-90-D2-180-D2-90-FID

; CORRELATION CROSS-PEAKS CAN BE OBTAINED FROM SPINS A AND X
; IN AN AMX SYSTEM WHEN J(AX) IS TOO SMALL.

1 ZE
2 D1            ;RELAXATION DELAY
3 P1 PH1        ;90 DEG PULSE CREATES XY-MAGN.
4 D0            ;EVOLUTION OF SHIFTS
5 P1 PH2        ;COMPLETE FIRST COHERENCE TRANSFER, E.G.
                ; SPIN A TO M DEPENDS ON SIN(PI*J(AM)*D0)
6 D2            ;SECOND COHERENCE PERIOD
7 P2 PH2        ;REFOCUS CHEMICAL SHIFTS
8 D2
9 P1 PH3        ;COMPLETE SECOND TRANSFER (EG. M TO X)
                ; EFFICIENCY DEPENDS ON
                ; SIN(PI*2D2*J(AM))*SIN(PI*2D2*J(MX)
10 GO=2 PH4     ;ACQUIRE FID
11 WR #1        ;STORE FID IN .SER FILE
12 IF #1
13 IN=1         ;LOOP FOR NEXT EXPERIMENT
14 EXIT

PH1=A0 A0 A0 A0 A0 A0 A0 A0     ;SCANS 1-2 SUPPRESS AXIAL PEAKS
    A1 A1 A1 A1 A1 A1 A1 A1     ;SCANS 3-4 FOR F1 QUAD (N-TYPE)
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH2=A0 A0 A1 A1 A2 A2 A3 A3     ;SCANS 5-8 SUPPRESS NOESY PEAKS
    A1 A1 A2 A2 A3 A3 A0 A0
    A2 A2 A3 A3 A0 A0 A1 A1
    A3 A3 A0 A0 A1 A1 A2 A2
PH3=A0 A2 A1 A3 A0 A2 A1 A3
    A1 A3 A2 A0 A1 A3 A2 A0
    A2 A0 A3 A1 A2 A0 A3 A1
    A3 A1 A0 A2 A3 A1 A0 A2

PH4=R0 R0 R2 R2 R0 R0 R2 R2
    R1 R1 R3 R3 R1 R1 R3 R3
    R2 R2 R0 R0 R2 R2 R0 R0
    R3 R3 R1 R1 R3 R3 R1 R1

;D2 = CA. 0.5/(J(AM)+J(MX)) WHEN COUPLINGS DO NOT DIFFER BY
;      MORE THAN FACTOR 2
;    = CA. 0.5/( 1.6*J(MAX) ) OR AT MOST 0.5/( 1.3*J(MAX) )
;      TO COVER A WIDER RANGE OF J.
;      NULLING CONDITIONS OF D2=0.5/J SHOULD BE AVOIDED

;NS=8*N
;P1=90, P2=180
;  OTHERWISE PARAMETERS AS FOR COSY.
```

COSYRCT2.AUR

; Cosy With 2-Step Relayed Coherence Transfer (Magnitude Mode), See Also RECOSY2.AUR

```
; G.WAGNER, JMR 55, 151 (83).
```

```
; A.BAX & G.DROBNY, JMR 61,306 (85)

;  90-D0-90-D2-180-D2-90-D3-180-D3-90-FID

; CORRELATION CROSS-PEAKS CAN BE OBTAINED FOR SPIN A
; FROM SPINS M,Q,X IN AN AMQX SPIN SYSTEM.

1 ZE
2 D1            ;RELAXATION DELAY
3 P1 PH1        ;90 DEG PULSE CREATES XY-MAGN.
4 D0            ;EVOLUTION OF SHIFTS
5 P1 PH2        ;COMPLETE FIRST COHERENCE TRANSFER, E.G.
                ; SPIN A TO M DEPENDS ON SIN(PI*J(AM)*D0)
6 D2            ;SECOND COHERENCE PERIOD
7 P2 PH2        ;REFOCUS CHEMICAL SHIFTS
8 D2
9 P1 PH3        ;COMPLETE SECOND TRANSFER (EG. M TO Q)
10 D3
11 P2 PH2
12 D3
13 P1 PH4       ;THIRD TRANSFER FROM Q TO X
14 GO=2 PH5     ;ACQUIRE FID
15 WR #1        ;STORE FID IN .SER FILE
16 IF #1
17 IN=1         ;LOOP FOR NEXT EXPERIMENT
18 EXIT

PH1=A0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
                                ;SCANS 3-4 FOR F1 QUAD (N-TYPE)
                                ;SCANS 5-8 SUPPRESS NOESY PEAKS
PH2=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3

PH3=A0 A0 A2 A2 A1 A1 A3 A3

PH4=A0 A2 A0 A2 A1 A3 A1 A3

PH5=R0 R0 R0 R0 R2 R2 R2 R2

;FOR LINEAR SPIN SYSTEM AMQX, THE TRANSFER FUNCTION IS
; SIN(PI*J(AM)*2D2)SIN(PI*J(MQ)*2D2)*
; SIN(PI*J(MQ)*2D3)SIN(PI*J(QX)*2D3)
;SET D2 = CA. 0.5/(1.6*J), WHERE J = LARGER OF J(AM), J(MQ)
;SET D3 =      "     "        "                J(MQ), J(QX)

;NS=16*N
;P1=90, P2=180
;  OTHERWISE PARAMETERS AS FOR COSY.
```

COSYRCT3.AUR

; Cosy With 3-Step Relayed Coherence Transfer (Magnitude Mode), See Also RECOSY2.AUR

```
; G.WAGNER, JMR 55, 151 (83).
; A.BAX & G.DROBNY, JMR 61,306 (85)

;  90-D0-90-D2-180-D2-90-D3-180-D3-90-
;     D4-180-D4-90-FID

; CORRELATION CROSS-PEAKS CAN BE OBTAINED FOR SPIN A
; FROM SPINS M,Q,R,X IN AN AMQRX SPIN SYSTEM.

1 ZE
2 D1              ;RELAXATION DELAY
3 P1 PH1          ;90 DEG PULSE CREATES XY-MAGN.
4 D0              ;EVOLUTION OF SHIFTS
5 P1 PH2          ;COMPLETE FIRST COHERENCE TRANSFER, E.G.
                  ; SPIN A TO M DEPENDS ON SIN(PI*J(AM)*D0)
6 D2              ;SECOND COHERENCE PERIOD
7 P2 PH2          ;REFOCUS CHEMICAL SHIFTS
8 D2
9 P1 PH3          ;COMPLETE SECOND TRANSFER (EG. M TO Q)
10 D3
11 P2 PH2
12 D3
13 P1 PH4         ;THIRD TRANSFER FROM Q TO R
14 D4
15 P2 PH2
16 D4
17 P1 PH5         ;4-TH TRANSFER FROM R TO X
14 GO=2 PH6       ;ACQUIRE FID
15 WR #1          ;STORE FID IN .SER FILE
16 IF #1
17 IN=1           ;LOOP FOR NEXT EXPERIMENT
18 EXIT

PH1=A0

PH2=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH3=A0 A0 A0 A0 A2 A2 A2 A2
    A1 A1 A1 A1 A3 A3 A3 A3

PH4=A0 A0 A2 A2 A0 A0 A2 A2
    A1 A1 A3 A3 A1 A1 A3 A3

PH5=A0 A2 A0 A2 A0 A2 A0 A2
    A1 A3 A1 A3 A1 A3 A1 A3

PH6=R0 R0 R0 R0 R0 R0 R0 R0
    R2 R2 R2 R2 R2 R2 R2 R2

;FOR LINEAR SPIN SYSTEM AMQRX, THE TRANSFER FUNCTION IS
; SIN(PI*J(AM)*2D2)SIN(PI*J(MQ)*2D2)*
; SIN(PI*J(MQ)*2D3)SIN(PI*J(QR)*2D3)*
; SIN(PI*J(QR)*2D4)SIN(PI*J(RX)*2D4)

;SET D2 = CA. 0.5/(1.6*J), WHERE J = LARGER OF J(AM), J(MQ)
;SET D3 =      "     "        "                J(MQ), J(QR)
;SET D4 =      "     "        "                J(QR), J(RX)
;NS=32*N
```

```
;P1=90, P2=180
;  OTHERWISE PARAMETERS AS FOR COSY.
```

COSYTQF.AUR

; Cosy-90 With Triple Quantum Filter Based On Double Quantum Filter Scheme

```
; See WOKAUN & ERNST, CHEM. PHYS. LETT. 52, 407(77)
;PIANTINI ET AL., JACS 104, 6800(82)
;SHAKA & FREEMAN, J.MAGN.RES. 51, 169(83)

;  D1-90-D0-D2-90-D3-90-D2-FID

1 ZE
2 D1             ;RELAXATION
3 P1 PH1         ;90 DEG EXCITATION
4 D0             ;EVOLUTION
  D2             ;FIXED DELAY TO ENHANCE MQC
5 P1 PH2         ;GENERATE MULTIPLE QUANTUM COHERENCE
6 D3             ;3 USEC FOR PHASE SWITCHING
7 P1 PH3         ;SELECTION PULSE
  D2
8 GO=2 PH4       ;ACQUIRE
9 WR #1          ;STORE FID
10 IF #1         ;INCREMENT FILE NUMBER
11 IN=1
   EXIT

;PHASES SPECIFIED AS MULTIPLES OF 30=360/12 DEG.

PH1=(12) 0 2 4 6 8 10 0 2 4 6 8 10
         3 5 7 9 11 1 3 5 7 9 11 1
         6 8 10 0 2 4 6 8 10 0 2 4
         9 11 1 3 5 7 9 11 1 3 5 7

PH2=(12) 0 2 4 6 8 10 3 5 7 9 11 1
         3 5 7 9 11 1 6 8 10 0 2 4
         6 8 10 0 2 4 9 11 1 3 5 7
         9 11 1 3 5 7 0 2 4 6 8 10

PH3=(12) 0 0 0 0 0 0 3 3 3 3 3 3
         3 3 3 3 3 3 6 6 6 6 6 6
         6 6 6 6 6 6 9 9 9 9 9 9
         9 9 9 9 9 9 0 0 0 0 0 0

PH4= R0 R2 R0 R2 R0 R2 R2 R0 R2 R0 R2 R0
     R1 R3 R1 R3 R1 R3 R3 R1 R3 R1 R3 R1
     R2 R0 R2 R0 R2 R0 R0 R2 R0 R2 R0 R2
     R3 R1 R3 R1 R3 R1 R1 R3 R1 R3 R1 R3

;PHASE PROGRAMS SELECT FOR TRIPLE QUANTA
;  WITH N-TYPE SELECTION
;NB:   NS=12*N,

;RD=PW=0
;D1=1-5*T1
;P1=90 DEG PULSE
;D0=3E-6
;D3=3E-6
;D2=0.05-0.3 SEC MAY BE USEFUL TO ENHANCE INTENSITY OF MQC
```

COSYWATR.AU

; Homonuclear Shift-Correlated 2-D Nmr (Jeener) With Water Suppression By T2-Relaxation (WATR)

```
;  D1 - 90 - D0 - 90 OR 45 - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
```

```
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; SCALAR COUPLING J.

1 ZE
2 D1              ;RELAXATION
3 P1 PH1          ;90 DEG EXCITATION PULSE
4 D0              ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P1 PH2          ;MIXING PULSE, 90 OR 45 DEG
6 D2              ;ECHO PERIOD
7 P2 PH3          ;180 DEGREE REFOCUSING PULSE
8 D2
9 LO TO 6 TIMES C ;C HAS TO BE ADJUSTED FOR OPTIMAL WATER SUP-
                  ;PRESSION
10 GO=2 PH4       ;ACQUIRE FID
11 WR #1          ;STORE FID
12 IF #1          ;INCREMENT FILE NUMBER
13 IN=1           ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
10 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH4=R0 R0 R2 R2 R1 R1 R3 R3


;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG
;P2 = 180 DEG
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2
;D2= CA 2MSEC
;VC LIST DEFINES NUMBER OF LOOPS THROUGH ECHO PERIOD, HAS TO BE
;OPTIMIZED FOR OPTIMAL WATER ATTENUATION

;N-TYPE PEAK SELECTION
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL
```

COSYX.AUR

; Homonuclear Shift-Correlated 2-D NMR (Jeener) For X-Nuclei With 1H Decoupling And With Delay Period To Emphasize Small Couplings

```
; 1H: BB(S1) -(S2)- - - - - - - - - - - - - - - - - - - -
;  X: D1 -    D3 -  90 - D0 - D2 - 90 OR 45 - D2 - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; SCALAR COUPLING J.

1 ZE
2 D1 CPD S1     ;RELAXATION, LOW-POWER DEC. FOR NOE
```

```
  D3 S2          ;SWITCH TO HIGHER DEC. POWER
3 P1 PH1         ;90 DEG EXCITATION PULSE
4 D0             ;EVOLUTION OF SHIFTS AND COUPLINGS
  D2             ;FIXED DELAY TO EMPHASIZE EFFECT OF SMALL J
5 P2 PH2         ;MIXING PULSE, 90 OR 45 DEG
  D2             ;WAIT FOR EFFECT OF SMALL J
6 GO=2 PH3       ;ACQUIRE FID
  D3 CPD S1      ;TURN DOWN DEC. POWER
7 WR #1          ;STORE FID
8 IF #1          ;INCREMENT FILE NUMBER
9 IN=1           ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
  D3 DO
10 EXIT          ;EXIT WITH DEC. GATED OFF

PH1=A0 A0 A0 A0 A1 A1 A1 A1 A2 A2 A2 A2 A3 A3 A3 A3
PH2=A0 A2 A1 A3 A1 A3 A2 A0 A1 A3 A2 A0 A2 A0 A3 A1
PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1, S1= CA 0.5 WATT FOR NOE
;D3 = 2 MSEC
;S2 = NORMAL POWER FOR GOOD CPD DECOUPLING (P9) TO MINIMIZE
;     DECAY OF TRANSVERSE MAGNETIZATION DURING EVOLUTION!!
;P1 = 90 DEG
;P2 = 90 DEG FOR MAX. SENSITIVITY
;   = 45 DEG FOR MINIMAL DIAGONAL (GOOD FOR TIGHT AB SYSTEMS)
;        AND 'TILTED' CORREL. PEAKS (SIGNS OF COUPLINGS).
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;D2 =< CA. 0.25/J TO MAXIMIZE CORREL. PEAKS FROM SMALL J

;N-TYPE PEAK SELECTION
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL
```

CPDCHECK.AUR

; Test Program To Check Quality Of CPD Decoupling As A Function Of O2 Offset.

```
1 ZE
2 D1 CPD S1 O2   ;RELAXATION WITH DECOUPLING FOR NOE,
                 ;SET OFFSET O2
3 D2 CPD S2      ;INCREASE POWER FOR GOOD DECOUPLING
4 GO=4           ;ACQUIRE FID
  D3 CPD S1      ;TIME TO OBSERVE FID
5 IN=1           ;LOOP TO 1 FOR NEXT O2 VALUE
6 D2 CPD S1
7 EXIT           ;EXIT WITH DEC. ON FOR NOE

;USE SUITABLE TEST SAMPLE SUCH AS C6H6
;DEFINE FL LIST CONTAINING O2 VALUES FOR 1H DECOUPLING,
;  TYPICALLY WITH VALUES FOR 0,1,2,3,4,5,6, AND 0 PPM
;  OFF-RESONANCE FROM 1H SIGNAL (IE. NE=8).
;D1=CA. 1 SEC, D2=.01, D3= CA. 3 SEC
;S1=CA. 0.4 WATT,  S2=POWER NORMALLY NEEDED FOR GOOD DEC.,
;  IE. S2 SHOULD GIVE RF FIELD STRENGTH OF CA. 5 PPM.
;  WHEN USING CPD,
```

```
;NS=1, RD=0, AQ=CA. 1 SEC, PW=CA. 20 DEG FLIP.
```

CPMAS.AUR

; Cross-Polarization With Hartmann-Hahn Spin-Lock For MAS Applications With AR Amplifier.

```
; ************** USE TLO ***************

;  1H: D1 - 90(+/-Y) - P2(X) - CW
;   X:                 P2(X) - FID(+/-)

1 ZE
2 D1 DO                  ;1H RELAXATION, DEC. OFF
  (P4):C8                ;UNBLANK AR
3 (P1 PH1):D:E (P1):C8   ;90 DEG 1H PULSE (POWER S1), BLANK RECEIVER
4 (P2 PH3):D (P2 PH2):T:E:C8     ;SPIN-LOCK 1H WITH PHASE 0
                         ;CROSS-POLARIZE TO X
                         ;VIA HARTMANN-HAHN MATCH.
5 GO=2 PH4 CW            ;ACQUIRE X-NUCLEUS FID WITH CW 1H DECOUPLING
6 D3 DO
7 EXIT                   ;EXIT WITH DEC. GATED OFF

PH1=1 3
PH2=0 0 2 2 1 1 3 3
PH3=0
PH4=R0 R2 R2 R0 R1 R3 R3 R1

;NORMALLY 1H-5H WITH A MAX. OF CA. 16W TO FEED HP 1H ENDSTAGE.
;RD=PW=0
;NS=8*N
;D1 = 1-5*T1 FOR 1H BUT > 20*(AQ+P2) TO AVOID EXCESSIVE HEATING.
;D3 = 3 MSEC TO ENSURE DEC. IS OFF BEFORE EXITING.
;P1 = 90 DEG 1H PULSE = 90 DEG X PULSE AT HARTMANN-HAHN CONDITION
;P2 = SPIN-LOCK TIME (E.G. 0.5-5 MSEC)
;C8 IS A  DC PULSE ON PIN V OF BURNDY P2 ON PROCESS-CONTROLLER
;USED TO UNBLANK THE AR TRANSMITTER: SET P4=40US
```

CPMASGS.AUR

; Set-Up Mode: Cross-Polarization With Hartmann-Hahn Spin-Lock For MAS Applications With AR Amplifier.

```
; ************** USE TLO ***************

;  1H: D1 - 90(+/-Y) - P2(X) - CW
;   X:                 P2(X) - FID(+/-)

1 ZE
2 D1 DO                  ;1H RELAXATION, DEC. OFF
  (P4):C8                ;UNBLANK AR : P4=40US
3 (P1 PH1):D:E (P1):C8   ;90 DEG 1H PULSE (POWER S1), BLANK RECEIVER
4 (P2 PH3):D (P2 PH2):T:E:C8     ;SPIN-LOCK 1H WITH PHASE 0
                         ;CROSS-POLARIZE TO X
                         ;VIA HARTMANN-HAHN MATCH.
5 GS=2 PH4 CW            ;ACQUIRE X-NUCLEUS FID WITH CW 1H DECOUPLING
6 D3 DO
7 EXIT                   ;EXIT WITH DEC. GATED OFF

PH1=1 3
PH2=0 0 2 2 1 1 3 3
PH3=0
PH4=R0 R2 R2 R0 R1 R3 R3 R1

;NORMALLY 1H-5H WITH A MAX. OF CA. 16W TO FEED HP 1H ENDSTAGE.
;RD=PW=0
;NS=8*N
```

```
;D1 = 1-5*T1 FOR 1H BUT > 20*(AQ+P2) TO AVOID EXCESSIVE
;HEATING.
;D3 = 3 MSEC TO ENSURE DEC. IS OFF BEFORE EXITING.
;P1 = 90 DEG 1H PULSE = 90 DEG X PULSE AT HARTMANN-HAHN CONDITION
;P2 = SPIN-LOCK TIME (E.G. 0.5-5 MSEC)
```

CPMASVC.AUR

; Cross-Polarization With Variable Contact Time For MAS Applications With AR Amplifier.

```
; ************** USE TLO ***************

;  1H: D1 - 90(+/-Y) - P2(X) - CW
;   X:                 P2(X) - FID(+/-)

1 ZE
2 D1 DO                  ;1H RELAXATION, DEC. OFF
  (P4):C8                ;UNBLANK AR
3 (P1 PH1):D:E (P1):C8   ;90 DEG 1H PULSE (POWER S1), BLANK RECEIVER
4 (P2 PH3):D (P2 PH2):T:E:C8     ;SPIN-LOCK 1H WITH PHASE 0
  L1 TO 4 TIMES UPR      ;CROSS-POLARIZE TO X
                         ;VIA HARTMANN-HAHN MATCH.
5 GO=2 PH4 CW            ;ACQUIRE X-NUCLEUS FID WITH CW 1H DECOUPLING
6 D3 DO                  ;DEC. GATED OFF
7 WR #1                  ;WRITE DATA
8 IF #1
9 IU1                    ;INCREMENT LOOP COUNTER
10 IN =1
11 EXIT

PH1=1 3
PH2=0 0 2 2 1 1 3 3
PH3=0
PH4=R0 R2 R2 R0 R1 R3 R3 R1

;NORMALLY 1H-5H WITH A MAX. OF CA. 16W TO FEED HP 1H ENDSTAGE.
;RD=PW=0
;NS=8*N
;D1 = 1-5*T1 FOR 1H BUT > 20*(AQ+P2) TO AVOID EXCESSIVE
;HEATING.
;D3 = 3 MSEC TO ENSURE DEC. IS OFF BEFORE EXITING.
;P1 = 90 DEG 1H PULSE = 90 DEG X PULSE AT HARTMANN-HAHN CONDITION
;P2 = SPIN-LOCK TIME (E.G. 0.5-5 MSEC)
;SET NE SO AS NOT TO EXCEED 15MS CONTACT PULSE, I.E. NE*P2 < 15MS
```

CPMG.AUR

; Carr-Purcell-Meiboom-Gill Spin-Echo For T2 Eliminates J-Modulation, 180 Deg Pulse Errors, And Diffusion Effects. Sample Spinning Can Modulate The Echo Decay Rate.

```
;   D1 - 90 - (D2 - 180 - D2)*C - FID

1 ZE
2 D1            ;RELAX TO EQUILIBRIUM
3 P1 PH1        ;90 DEG PULSE
4 D2            ;DEPHASE
5 P2 PH2        ;180 DEG PULSE SHIFTED 90 DEG REL. TO P1
6 D2            ;REPHASE
7 LO TO 4 TIMES C       ;VARIABLE LOOP TO 4 FOR 'C' ECHOES
8 GO=2 PH3      ;ACQUIRE FID, LOOP TO 2
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE EXTENSION
11 VC           ;INCREMENT POINTER FOR VC LIST (NEXT VALUE OF 'C')
12 IN=1         ;LOOP FOR NEXT EXPERIMENT
13 EXIT

PH1=A0 A0 A2 A2 A1 A1 A3 A3
PH2=A1 A3
PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES THE NUMBER OF EXPERIMENTS, NO. OF ITEMS IN VC LIST.
;CURRENT VC LIST MUST CONTAIN A SERIES OF VALUES FOR THE LOOP
;  COUNTER 'C'. THESE MUST BE EVEN NUMBERS TO PROVIDE FOR
;  CANCELLATION OF 180 DEG PULSE ERRORS.
;D2 = A SHORT FIXED ECHO TIME TO ALLOW ELIMINATION OF DIFFUSION
;  AND J-MOD. EFFECTS. THE MIN. VALUE FOR D2 IS 0.6 MSEC TO ALLOW
;  FOR VARIABLE LOOP IN STEP 7. SHORTER VALUES OF D2 ARE
;  POSSIBLE WHEN A FIXED HARDWARE LOOP IS USED.
;  D2 SHOULD BE << 1/J  BUT > CA. 50*P2 TO
;  AVOID EXCESSIVE AVE. POWER OUTPUT FOR TRANSMITTER.

;D1 = 5*T1 FOR COMPLETE RELAXATION
;P1 = 90 DEG
;P2 = 180 DEG
;RD=PW=0, NS= MULTIPLE OF 8
```

CPSET.AUR

```
; Set-Up Cross-Polarization With Hartmann-Hahn Spin-Lock
; For MAS Applications.

;  1H: D1 - 90(+/-Y) - P2(X) - CW
;   X:                 P2(X) - FID(+/-)

1 ZE
2 D1 DO S1          ;1H RELAXATION, DEC. OFF
3 (P1 PH1):D:E      ;90 DEG 1H PULSE (POWER S1), BLANK RECEIVER
4 (P2 PH2 CW):T:E        ;SPIN-LOCK 1H WITH PHASE 0
                         ; CROSS-POLARIZE TO X
                         ; VIA HARTMANN-HAHN MATCH.
5 GS=2 PH3 CW       ;ACQUIRE X-NUCLEUS FID WITH CW 1H DECOUPLING
6 D3 DO
7 EXIT          ;EXIT WITH DEC. GATED OFF

PH1=B1 B1 B3 B3
PH2=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH3=R0 R0 R2 R2 R1 R1 R3 R3
    R2 R2 R0 R0 R3 R3 R1 R1

;S1 DEFINES DEC. POWER FOR ENTIRE SEQUENCE
; NORMALLY 1H-5H WITH A MAX. OF CA. 16W TO FEED HP 1H ENDSTAGE.
;RD=PW=0
;NS=8*N
;D1 = 1-5*T1 FOR 1H BUT > 20*(AQ+P2) TO AVOID EXCESSIVE
;  HEATING.
;D3 = 1 MSEC TO ENSURE DEC. IS OFF BEFORE EXITING.
;P1 = 90 DEG 1H PULSE = 90 DEG X PULSE AT HARTMANN-HAHN CONDITION
;P2 = SPIN-LOCK TIME (E.G. 0.5-5 MSEC)
```

CPT1.TEST

```
;                      BE/HF 6/89
;*****UNTESTED******
; Cross-Polarization With Hartmann-Hahn Spin-Lock For MAS Applications With AR
  Amplifier.
; ************** USE TLO ***************

;  1H: D1 - 90(+/-Y) - P2(X) - CW
;   X:                 P2(X) - FID(+/-)

1 ZE
2 D1 DO                   ;1H RELAXATION, DEC. OFF
  (P4):C8                 ;UNBLANK AR
3 (P1 PH1):D:E (P1):C8   ;90 DEG 1H PULSE (POWER S1), BLANK RECEIVER
4 (P2 PH3):D (P2 PH2):T:E:C8     ;SPIN-LOCK 1H WITH PHASE 0
                          ;CROSS-POLARIZE TO X
                          ;VIA HARTMANN-HAHN MATCH.
  (P3 PH5):T:E:C8         ;90 DEG X-PULSE, STORE IN Z-AXIS
  VD
  (P4):C8                 ;UNBLANK AR
  (P3 PH6):T:E:C8         ;90 DEG X-PULSE, FLIP BACK TO XY-PLANE
5 GO=2 PH4 CW             ;ACQUIRE X-NUCLEUS FID WITH CW 1H DECOUPLING
  D3 DO
  WR #1
  IF #1
  IN=1
7 EXIT                    ;EXIT WITH DEC. GATED OFF

PH1=1 3
PH2=0
PH3=0
PH4=R0 R2 R2 R0 R1 R3 R3 R1
PH5=1
PH6=0 0 2 2 1 1 3 3

;NORMALLY 1H-5H WITH A MAX. OF CA. 16W TO FEED HP 1H ENDSTAGE.
;RD=PW=0
;NS=8*N
;D1 = 1-5*T1 FOR 1H BUT > 20*(AQ+P2) TO AVOID EXCESSIVE
;HEATING.
;D3 = 3 MSEC TO ENSURE DEC. IS OFF BEFORE EXITING.
;P1 = 90 DEG 1H PULSE = 90 DEG X PULSE AT HARTMANN-HAHN CONDITION
;P2 = SPIN-LOCK TIME (E.G. 0.5-5 MSEC)
;C8 IS A  DC PULSE ON PIN V OF BURNDY P2 ON PROCESS-CONTROLLER
;USED TO UNBLANK THE AR TRANSMITTER: SET P4=40US
;P4 ONLY 1US IF AR-AMPLIFIER IS NOT USED
;P3 = 90 DEG X-PULSE
;SET VD-LIST ACCORDING TO X-NUCLEUS T1
;USE T1 ROUTINE FOR EVALUATION
```

CPT1RHO.TEST

```
;*****UNTESTED******                BE/HF 6/89
; Cross-Polarization With Hartmann-Hahn Spin-Lock For MAS Applications With AR
  Amplifier.
; ************** USE TLO ***************

;  1H: D1 - 90(+/-Y) - P2(X) - CW
;   X:                 P2(X) - FID(+/-)

1 ZE
2 D1 DO                  ;1H RELAXATION, DEC. OFF
  (P4):C8                ;UNBLANK AR
3 (P1 PH1):D:E (P1):C8   ;90 DEG 1H PULSE (POWER S1), BLANK RECEIVER
4 (P2 PH3):D (P2 PH2):T:E:C8     ;SPIN-LOCK 1H WITH PHASE 0
                         ;CROSS-POLARIZE TO X
                         ;VIA HARTMANN-HAHN MATCH.
8 (P3 PH2):T:E:C8        ;SPINLOCK FOR T1RHO
  L1 TO 8 TIMES UPR      ;L1 >=1
5 GO=2 PH4 CW            ;ACQUIRE X-NUCLEUS FID WITH CW 1H DECOUPLING
  D3 DO
  WR #1
  IF #1
  IU1
  IN=1
7 EXIT                   ;EXIT WITH DEC. GATED OFF

PH1=1 3
PH2=0 0 2 2 1 1 3 3
PH3=0
PH4=R0 R2 R2 R0 R1 R3 R3 R1

;NORMALLY 1H-5H WITH A MAX. OF CA. 16W TO FEED HP 1H ENDSTAGE.
;RD=PW=0
;NS=8*N
;D1 = 1-5*T1 FOR 1H BUT > 20*(AQ+P2) TO AVOID EXCESSIVE
;HEATING.
;D3 = 3 MSEC TO ENSURE DEC. IS OFF BEFORE EXITING.
;P1 = 90 DEG 1H PULSE = 90 DEG X PULSE AT HARTMANN-HAHN CONDITION
;P2 = SPIN-LOCK TIME (E.G. 0.5-5 MSEC)
;C8 IS A  DC PULSE ON PIN V OF BURNDY P2 ON PROCESS-CONTROLLER
;USED TO UNBLANK THE AR TRANSMITTER: SET P4=40US
;L1 >= 1
;P3 = SPINLOCK FOR T1RHO
;SET P3, L1, NE TO ACHIEVE DESIRED SPINLOCK TIME VARIATION
```

DEC90.AUR

; Calibrate The 1H Decoupler 90 Deg Pulse Length By Observing A X-Nucleus (e.g. Carbon) Which Has One Attached Proton With Coupling J(XH).

```
1 ZE
2 D1 DO         ;RELAX, DEC. GATED OFF
3 P3            ;90 DEG X-NUCLEUS PULSE
4 D2            ;EVOLUTION
5 P1:D          ;VARIABLE 1H DEC. PULSE
6 GO=2          ;ACQUIRE X-NUCLEUS FID WITHOUT DEC.
7 EXIT

;USE SUITABLE TEST SAMPLE OR SELECT SIGNAL WHICH HAS
;  DOUBLET MULTIPLICITY (XH).
;D1=CA. T1
;D2=0.5/J(XH)
;P3=90 DEG X PULSE
;P1=1H PULSE WHICH WILL BE VARIED
;PW=RD=0, WHEN POSSIBLE USE NS=1, DS=0

;FOR SMALL P1, X-NUCLEUS SPECTRUM IS AN ANTI-PHASE DOUBLET.
;FOR P1=CA. 180 DEG, ANTI-PHASE DOUBLET IS INVERTED.
;FOR P1=90 DEG, BOTH COMPONENTS OF DOUBLET GO THROUGH A MIN.

;THE COMPLETE DEPT SEQUENCE MAY ALSO BE USED FOR CALIBRATION
; PURPOSES, WHERE A 90 DEG PULSE GIVES NULLING OF XH2 TRIPLET
; MULTIPLICITIES.
```

DECPHASE.AUR

```
; Check Or Calibrate Decoupler Phase Shifts Using Decoupler As Transmitter For
  1H Spectra.
; USE AUTOMATIC BLOCK ADDRESS ADVANCE (ASTI=1)

1 ST0           ;SET START ADDRESS TO BEGINNING OF MEMORY REGION
2 ZE            ;ZERO BLOCK
3 ST
4 LO TO 2 TIMES C       ;LOOP TO ZERO ALL BLOCKS
5 ST0           ;RESET TO FIRST BLOCK
6 P1:D PH1      ;DECOUPLER PULSE
7 GO=6          ;ACQUIRE FID (DETECTOR PHASE=0), LOOP AND
                ;AUTOMATICALLY INCREMENT BLOCK ST.
8 ST0              ;RESET TO FIRST BLOCK
9 WR #1            ;STORE FID
10 IF #1
11 ST
12 LO TO 9 TIMES C  ;WRITE ALL BLOCKS TO DISK
  EXIT

PH1=(8) 0 1 2 3 4 5 6 7        ;DEFINES 45 DEG PHASE SHIFTS

;NB: USE CP MODE AND SET ASTI=1 TO AUTOMATICALLY
;  INCREMENT BLOCK START ADDRESS AFTER EACH SCAN.
;RD=PW=0
;NS=NUMBER OF FIDS, EG. 5
;NBL = NUMBER OF BLOCKS = NS
;VC LIST HAS ONE ITEM = NS
;FOR EXAMPLE,  SET NS=NBL=5
; FIRST AND LAST SPECTRA HAVE PHASE 0.

;O1 AND O2 MUST BE AS NEARLY EQUAL AS POSSIBLE TO AVOID
; ADDITIONAL PHASE SHIFTS DURING THE TIME OF THE
; EXPERIMENT.

;PHASE CORRECT FIRST SPECTRUM AND USE 'PK' FOR ALL OTHER
;  SPECTRA.
```

DEPT.AUR

; DEPT Polarization Transfer From 1H To X-Nuclei For Refocussed Decoupled Spectra

```
;  1H: D1 - 90 - D2 - 180 - D2 - P0 - D2 - CPD
;   X:                 90        180       FID

1 ZE
2 D1 S1  DO                     ;1H RELAXATION, SET DEC. POWER
                                ;FOR PULSING
3 (P1 PH1 D2):D                 ;90 DEG 1H PULSE, SHIFTS AND
                                ;J(XH) EVOLVE
4 (P2 PH2):D  (P3 PH4 D2)       ;180 DEG 1H PULSE TO REFOCUS
                                ;SHIFTS, 90 DEG X PULSE FOR MQ
                                ;COHERENCE
5 (P0 PH3):D (P4 PH5 D2 S2)     ;VARIABLE PULSE FOR 1H TO COMPLETE
                                ;POLARIZATION TRANSFER, 180 X PULSE
                                ;TO REFOCUS X SHIFTS, SET DEC. POWER
6 GO=2 PH7 CPD           ;ACQUIRE FID WITH DEC.
7 D2 DO
8 EXIT                   ;EXIT WITH DEC. OFF

PH1=0                    ;DECOUPLER PHASES

PH2=0 2 1 3
PH3=1 1 1 1 3 3 3 3

PH4= 0 0 0 0 0 0 0 0         ;TRANSMITTER PHASES
     1 1 1 1 1 1 1 1
     2 2 2 2 2 2 2 2
     3 3 3 3 3 3 3 3
PH5= 0 2 0 2 0 2 0 2
     1 3 1 3 1 3 1 3
PH7= R0 R0 R2 R2 R2 R2 R0 R0
     R1 R1 R3 R3 R3 R3 R1 R1
     R2 R2 R0 R0 R0 R0 R2 R2
     R3 R3 R1 R1 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.

; NS=4*N (MIN. 4 SCANS, 32 FOR COMPLETE PHASE CYCLE)
;RD=PW=0
```

DEPTC.AUR

; DEPT Polarization Transfer From 1H To X-Nuclei For Refocussed 1H-Coupled Spectra

```
;  1H: D1 - 90 - D2 - 180 - D2 - P0 - D2 - DO
;   X:                 90        180       FID

1 ZE
2 D1 S1 DO                      ;1H RELAXATION, SET DEC. POWER
                                ;FOR PULSING
3 (P1 PH1 D2):D                 ;90 DEG 1H PULSE, SHIFTS AND
                                ;J(XH) EVOLVE
4 (P2 PH2):D  (P3 PH4 D2)       ;180 DEG 1H PULSE TO REFOCUS
                                ;SHIFTS, 90 DEG X PULSE FOR MQ
                                ;COHERENCE
5 (P0 PH3):D  (P4 PH5 D2)       ;VARIABLE PULSE FOR 1H TO COMPLETE
                                ;POLARIZATION TRANSFER, 180 X PULSE
                                ;TO REFOCUS X SHIFTS
6 GO=2 PH7              ;ACQUIRE FID
7 D2 DO
8 EXIT                  ;EXIT WITH DEC. OFF

PH1=B0                          ;DECOUPLER PHASES
PH2=B0 B2 B1 B3
PH3=B1 B1 B1 B1 B3 B3 B3 B3
PH4=A0 A0 A0 A0 A0 A0 A0 A0     ;TRANSMITTER PHASES
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH5=A0 A2 A0 A2 A0 A2 A0 A2
    A1 A3 A1 A3 A1 A3 A1 A3
PH7=R0 R0 R2 R2 R2 R2 R0 R0     ;RECEIVER PHASE
    R1 R1 R3 R3 R3 R3 R1 R1
    R2 R2 R0 R0 R0 R0 R2 R2
    R3 R3 R1 R1 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.

; NS=4*N (MIN. 4 SCANS, 32 IS COMPLETE PHASE CYCLE)
;RD=PW=0
```

DEPTPP.AUR

```
; DEPT++ Polarization Transfer From 1H To X-Nuclei For Refocussed Coupled
  Spectra With Elimination Of Multiplet Anomalies.
; O.W.SORENSEN & R.R.ERNST, J.MAGN.RES. 51, 477 (83)

;  1H: D1-90-D2-180-D2-P0-D3-180-D3-90-D3-180-D3- DO
;   X:         90          180                 - FID

1 ZE
2 D1 S1 DO                        ;1H RELAXATION, SET DEC. POWER
                                  ;FOR PULSING
3 (P1 PH1 D2):D                   ;90 DEG 1H PULSE, SHIFTS AND
                                  ;J(XH) EVOLVE
4 (P2 PH2):D  (P3 PH4 D2)         ;180 DEG 1H PULSE TO REFOCUS
                                  ;SHIFTS, 90 DEG X PULSE FOR MQ
                                  ;COHERENCE
5 (P0 PH3 D3):D            ;VARIABLE PULSE FOR 1H TO COMPLETE
                           ;POLARIZATION TRANSFER, BEGIN PURGE
  (P2 PH9 D3):D  P4 PH5    ;TO REFOCUS 1H AND X SHIFTS
  (P1 PH8 D3):D            ;90 DEG 1H PULSE (PURGING)
  (P2 PH9 D3):D            ;REFOCUS 1H SHIFTS AGAIN
6 GO=2 PH7                 ;ACQUIRE FID WITHOUT DEC.
7 D2 DO
8 EXIT              ;EXIT WITH DEC. OFF

PH1=B0
PH2=B0 B2 B1 B3
PH3=B1 B1 B1 B1 B3 B3 B3 B3
PH4=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH5=A0 A2 A0 A2 A0 A2 A0 A2
    A1 A3 A1 A3 A1 A3 A1 A3
PH7=R0 R0 R2 R2 R2 R2 R0 R0
    R1 R1 R3 R3 R3 R3 R1 R1
    R2 R2 R0 R0 R0 R0 R2 R2
    R3 R3 R1 R1 R1 R1 R3 R3
PH8=B0
PH9=B0 B2
;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;D3 = D2/2
;S1 = 0H FOR MAX. POWER PULSES
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.

; NS=4*N (MIN. 4 SCANS, 32 IS COMPLETE CYCLE)
;RD=PW=0
```

DEPTSAT.AUR

; DEPT Polarization Transfer From 1H To X-Nuclei For Refocussed Decoupled Spectra With 90 Deg Pre-Saturation Pulse For X. Useful For Long Relaxing X With Neg. NOE.

```
;  1H: D1 - 90 - D2 - 180 - D2 - P0 - D2 - BB
;   X:      90         90        180       FID

1 ZE
2 D1 S1 DO                       ;1H RELAXATION, SET DEC. POWER
                                 ;FOR PULSING
3 (P1 PH1 D2):D  P3              ;90 DEG 1H PULSE, SHIFTS AND
                                 ;J(XH) EVOLVE, 90 DEG X PULSE
                                 ;TO DESTROY Z-MAGN.
4 (P2 PH2):D  (P3 PH4 D2)        ;180 DEG 1H PULSE TO REFOCUS
                                 ;SHIFTS, 90 DEG X PULSE FOR MQ
                                 ;COHERENCE
5 (P0 PH3):D  (P4 PH5 D2 S2)     ;VARIABLE PULSE FOR 1H TO COMPLETE
                                 ;POLARIZATION TRANSFER, 180 X PULSE
                                 ;TO REFOCUS X SHIFTS, SET DEC. POWER
6 GO=2 PH7 CPD           ;ACQUIRE FID WITH DEC.
7 D2 DO
8 EXIT              ;EXIT WITH DEC. OFF

PH1=B0
PH2=B0 B2 B1 B3
PH3=B1 B1 B1 B1 B3 B3 B3 B3
PH4=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH5=A0 A2 A0 A2 A0 A2 A0 A2
    A1 A3 A1 A3 A1 A3 A1 A3
PH7=R0 R0 R2 R2 R2 R2 R0 R0
    R1 R1 R3 R3 R3 R3 R1 R1
    R2 R2 R0 R0 R0 R0 R2 R2
    R3 R3 R1 R1 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR OPTIMAL DEC. WITH CPD (P9)
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;           = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;           = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.

;NS=4*N (MIN. 4 SCANS, 32 IS COMPLETE CYCLE)
;RD=PW=0
```

DEPTVAR.AUR

; DEPT With Variable Pulse P0 For A Series Of Decoupled Spectra, Using Job Parameter Files And Cycling Of 3 Experiments For Long-Term Averaging.

```
1 RJ #1         /ENTER JOB PARAMETER FILE
                ;READ JOB PARAMETER FILE
  IF #1         ;INCREMENT JOB PARAMETER FILE EXT.
  ZE            ;ZERO MEMORY
  WR #2         /ENTER FID
                ;STORE ZEROED FILE AND PARAMETERS
  IF #2         ;INCREMENT FID FILE EXT.
  LO TO 1 TIMES 3       ;REPEAT FOR 3 EXPERIMENTS (3 VALUES OF P0)

2 RF #2.001     ;RESET FID FILE EXT., BEGIN CYCLE
3 RE #2         ;READ CURRENT FID WITH PARAMETERS
4 D1 S1 DO      ;PERFORM DEPT WITH DECOUPLING
  (P1 PH1 D2):D
  (P2 PH2):D  (P3 PH4 D2)
  (P0 PH3):D  (P4 PH5 D2 S2)
6 GO=4 PH7 CPD
7 D2 DO         ;GATE DEC. OFF
  WR #2         ;STORE CURRENT FID
  IF #2         ;INCREMENT FID FILE EXT.
8 LO TO 3 TIMES 3       ;LOOP TO 3 FOR 3 EXPERIMENTS
9 IN=2          ;LOOP TO 2 TO REPEAT CYCLE NE TIMES
  EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=B1 B1 B1 B1 B3 B3 B3 B3
PH4=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH5=A0 A2 A0 A2 A0 A2 A0 A2
    A1 A3 A1 A3 A1 A3 A1 A3
PH7=R0 R0 R2 R2 R2 R2 R0 R0
    R1 R1 R3 R3 R3 R3 R1 R1
    R2 R2 R0 R0 R0 R0 R2 R2
    R3 R3 R1 R1 R1 R1 R3 R3

;REQUESTS FILENAME #1 = JOB PARAMETER FILE (CREATED WITH 'WJ').
;  THERE MUST BE 3 FILES WITH P0 = 45, 90, 135 DEG RESPECTIVELY.
;  FOR P0 = 90, NS SHOULD BE TWICE THE VALUE USED FOR THE OTHER 2
;  EXPERIMENTS. NE MUST BE THE NUMBER OF DESIRED CYCLES TO GIVE
;  NE*NS TOTAL TRANSIENTS.
;FILENAME #2 = FID DATA FILE.

;ALL PARAMETERS AS FOR NORMAL DEPT.
```

DQ2D.AUR

```
; Double Quantum 2-D For Protons
;T.H.MARECI & R.FREEMAN, J.MAGN.RES. 51, 531 (83)

; D1-90-D2-180-D2- 90  -D0-ALPHA-AQ(FID)  +
; D1-90-D2-180-D2-45-90-D0-ALPHA-AQ(FID)

; MATRIX REPRESENTATION LIKE INADEQUATE WITH DOUBLE QUANTUM
; FREQ. F(A)+F(B) IN F1 DOMAIN, DIAGONAL PEAKS SUPPRESSED.

1 ZE
2 D1            ;RELAX
3 P1 PH1        ;90 DEG
4 D2            ;(2N+1)/4J
5 P2 PH2        ;180 DEG
6 D2            ;REFOCUS SHIFTS
7 P1 PH1        ;CREATE DBL.QUANTUM COHERENCE
8 D0            ;EVOLUTION
9 P3 PH3        ;CONVERT TO SINGLE QUANTUM FOR DETECTION
                ; P3=135-160 DEG AVOIDS PASSIVE SPINS
10 GO=2 PH4     ;ACQUIRE FID USING PHASE PROGRAM, NS TRANSIENTS

11 D1           ;REPEAT SEQUENCE WITH ADDITIONAL 45 DEG PULSE
12 P1 PH1       ; FOR 90 DEG PHASE SHIFT (F1 QUAD DETECTION)
13 D2
14 P2 PH2
15 D2
16 P5 PH5       ;45 DEG PULSE
17 D3
18 P1 PH1
19 D0
20 P3 PH3
21 GO=11 PH6    ;NS TRANSIENTS
22 LO TO 2 TIMES C      ;LOOP TO REPEAT SEQUENCE 'C' TIMES
23 WR #1
24 IF #1
25 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
   EXIT

PH1=A0 A0 A0 A0 A0 A0 A0 A0
    A2 A2 A2 A2 A2 A2 A2 A2
    A1 A1 A1 A1 A1 A1 A1 A1
    A3 A3 A3 A3 A3 A3 A3 A3

PH2=A0 A2 A0 A2 A0 A2 A0 A2
    A0 A2 A0 A2 A0 A2 A0 A2
    A1 A3 A1 A3 A1 A3 A1 A3
    A1 A3 A1 A3 A1 A3 A1 A3

PH3=A0 A0 A1 A1 A2 A2 A3 A3
    A2 A2 A3 A3 A0 A0 A1 A1
    A1 A1 A2 A2 A3 A3 A0 A0
    A3 A3 A0 A0 A1 A1 A2 A2
PH4=R0 R0 R3 R3 R2 R2 R1 R1
    R2 R2 R1 R1 R0 R0 R3 R3
    R1 R1 R0 R0 R3 R3 R2 R2
    R3 R3 R2 R2 R1 R1 R0 R0
PH5=A3 A3 A3 A3 A3 A3 A3 A3
    A1 A1 A1 A1 A1 A1 A1 A1
    A0 A0 A0 A0 A0 A0 A0 A0
    A2 A2 A2 A2 A2 A2 A2 A2
PH6=R1 R1 R0 R0 R3 R3 R2 R2
    R3 R3 R2 R2 R1 R1 R0 R0
    R2 R2 R1 R1 R0 R0 R3 R3
    R0 R0 R3 R3 R2 R2 R1 R1
```

```
;RD=PW=0
;P1=90 DEG    P2=180 DEG   P3=  135-160 DEG   P5=45 DEG
;D1=1-5*T1
;D2=(2N+1)/4J ,   D0=D3=3E-6
;NS=8*N,  TOT. TRANS.=2*C*NS , NB: MIN. NS=8
;DS=2 OR 4
;ND0=1, IN=0.5/SW1,  CHOOSE SW1 AT MOST =SW
```

DQF.AUR

```
; Double Quantum Filter For 1-D NMR - This Is A Spin-Echo Version Of The
  Multiple Quantum Filter.  This Eliminates Single Quantum Transitions (Such
  As Solvent).
; With The Phase Cycling Given.
; PIANTINI ET AL., JACS 104, 6800-6801 (1982)
; HOMONUC. J-MODULATION IS PRESENT SO THAT TYPICALLY ONE
; SHOULD USE RESOLUTION ENHANCEMENT (E.G. SINE-BELL) AND
; MAGNITUDE CALCULATION (MC).

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG
4 D2            ;CA. 1/(4J), EG. 35 MSEC FOR 1H
5 P2 PH2        ;180 DEG
6 D2
7 P1 PH1        ;90 DEG, CREATE MQ
8 D3            ;MQ PRECESSION, NORMALLY < 5 USEC
9 P1 PH3        ;90 DEG CONVERSION PULSE
10 GO=2 PH4        ;ACQUIRE
   EXIT

PH1=A0 A2 A1 A3 A1 A3 A2 A0     ;PHASE CYCLE FOR DBL.QUANTUM
PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A0 A2 A1 A3 A1 A3 A2 A0
    A2 A0 A3 A1 A3 A1 A0 A2
    A2 A0 A3 A1 A3 A1 A0 A2
PH3=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH4=R0 R0 R2 R2 R1 R1 R3 R3
    R2 R2 R0 R0 R3 R3 R1 R1

;   NS=4*N
;D1 = 1-5*T1
;P1,P2 = 90,180 DEG
;D2 = CA. 1/(4J)
;D3 = 3E-6
```

ECOSY.AUR

; Generalized Microprogram

```
1 ZE
  RCV VCLIST          ;read VCLIST with calculated NS and DC values
  WR #1               ;first .SER file
  WR #2               ;second .SER file
  .
  .
  WR #M               ;Mth .SER file,
                      ;M - number of spin systems wanted
  ECOS                ;set NS and set DC for first .SER file
2 ZE
3 D1
  P1 PH1              ;Pulse sequence: ECOSY
  D0
  P1 PH2
  P1 PH3
  GO=3 PH4
  WR IM               ;write intermediate result
  AT #1               ;add with DC to first .SER file
  WR #1               ;write result
  RE IM               ;read in intermediate result
  ECOS                ;set DC for second .SER file
  AT #2               ;add second .SER file
  WR #2
  .
  .
  RE IM               ;read in intermediate result
  ECOS                ;set DC for Mth .SER file
  AT #M               ;add Mth .SER file
  WR #M
  IP1
  IP2                 ;next phase BETA
  LO TO 2 TIMES 2*N   ;loop for 2*N phases,
                      ;N - maximum size of spin system
  IF #1               ;increment extension
  IF #2
  .
  .
  IF #M
4 IP1
  LO TO 4 TIMES N/2   ;TPPI
  IN=2
5 EXIT

PH1=(2*N) 0
PH2=(2*N) 0
PH3=      2
PH4=      R0
```

ECOSY3B.AUR

; E. COSY Microprogram for KcMAX=3 and Normal and Complementary E. COSY

```
;You may create the relevant NS and DC values with the PASCAL ECOSY
;program,
;however there are files VC3B6.001     for NS=6
;                        VC3B14.001    for NS=14
;                        VC3B26.001    for NS=26
;                   and  VC3B52.001    for NS=52
;which contain already all you need for 6, 14, 26, or 52 scans per
;t1 experiment.

;*****************************************************************
;In the case that you use one of these lists line 2 of the micro-
;program must read for example for 26 scans: RVC VC3B26
;*****************************************************************

1 ZE
  RVC VCLIST          ;read VCLIST with NS and DC values created by
                      ;the ECOSY Pascal program
  WR #1               ;first .SER file: normal E.COSY with KcMAX=3
  WR #2               ;second .SER file: complementary E.COSY with
                      ;KcMAX=3
2 ECOS                ;set NS and DC for next phase setting
  ZE
3 D1
  P1 PH1              ;Pulse sequence: ECOSY
  D0
  P1 PH2
  P1 PH3
  GO=3 PH4
  WR IM               ;write intermediate result
  AT #1               ;add with DC to normal E.COSY
  WR #1               ;write result
  ECOS                ;read DC constant for complementary E.COSY
  RE IM               ;read intermediate result
  AT #2               ;add with second DC to complementary E.COSY
  WR #2               ;write result
  IP1                 ;next phase BETA
  IP1
  IP2
  LO TO 2 TIMES 6     ;loop for 2*N=6 BETA-phases,
  IF #1               ;increment extension of .SER files
  IF #2
4 IP1
  LO TO 4 TIMES 3     ;TPPI
  IN=2                ;next t1 value
5 EXIT

PH1=(12) 3
PH2=(6) 0
PH3= 2
PH4= R1
```

ECOSY3C.AU

; Complementary E.COSY Program for KcMAX=3

```
1 ZE
2 D1
P1 PH1
D0
P1 PH2
D2              ; D2=4U
P1 PH3
GO=2 PH4
WR #1
IF #1
IP1
IP1
IP1
IN=1
EXIT
PH1=(12) 3 3 3 3 5 5 5 7 11 1 1 1
PH2=(12) 0 0 0 0 2 2 2 4 8 10 10 10
PH3= 0
PH4=R1 R1 R1 R1 R3 R3 R3 R1 R1 R3 R3 R3
```

ECOSY3N.AU

;E.COSY Program for KcMAX=3

```
1 ZE
2 D1
P1 PH1
D0
P1 PH2
D2              ; D2=4U
P1 PH3
GO=2 PH4
WR #1
IF #1
IP1
IP1
IP1
IN=1
EXIT
PH1=(12) 3 3 3 3 5 5 5 7 11 1 1 1
PH2=(12) 0 0 0 0 2 2 2 4 8 10 10 10
PH3= 2
PH4=R1 R1 R1 R1 R3 R3 R3 R1 R1 R3 R3 R3
```

ECOSY4B.AU

; E. COSY Microprogram for KcMAX=4 and Normal and Complementary E. COSY

```
;You may create the relevant NS and DC values with the PASCAL ECOSY
;program,
;however, there are files VC4B16.001 for NS=16
;                         VC4B24.001 for NS=24
;                         VC4B54.001 for NS=54
;which contain already all that you need for 16, 24, or 54 scans
;per t1 increment.

;**********************************************************************
;In the case that you use one these list files line 2 of the micro-
;program changes for example for NS=16 to:  RVC VC4B24
;**********************************************************************


1 ZE
  RVC VCLIST          ;read VCLIST with NS and DC values created by
                      ;the ECOSY Pascal program
  WR #1               ;first .SER file: normal E.COSY with KcMAX=4
  WR #2               ;second .SER file: complementary E.COSY with
                      ;KcMAX=4
2 ECOS                ;set NS and DC for next phase setting
  ZE
3 D1
  P1 PH1              ;Pulse sequence: ECOSY
  D0
  P1 PH2
  P1 PH3
  GO=3 PH4
  WR IM               ;write intermediate result
  AT #1               ;add with DC to normal E.COSY
  WR #1               ;write result
  ECOS                ;read DC constant for complementary E.COSY
  RE IM               ;read intermediate result
  AT #2               ;add with second DC to complementary E.COSY
  WR #2               ;write result
  IP1                 ;next phase BETA
  IP1
  IP2
  LO TO 2 TIMES 8     ;loop for 2*N=8 BETA-phases,
  IF #1               ;increment extension of .SER files
  IF #2
4 IP1
  LO TO 4 TIMES 4     ;TPPI
  IN=2                ;next t1 value
5 EXIT

PH1=(16) 4
PH2=(8) 0
PH3= 2
PH4= R1
```

ECOSY4C.AU

; Complementary E. COSY Programm for Kc=4

```
;32, 64, or 128  Scans per t1 value when you use VC4NC32.001
;                                                VC4NC64.001
;                                                VC4NC128.001

;No time consuming addition during acquisition is done with this
;E.COSY version.

  RVC VC4NC32         ;read VCLIST with total NS=32 per t1 value.
1 ZE
2 D2                  ;D2=1ms
  ECOS                ;read next NS value from VCLIST
3 D1                  ;Relaxation delay
  P1 PH1              ;Pulse sequence: ECOSY
  D0                  ;D0=0.9U;IN=DW
  P1 PH2              ;P1=90 PULSE
  D3                  ;D3=4U
  P1 PH3
  GO=3 PH4
  IP1                 ;next phase BETA
  IP2
  NM
  LO TO 2 TIMES 8     ;loop for 8 BETA-phases,
  WR #1
  IF #1
4 IP1
  IP1
  IN=1                ;next t1 value
5 EXIT

PH1=(8) 2
PH2=(8) 0
PH3= 0
PH4= R1
```

ECOSY4N.AU

;This Program Can Be Started Without Using The Pascal Program ECOSY In Advance.

```
; E. COSY Programm for Kc=4

;32, 64 or 128 Scans per t1 value, when you read VC4NC32.001
;                                                VC4NC64.001
;                                             or VC4NC128.001

;No time consuming addition is done during acquisition in this
;E.COSY version.

  RVC VC4NC64         ;read VCLIST with total NS=64 per t1 value
1 ZE
2 D2                  ;D2=1ms
  ECOS                ;read next NS value from VCLIST
3 D1                  ;Relaxation delay
  P1 PH1              ;Pulse sequence: ECOSY
  D0                  ;D0=0.9U;IN=DW
  P1 PH2              ;P1=90 PULSE
  D3                  ;D3=4U
  P1 PH3
  GO=3 PH4
  IP1                 ;next phase BETA
  IP2
  NM
  LO TO 2 TIMES 8     ;loop for 8 BETA-phases,
  WR #1
  IF #1
4 IP1
  IP1
  IN=1                ;next t1 value
5 EXIT

PH1=(8) 2
PH2=(8) 0
PH3= 2
PH4= R1

;This program can be started without using the PASCAL program
;ECOSY in advance. The relevant NS values are stored in the VCLIST
;VC4NC32.001.
```

EXTAD.AUR

; Data Acquisition Under External Address Advance

```
1 ZE
2 D1             ;RELAXATION DELAY
4 P1 PH1         ;TRANSM. PULSE , RECEIVER BLANKING
5 D5             ;2*D5+D7=DE ,HOLD PHASE AND BLANKING
6 D7 ADC         ;D7=2 USEC; INITIALIZE ADC
7 D5 PH2         ;SET DETECTOR PHASE, OPEN RECEIVER GATE
8 P8:X           ;EXT. DWELL PULSE (1 USEC), DIGITIZE DATA
                 ; POINT AND INCREMENT ADDRESS.
9 D8             ;D8=DW-1USEC, DWELL TIME = P8 + D8
10 L1 TO 8 TIMES C      ;HARDWARE LOOP FOR C=TD DATA POINTS
11 RCYC=2 PH3       ;LOOP FOR NEXT SCAN, OPTIONAL PHASE PROG.
12 EXIT

PH1=0 0 2 2                     ;TRANSM. PHASE  (OR USE :A)
PH2=0 0 0 0 3 3 3 3             ;DETECTOR PHASE (NORMALLY 0)
PH3=R0 R0 R2 R2 R1 R1 R3 R3     ;RECEIVER PHASE (OR USE QP)

;P1 IS EQUIVALENT TO PW
;D5 IS EQUIV. TO CA. DE/2
;D7=2 USEC IS SUFFICIENT FOR ADC COMMAND
;P8=1 USEC IS SUFFICIENT FOR DWELL PULSE
;D8=DW-P8
;VC LIST CONTAINS VALUE OF TD
```

GATEDEC.AU

; Heteronuclear Gated Decoupling Gives 1H-Coupled Spectrum With Full NOE

```
1 ZE             ;ZERO MEMORY, RESET SCAN COUNTER
2 D1 CPD S1      ;APPLY CPD DURING D1 WITH POWER S1
3 GO=2 DO        ;GATE DEC. OFF DURING AQ. (RD=0)
4 D2 BB
  EXIT           ;EXIT WITH DEC. ON FOR THERMAL EQUILIBRIUM

;NOE IS GENERATED DURING D1 WHICH SHOULD BE (2-4)*AQ
;OR 4*T1 FOR THE OBSERVED NUCLEUS.
;DEC. POWER S1 NEED ONLY BE HIGH ENOUGH TO GENERATE NOE.
;D2=1 MSEC
;P9 = 90 DEG DEC. PULSE AT POWER S1.
```

HAHNECHO.AUR

```
; Hahn Spin-Echo Sequence For Homonuc. J-Modulation
;   D1 - 90 - VD - 180 - VD - FID

1 ZE
2 D1            ;RELAX TO EQUILIBRIUM
3 P1 PH1        ;90 DEG PULSE WITH QP PHASE CONTROL
4 VD            ;PRECESSION OF SHIFT AND J-COUPLING
5 P2 PH2        ;180 DEG PULSE, PHASE PROG. PH2
6 VD            ;REFOCUS SHIFTS BUT NOT J-MODULATIONS
7 GO=2 PH3      ;ACQUIRE FID, LOOP TO 2
8 WR #1         ;STORE FID
9 IF #1         ;INCREMENT FILE EXTENSION
10 IN=1         ;LOOP FOR NEXT EXPERIMENT, INCREMENT VD
                ;LIST POINTER
11 EXIT

PH1= A0 A2 A2 A0 A1 A3 A3 A1    ;EXORCYCLE
     A2 A0 A0 A2 A3 A1 A1 A3
PH2= A0 A2 A0 A2 A1 A3 A1 A3
     A1 A3 A1 A3 A0 A2 A0 A2
PH3= R0 R2 R2 R0 R1 R3 R3 R1

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES NO. OF EXPERIMENTS = NO. OF DELAYS IN VD LIST
;CURRENT VD LIST MUST CONTAIN DELAYS (IN SEC) FOR ECHO.
;SIGNAL AMPLITUDES DECAY FOR SHORT VD ACCORDING TO
;  EXP(-2*VD/T2), FOR LONGER VD DECAY IS MORE RAPID DUE TO
;  DIFFUSION.
; D1 = CA. 5*T1 FOR COMPLETE RELAXATION
;P1 = 90 DEG PULSE
;P2 = 180 DEG PULSE
;RD=PW=0,
```

HCCCOSY.AU

; C-Relayed H,C-COSY - For Correlation Of Quaternary Carbons With Protons Bound To Neighboured Carbons Via Carbon-Carbon Double Quantum Coherence With Suppression Of Signals Of Protonated Carbons (for 1D version, see INEPREL1.AU)

```
;H. Kessler, W. Bermel & C.Griesinger, J. Magn. Reson. 62, 573 (1985)


1 ZE
2 D1 S1 DO                ;relaxation delay
3 P1:D PH1                ;90 deg H-1 pulse
4 D0                      ;t1/2
5 P4 PH4                  ;180 deg C-13 pulse
6 D0                      ;t1/2
7 D2                      ;1/2J(CH)
8 (P1 PH3):D (P3 PH5)     ;90 deg H-1 and C-13 pulse
9 D3 S2                   ;1/3J(CH)
10 D6 CPD                 ;D3+D6=D7  , D7=1/4J(CC)
11 P4 PH6                 ;180 deg C-13 pulse
12 D7                     ;1/4J(CC)
13 P3 PH7                 ;90 deg C-13 pulse
14 D5 DO                  ;1/2J(CH),
                          ;to suppress signals of protonated carbons
15 GO=2 PH8 CPD
16 WR #1
17 IF #1
18 IN=1
19 D2 DO
20 EXIT

PH1= 0 1 0 1 0 1 0 1 2 3 2 3 2 3 2 3
     2 3 2 3 2 3 2 3 0 1 0 1 0 1 0 1
PH3= 0 1 2 3 1 2 3 0 2 3 0 1 3 0 1 2
PH4= 0
PH5= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH7= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH8= R0 R0 R2 R2 R1 R1 R3 R3


;D1 : 1-5 T1
;S1 = 0H
;P1 : 90 deg H-1 pulse
;D0 = 3 usec
;P3,P4: 90, 180 deg C-13 pulse
;D2 : 1/(2J)CH
;D3 : 1/(3J)CH for all multiplicities (like in INEPTRD.AU)
;S2 : decoupler power level for CPD
;P9 : 90 deg H-1 pulse at power level S2
;D6 : D7 - D3
;D7 : 1/(4J)CC
;D5 : 1/(2J)CH
;NS : 128 * n
;DS : 4
```

```
;ND0 = 2
;IN : 1 / 2SW(H-1)
```

HETJRES.AUR

; Heteronuc. J-Resolved 2-D NMR Using The Gated Decoupling Technique

```
;  1H: BB-------------DO---------BB
;   X: D1 - 90 - D0 - 180 - D0 - FID

; F2 DOMAIN: X-NUCLEUS CHEMICAL SHIFTS
; F1 DOMAIN: 1/2 THE X-H J-COUPLINGS

1 ZE
2 D1 CPD S1      ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D2 S2          ;SWITCH TO HIGHER POWER FOR GOOD DEC.
4 P1 PH1         ;90 DEG X PULSE
5 D0             ;FIRST HALF OF EVOLUTION FOR CHEM. SHIFTS
6 P2 PH2 DO      ;180 DEG X PULSE, GATE DEC. OFF
7 D0             ;REFOCUS SHIFTS AND LET J(XH) EVOLVE
8 GO=2 PH3       ;ACQUIRE FID WITH BB DEC.
9 D2 CPD S1      ;TURN DOWN DEC. POWER
10 WR #1         ;STORE FID
11 IF #1         ;INCREMENT FILE NUMBER
12 IN=1          ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
13 D2 DO
14 EXIT          ;EXIT WITH DEC. GATED OFF

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;EXORCYCLE
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES THE NO. OF FIDS
;RD=PW=0
;NS CAN BE 4, 8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;D1 = 1-5*T1 FOR X-NUCLEUS, S1 SHOULD BE SUFFICIENT TO HOLD NOE
;D2 = 5 MSEC, S2 SHOULD BE FOR OPTIMAL DECOUPLING TO INSURE
; MINIMAL LOSS OF TRANSVERSE MAGNETIZATION DURING EVOLUTION.
;P1,P2 = 90,180 DEG X PULSE
;D0 = 3E-6 INITIAL DELAY
;IN = 0.25/SW1
;SW1 = CA. 0.8*(LARGEST J) FOR NO FOLDING OF QUARTETS
;ND0 = 2

;NO TILT NECESSARY
```

HOESY.AU

```
; Two-Dimensional Heteronuclear Overhauser Experiment
;LIT: C.YU AND G.LEVY , JACS 1984, 106, 6533

;   1H: D1 - 90 - D0 -     - D0 - 90 - D9 -    - BB
; XNUC: D1 -         - 180 -              - 90 - FID

1 ZE                 ;ZERO MEMORY
2 D1 S3 DO           ;RELAXATION DELAY, SWITCH DECOUPLER POWER
                     ;DECOUPLER GATED OFF
3 (P1 PH1):D         ;90 DEG PULSE DECOUPLER
4 D0                 ;FIRST HALF OF EVOLUTION TIME
5 P6 PH3             ;180 DEG PULSE XNUC TO REMOVE SCALAR
                     ;INTERACTIONS
6 D0                 ;SECOND HALF OF EVOLUTION TIME
7 (P1 PH2):D         ;CREATION OF LONGITUDINAL H-SPIN MAGNETIZATION
8 D9 S2              ;MIXING TIME FOR MAGNETIZATION EXCHANGE BY
                     ;DIPOLAR INTERACTION,SWITCH DECOUPLER POWER
9 P5 PH4             ;OBSERVE PULSE
10 GO=2 PH5 CPD      ;ACQUISITION WITH BROADBAND OR WALTZ-DECOUPLING
11 IPHA              ;INCREMENT ALL PHASES BY 90 DEGREE
12 LO TO 2 TIMES C
13 WR #1
14 IF #1
15 IN=1
16 EXIT

PH1=B0 B0 B0 B0 B1 B1 B1 B1 B2 B2 B2 B2 B3 B3 B3 B3
PH2=B0 B1 B2 B3 B1 B2 B3 B0 B2 B3 B0 B1 B3 B0 B1 B2
PH3=A0
PH4=A0 A1 A2 A3
PH5=R0 R2

;P1=90 DEG 1H-DEC; P5,P6=90,180DEG X-NUCLEUS;
;S3=0H;S2=FOR BB OR WALTZ-DECOUPLING; D1=RELAXATION DELAY,
;D9=MIXING TIME; D0=0.000003; ND0=2, SW1=0.5SW 1H
;NS=N*16(N=1,2,3...); DS=2,4; TOTAL SCAN-NUMBER FOR ONE
;EXPERIMENT=16*N*C; MINIMUM PHASE-CYCLE NS=16 AND C=1
```

HOMODEC.AU

; Homonuclear Decoupling Using One Freq. List And One Power Setting.

```
1 FL #1          /INPUT FREQ. LIST
                 ;READ IN FREQ. LIST AND INITIALIZE POINTER
2 ZE
3 D1 HD O2 S3    ;TURN ON HOMODEC. WITH POWER S3 AND SET
                 ;O2 FREQ. FROM CURRENT FL LIST, INCREMENT
                 ;FREQ. POINTER.
4 GO=4           ;ACQUIRE DATA
5 WR #2          /INPUT FID
                 ;STORE FID
6 IF #2          ;INCREMENT FILE EXTENSION
7 IN=2           ;LOOP FOR NEXT EXPERIMENT
8 D1 DO
  EXIT           ;EXIT WITH DEC. OFF

;PROGRAM REQUESTS A FILENAME FOR FREQ. LIST AND FID STORAGE
;D1 = 1 SEC TO SET FREQ.
;S3 = DESIRED DEC. POWER
;NE DEFINES NUMBER OF ITEMS IN FL LIST = NO. OF EXPERIMENTS
;USE RD AS DESIRED AND AT LEAST 2 DUMMY SCANS
```

HOMODEC2.AUR

; Homonuclear Decoupling Using Several Freq. Lists And A Different Power Setting (Or Variation Of Other Parameters) For Each List.

```
1 FL #1         /INPUT FREQ. LIST
                ;READ IN FREQ. LIST AND INITIALIZE POINTER
2 RJ #2         /INPUT JOB PARAMETER
                ;READ IN JOB PARAMETER FILE TO SET DEC. POWER
                ; OR OTHER PARAMETERS.
3 IF #1
4 IF #2
5 ZE
6 D1 HD O2 S3   ;TURN ON HOMODEC. WITH POWER S3 AND SET
                ;O2 FREQ. FROM CURRENT FL LIST, INCREMENT
                ;FREQ. POINTER.
7 GO=7          ;ACQUIRE DATA
8 WR #3         /INPUT FID
                ;STORE FID
9 IF #3                 ;INCREMENT FILE EXTENSION
10 LO TO 5 TIMES C      ;LOOP FOR NEXT FREQ. IN CURRENT LIST
11 VC                   ;INCREMENT VC LIST POINTER
12 IN=1         ;LOOP FOR NEXT FREQ. LIST
13 D1 DO
  EXIT          ;EXIT WITH DEC. OFF

;PROGRAM REQUESTS A FILENAME FOR FREQ. LIST,
; JOB PARAMETER FILE AND FID STORAGE
;D1 = 1 SEC TO SET FREQ.
;S3 = DESIRED DEC. POWER
;NE DEFINES NUMBER OF DIFFERENT FL LISTS AND THERE MUST BE
;  ONE JOB PARAMETER FILE FOR EACH LIST.
;VC LIST CONTAINS NE ENTRIES SPECIFYING THE NUMBER OF O2
;  VALUES IN EACH LIST.
;USE RD AS DESIRED AND AT LEAST 2 DUMMY SCANS

;NB: FIRST FL LIST SHOULD BE THE LONGEST ONE AND IN MEMORY
; WHEN AU IS STARTED.
```

HOMOINVG.AUR

; Homonuclear Decoupling Using Several Freq. Lists And A Different Power Setting (Or Variation Of Other Parameters) For Each List. Using The Inverse-Gating Technique To Minimize NOE.

```
1 FL #1         /INPUT FREQ. LIST
                ;READ IN FREQ. LIST AND INITIALIZE POINTER
2 RJ #2         /INPUT JOB PARAMETER
                ;READ IN JOB PARAMETER FILE TO SET DEC. POWER
                ; OR OTHER PARAMETERS.
3 IF #1
4 IF #2
5 ZE
6 D1 O2 S3      ;SET POWER S3 AND SET
                ;O2 FREQ. FROM CURRENT FL LIST, INCREMENT
                ;FREQ. POINTER.
7 D2 DO         ;RELAXATION WITHOUT DEC. (SUPPRESS NOE)
  D3 HD         ;TURN ON DEC.
  GO=7          ;ACQUIRE DATA
8 WR #3         /INPUT FID
                ;STORE FID
9 IF #3                 ;INCREMENT FILE EXTENSION
10 LO TO 5 TIMES C      ;LOOP FOR NEXT FREQ. IN CURRENT LIST
11 VC                   ;INCREMENT VC LIST POINTER
12 IN=1         ;LOOP FOR NEXT FREQ. LIST
13 D1 DO
  EXIT          ;EXIT WITH DEC. OFF

;PROGRAM REQUESTS A FILENAME FOR FREQ. LIST,
; JOB PARAMETER FILE AND FID STORAGE
;RD=0.
;D1 = 0.5 SEC TO SET FREQ.
;S3 = DESIRED DEC. POWER
;D2+D3+AQ IS THE TOTAL RELAXATION TIME; TO MINIMIZE NOE SET
;  D2=1-5*T1, D3=0.1-5 MSEC TO MINIMIZE TRANSIENT EFFECTS WHEN
;  DECOUPLER IS TURNED ON.
;NE DEFINES NUMBER OF DIFFERENT FL LISTS AND THERE MUST BE
;  ONE JOB PARAMETER FILE FOR EACH LIST.
;VC LIST CONTAINS NE ENTRIES SPECIFYING THE NUMBER OF O2
;  VALUES IN EACH LIST.
;USE RD AS DESIRED AND AT LEAST 2 DUMMY SCANS

;NB: FIRST FL LIST SHOULD BE THE LONGEST ONE AND SHOULD BE
;  IN MEMORY WHEN AU IS STARTED.
```

HPCYCL.AU

??

HPCYCLGS.AU

??

HPDEC.AUR

; High-Power Decoupling For Mas Applications Using Extra 1H Decoupler Endstage And Normal Low-Power Transmitter.

```
1 ZE
2 D1 DO         ;RELAXATION, GATE DEC. OFF
  P1:A CW       ;TRANSMITTER PULSE, TURN ON CW DEC.
3 GO=2          ;ACQUIRE FID WITH HIGH-POWER CW DEC.
4 D2 DO         ;GATE DEC. OFF
5 EXIT

;RD=PW=0
;P1 = TRANSMITTER PULSE FOR OBSERVATION
;AQ ONLY AS LONG AS NEEDED FOR EXPECTED RESOLUTION (E.G. T2*)
;  BUT MAX. VALUE OF 0.5 SEC TO AVOID DESTROYING PROBEHEAD!!
;D1 = 1-5*T1 OF X-NUCLEUS BUT > 20*AQ TO AVOID EXCESSIVE
;  HEATING FROM DECOUPLER.
;D2 = 1 MSEC
;DP DEFINES DEC. POWER, NORMALLY 1H-5H WITH MAX. 16W TO FEED
;  HP 1H ENDSTAGE.
```

HPDECGS.AU

; ??

HPDECHP.AUR

; High-Power Decoupling For MAS Applications Using Extra 1H Decoupler Endstage And AR Amplifier For X-Nuclei Irradiation

```
; *************** USE TLO ***************

1 ZE
2 D1 DO         ;RELAXATION, GATE DEC. OFF
  P4:C8         ;UNBLANKING OF AR AMPLIFIER
  (P1 PH1 CW):T:C8   ;AR TRANSMITTER PULSE, TURN ON CW DEC.
3 GO=2 CW       ;ACQUIRE FID WITH HIGH-POWER CW DEC.
4 D2 DO         ;GATE DEC. OFF
5 EXIT

PH1= 0 0 2 2 1 1 3 3

;RD=PW=0
;P1 = TRANSMITTER PULSE FOR OBSERVATION
;AQ ONLY AS LONG AS NEEDED FOR EXPECTED RESOLUTION (E.G. T2*)
;  BUT MAX. VALUE OF 0.5 SEC TO AVOID DESTROYING PROBEHEAD!!
;D1 = 1-5*T1 OF X-NUCLEUS BUT > 20*AQ TO AVOID EXCESSIVE
;  HEATING FROM DECOUPLER.
;D2 = 1 MSEC
;P4 = 40 USEC FOR UNBLANKING
;DP DEFINES DEC. POWER, NORMALLY 1H-5H WITH MAX. 16W TO FEED
;  HP 1H ENDSTAGE.
```

HPDECHPS.AUR

; High-Power Decoupling For MAS Applications Using Extra 1H Decoupler Endstage And AR Amplifier For X-Nuclei Irradiation

```
; ************* USE TLO *************

1 ZE
2 D1 DO         ;RELAXATION, GATE DEC. OFF
  P4:C8         ;UNBLANKING OF AR AMPLIFIER
  (P1 PH1 CW):T:C8   ;AR TRANSMITTER PULSE, TURN ON CW DEC.
3 GS=2 CW       ;ACQUIRE FID WITH HIGH-POWER CW DEC.
4 D2 DO         ;GATE DEC. OFF
5 EXIT

PH1= 0 0 2 2 1 1 3 3

;RD=PW=0
;P1 = TRANSMITTER PULSE FOR OBSERVATION
;AQ ONLY AS LONG AS NEEDED FOR EXPECTED RESOLUTION (E.G. T2*)
;  BUT MAX. VALUE OF 0.5 SEC TO AVOID DESTROYING PROBEHEAD!!
;D1 = 1-5*T1 OF X-NUCLEUS BUT > 20*AQ TO AVOID EXCESSIVE
;  HEATING FROM DECOUPLER.
;D2 = 1 MSEC
;P4 = 40 USEC FOR UNBLANKING
;DP DEFINES DEC. POWER, NORMALLY 1H-5H WITH MAX. 16W TO FEED
;  HP 1H ENDSTAGE.
```

HPDECSET.AUR

; Set-Up Mode - High-Power Decoupling For MAS Applications Using Extra 1H Decoupler Endstage.

```
1 ZE
2 D1 DO S1      ;RELAXATION, GATE DEC. OFF, SET POWER
```

```
  P1:A CW       ;TRANSMITTER PULSE, TURN ON CW DEC.
3 GS=2          ;ACQUIRE FID WITH HIGH-POWER CW DEC.
4 D2 DO         ;GATE DEC. OFF
5 EXIT

;RD=PW=0
;P1 = TRANSMITTER PULSE FOR OBSERVATION
;AQ ONLY AS LONG AS NEEDED FOR EXPECTED RESOLUTION (E.G. T2*)
;  BUT MAX. VALUE OF 0.5 SEC TO AVOID DESTROYING PROBEHEAD!!
;D1 = 1-5*T1 OF X-NUCLEUS BUT > 20*AQ TO AVOID EXCESSIVE
;  HEATING FROM DECOUPLER.
;D2 = 1 MSEC
;S1 DEFINES DEC. POWER, NORMALLY 1H-5H WITH MAX. 16W TO FEED
;  HP 1H ENDSTAGE.
```

INAD2D.AUR

; Inadequate 2-D NMR Using J(XX) To Give X-X Connectivities. An Extended Ernst-Type Phase Cycle Suppresses Single Quantum Peaks. A Ca. 125 Deg Conversion Pulse Suppresses Unwanted F1 Image Peaks By Selecting The Coherence Transfer Echo (N-Type Selection).

```
; T.H.MARECI AND R.FREEMAN, J.MAGN.RES., 48, 158 (1982)

;  D1-90-D2-180-D2-90-D0-125-FID

; F2 DOMAIN: X-NUCLEUS SHIFTS AND J(XX)
; F1 DOMAIN: DOUBLE QUANTUM FREQ. F(A)+F(B) WHERE F IS
;   THE RESONANCE FREQ. RELATIVE TO O1. PAIRS OF
;   CORRELATION PEAKS APPEAR ON INDIVIDUAL ROWS OF MATRIX.

1 ZE
2 D1 CPD S1     ;RELAX, LOW POWER DEC. FOR NOE
3 D3 S2         ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1        ;90 DEG PULSE
5 D2            ;SPIN-ECHO PERIOD= (2N+1)/4J(XX)
6 P2 PH2        ;180 DEG
7 D2            ;SECOND ECHO PERIOD
8 P1 PH3        ;CREATE DBL. QUANTUM COHERENCE
9 D0            ;EVOLUTION
10 P3 PH4       ;CONVERT DBL. QUANTUM TO SINGLE
12 GO=2 PH5     ;DETECTION (32 TRANSIENTS)
13 D3 S1        ;LOWER DEC. POWER
   IPHA         ;INCREMENT ALL PHASE PROGRAMS (CYCLOPS)
   LO TO 2 TIMES C   ;LOOP FOR FURTHER PHASE CYCLING (C=1,2 OR 4)
                ;C*NS TRANSIENTS PER FID
14 WR #1        ;STORE FID (SERIES FILE)
15 IF #1        ;INCREMENT FILE EXTENSION
16 IN=1         ;LOOP FOR NEXT EXPERIMENT
17 D2 DO        ;GATE DEC. OFF
18 EXIT

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1

PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2

PH3=A0 A1 A2 A3

PH4=A0

PH5=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;32-PHASE CYCLE FOR DBL. QUANTUM, INCREMENT ALL PHASES IN
;  90 DEG. STEPS FOR QP-TYPE QUADRATURE.
;NS=32 FOR DOUBLE QUANTUM SELECTION
;VC LIST: C=1, OR =2 FOR MIN. QUAD CYCLE, =4 FOR COMPLETE CYCLE

;PROGRAM REQUESTS FILENAME FOR .SER FILE
;NE DEFINES NO. OF FIDS = TD1, NS*C SCANS PER FID
;RD=PW=0
;P1,P2 = 90,180 DEG PULSES
;P3=120 FOR MAX. DBL. QUANTUM SIGNAL, =135 DEG FOR BETTER
;   F1 QUAD IMAGE SUPPRESSION.
;D1=1-5*T1 , S1 = CA. 0.5 WATT FOR NOE
;S2 = OPTIMAL DEC. POWER (P9=90 DEG DEC. PULSE) IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING D2.
;D2=(2N+1)/4J WHERE N=0,1,2,3,...
;D3=5 MSEC
;D0=2E-6
;ND0=1
```

```
;IN=0.5/SW1,  SW1=SW TO AVOID FOLDING IN F1
;SYMMETRIZATION OF MATRIX IS NOT POSSIBLE
;RECOMMEND:  SI=TD SO THAT HZ/PT2=< 0.5*J(XX)
;            NE=TD1=SI1/2  SO THAT HZ/PT1=10-40 HZ
```

INAD2D2(3?).AU

; Inadequate 2-D NMR Using J(XX) To Give X-X Connectivities. An Extended Ernst-Type Phase Cycle Suppresses Single Quantum Peaks Using 45° Phase Shifts

```
;  D1-90-D2-180-D2-90-D0-90-FID
; F2 DOMAIN: X-NUCLEUS SHIFTS AND J(XX)
; F1 DOMAIN: DOUBLE QUANTUM FREQ. F(A)+F(B) WHERE F IS
;   THE RESONANCE FREQ. RELATIVE TO O1. PAIRS OF
;   CORRELATION PEAKS APPEAR ON INDIVIDUAL ROWS OF MATRIX.

1 ZE
2 D1 S1 CPD     ;RELAX, LOW POWER DEC. FOR NOE
4 P1 PH1        ;90 DEG PULSE
5 D2            ;SPIN-ECHO PERIOD= (2N+1)/4J(XX)
6 P2 PH2        ;180 DEG
7 D2            ;SECOND ECHO PERIOD
8 P1 PH1        ;CREATE DBL. QUANTUM COHERENCE
9 D0            ;EVOLUTION
10 P3 PH3       ;CONVERT DBL. QUANTUM TO SINGLE
12 GO=2 PH4     ;DETECTION (32 TRANSIENTS)
14 WR #1        ;STORE FID (SERIES FILE)
15 IF #1        ;INCREMENT FILE EXTENSION
16 IN=1         ;LOOP FOR NEXT EXPERIMENT
17 D2 DO        ;GATE DEC. OFF
18 EXIT

PH1=(8) 0 4 2 6 3 7 1 5 3 7 5 1 6 2 4 0
        2 6 4 0 5 1 3 7 1 5 3 7 4 0 2 6

PH2=(8) 0 4 2 6 3 7 1 5 3 7 5 1 6 2 4 0
        2 6 4 0 5 1 3 7 1 5 3 7 4 0 2 6
        4 0 6 2 7 3 5 1 7 3 1 5 2 6 0 4
        6 2 0 4 1 5 7 3 5 1 7 3 0 4 6 2

PH3=    0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1
        2 2 2 2 2 2 2 2 3 3 3 3 3 3 3 3

PH4= R0 R0 R2 R2 R1 R1 R3 R3

;32-PHASE CYCLE FOR DBL. QUANTUM
;NS=32 FOR DOUBLE QUANTUM SELECTION

;PROGRAM REQUESTS FILENAME FOR .SER FILE
;NE DEFINES NO. OF FIDS = TD1, NS SCANS PER FID
;RD=PW=0
;P1,P2 = 90,180 DEG PULSES
;P3=90 DEG PULSE
;D1=1-5*T1 ,
;S1 = DEC. POWER FOR CPD
;D2=(2N+1)/4J WHERE N=0,1,2,3,...
;D3=5 MSEC
;D0=2E-6
;ND0=1
;IN=0.5/SW1,  SW1=SW TO AVOID FOLDING IN F1
;SYMMETRIZATION OF MATRIX IS NOT POSSIBLE
;RECOMMEND:  SI=TD SO THAT HZ/PT2=< 0.5*J(XX)
;            NE=TD1=SI1/2  SO THAT HZ/PT1=10-40 HZ
```

INADCOMP.AUR

; Inadequate Double Quantum 1-D NMR Using J(XX) With Composite Pulses (May Give Improvement When 90 Deg Pulse Is Longer Than Ca. 20 Usec). Suppression Of Single Quantum Signals Using Ernst-Type Double Quantum Phase Cycling And Fid Storage After Each 32 Transient Block.

```
;   D1 - 90 - D2 - 180 - D2 - 90 - D3 - 90 - FID

1 ZE
2 D1 CPD S1      ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D4 S2          ;SWITCH TO HIGHER DEC. POWER
4 P2 PH1         ;90 DEG COMPOSITE X PULSE
  P2 PH6
  P1 PH1
5 D2             ;EVOLUTION OF SHIFTS AND J(XX)
6 P3 PH2         ;180 DEG COMPOSITE PULSE
  P2 PH7
  P1 PH2
7 D2             ;REFOCUS SHIFTS, J-MOD. CONTINUES
8 P1 PH3         ;90 DEG PULSE, CREATE DBL. QUANTUM COHERENCE
9 D3             ;DBL. QUANTUM PRECESSION (NORMALLY 3 USEC)
10 P2 PH8        ;90 DEG COMPOSITE  CONVERSION PULSE
   P2 PH9
   P1 PH8
   P3 PH9
   P2 PH8
   P1 PH9
   P1 PH4
11 GO=2 PH5      ;ACQUIRE FID USING PHASE PROGRAM TO CANCEL
                 ;UNWANTED SINGLE QUANTUM SIGNALS.
                 ;ONE CYCLE OF 32 TRANSIENTS
12 D4 CPD S1         ;SWITCH BACK TO LOWER POWER
13 WR #1         ;SAVE FID AFTER EACH 32 TRANSIENT CYCLE
14 IPHA          ;INCREMENT ALL PHASE PROGRAMS
22 LO TO 2 TIMES 4      ;LOOP FOR 4 CYCLES (CYCLOPS) FOR
                        ;TOTAL OF 4*32=128 TRANSIENTS
23 IN=2          ;REPEAT CYCLE OF 128 NE TIMES
24 D4 CPD S1
25 EXIT          ;EXIT WITH LOW POWER DEC.

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1

PH6= 2 3 0 1 2 3 0 1 2 3 0 1 2 3 0 1
     0 1 2 3 0 1 2 3 0 1 2 3 0 1 2 3
PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2
PH7= 2 3 0 1 3 0 1 2 0 1 2 3 1 2 3 0
PH3=A0 A1 A2 A3

PH4=A0
PH8= 1
PH9= 3
PH5=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;D1 = 1-5*T1 X-NUCLEUS
;P1,P2,P3 = 90,180,270 DEG X PULSE
;D3 = 3E-6 TO ALLOW PHASE SWITCHING
;D4 = 5 MSEC FOR DEC. POWER SWITCHING
;S1 = CA. 0.5 WATT, LOW-POWER FOR NOE
;S2 = NORMAL POWER FOR OPTIMAL DEC. WITH CPD

;D2 = (2N+1)/(4J) WHERE N=0,1,2,... TO CREATE OPTIMUM
;  DBL. QUANTUM COHERENCE (ANTI-PHASE J(XX) DOUBLETS)
```

```
;RD=PW=0
;NS=32 , DS=4
;NE DEFINES NO. OF SUPERCYCLES
;TOTAL TRANSIENTS = 128*NE
```

INADEQ.AUR

; Inadequate Double Quantum 1-D NMR Using J(XX) With Suppression Of Single Quantum Signals Using The Basic 32-Phase Cycle Of Freeman With Automatic Storage Of Data.

```
;   D1 - 90 - D2 - 180 - D2 - 90 - D3 - 90 - FID

1 ZE
2 D1 CPD S1      ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D4 S2          ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1         ;90 DEG X PULSE
5 D2             ;EVOLUTION OF SHIFTS AND J(XX)
6 P2 PH2         ;180 DEG PULSE
7 D2             ;REFOCUS SHIFTS, J-MOD. CONTINUES
8 P1 PH1         ;90 DEG PULSE, CREATE DBL. QUANTUM COHERENCE
9 D3             ;DBL. QUANTUM PRECESSION (NORMALLY 3 USEC)
10 P1 PH3        ;90 DEG PULSE, CONVERT DBL. QUANTUM TO SINGLE
                 ;QUANTUM COHERENCE
11 GO=2 PH4      ;ACQUIRE FID USING PHASE PROGRAM TO CANCEL
                 ;UNWANTED SINGLE QUANTUM SIGNALS
12 D4 DO
13 WR #1         ;STORE FID AFTER COMPLETE CYCLE
14 IN=2          ;REPEAT FOR NE CYCLES
   EXIT          ;EXIT WITH DEC. OFF

PH1=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH2=A0 A2 A0 A2 A0 A2 A0 A2
    A1 A3 A1 A3 A1 A3 A1 A3
PH3=A0 A0 A1 A1 A2 A2 A3 A3
    A1 A1 A2 A2 A3 A3 A0 A0
    A2 A2 A3 A3 A0 A0 A1 A1
    A3 A3 A0 A0 A1 A1 A2 A2
PH4=R0 R0 R3 R3 R2 R2 R1 R1
    R1 R1 R0 R0 R3 R3 R2 R2
    R2 R2 R1 R1 R0 R0 R3 R3
    R3 R3 R2 R2 R1 R1 R0 R0

;D1 = 1-5*T1 X-NUCLEUS
;P1,P2 = 90,180 X PULSE
;D3 = 3E-6 TO ALLOW PHASE SWITCHING
;D4 = 5 MSEC FOR DEC. POWER SWITCHING
;S1 = CA. 0.5 WATT, LOW-POWER FOR NOE
;S2 = NORMAL POWER FOR OPTIMAL CPD DEC., IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING D2.

;D2 = (2N+1)/(4J) WHERE N=0,1,2,... TO CREATE OPTIMUM
;  DBL. QUANTUM COHERENCE (ANTI-PHASE J(XX) DOUBLETS)

;RD=PW=0
;NS=32*N , DS=4
;NE=NUMBER OF CYCLES OF NS SCANS.
```

INADEQ2.AUR

; Inadequate Double Quantum 1-D NMR Using J(XX) With Suppression Of Single Quantum Signals Using Ernst-Type Double Quantum Phase Cycling And FID Storage After Each 32 Transient Block.

```
;   D1 - 90 - D2 - 180 - D2 - 90 - D3 - 90 - FID

1 ZE
2 D1 CPD S1     ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D4 S2         ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1        ;90 DEG X PULSE
5 D2            ;EVOLUTION OF SHIFTS AND J(XX)
6 P2 PH2        ;180 DEG PULSE
7 D2            ;REFOCUS SHIFTS, J-MOD. CONTINUES
8 P1 PH3        ;90 DEG PULSE, CREATE DBL. QUANTUM COHERENCE
9 D3            ;DBL. QUANTUM PRECESSION (NORMALLY 3 USEC)
10 P1 PH4       ;90 DEG PULSE, CONVERT DBL. QUANTUM TO SINGLE
                ;QUANTUM COHERENCE
11 GO=2 PH5     ;ACQUIRE FID USING PHASE PROGRAM TO CANCEL
                ;UNWANTED SINGLE QUANTUM SIGNALS.
                ;ONE CYCLE OF 32 TRANSIENTS
12 D4 CPD S1        ;SWITCH BACK TO LOWER POWER
13 WR #1        ;SAVE FID AFTER EACH 32 TRANSIENT CYCLE
14 IPHA         ;INCREMENT ALL PHASE PROGRAMS
15 LO TO 2 TIMES 4      ;LOOP FOR 4 CYCLES (CYCLOPS) FOR
                        ;TOTAL OF 4*32=128 TRANSIENTS
16 IN=2         ;REPEAT CYCLE OF 128 NE TIMES
17 D4 CPD S1
18 EXIT         ;EXIT WITH LOW POWER DEC.

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1

PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2

PH3=A0 A1 A2 A3

PH4=A0

PH5=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;D1 = 1-5*T1 X-NUCLEUS
;P1,P2 = 90,180 X PULSE
;D3 = 3E-6 TO ALLOW PHASE SWITCHING
;D4 = 5 MSEC FOR DEC. POWER SWITCHING
;S1 = CA. 0.5 WATT, LOW-POWER FOR NOE
;S2 = NORMAL POWER FOR OPTIMAL CPD DEC., IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING D2.

;D2 = (2N+1)/(4J) WHERE N=0,1,2,... TO CREATE OPTIMUM
;  DBL. QUANTUM COHERENCE (ANTI-PHASE J(XX) DOUBLETS)

;RD=PW=0
;NS=32 , DS=4
;NE DEFINES NO. OF SUPERCYCLES
;TOTAL TRANSIENTS = 128*NE
```

INADEQR.AUR

; Refocussed Inadequate Double Quantum 1-D NMR Using J(XX) With Suppression Of Single Quantum Signals Using Ernst-Type Double Quantum Phase Cycling And FID Storage After Each 32 Transient Block.

```
;   D1 - 90 - D2 - 180 - D2 - 90 - D3 - 90 - FID

1 ZE
2 D1 CPD S1      ;RELAXATION, LOW-POWER DEC. FOR NOE
3 D4 S2          ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1         ;90 DEG X PULSE
5 D2             ;EVOLUTION OF SHIFTS AND J(XX)
6 P2 PH2         ;180 DEG PULSE
7 D2             ;REFOCUS SHIFTS, J-MOD. CONTINUES
8 P1 PH3         ;90 DEG PULSE, CREATE DBL. QUANTUM COHERENCE
9 D3             ;DBL. QUANTUM PRECESSION (NORMALLY 3 USEC)
10 P1 PH4        ;90 DEG PULSE, CONVERT DBL. QUANTUM TO SINGLE
                 ;QUANTUM COHERENCE
   D2            ;FIRST PART OF REFOCUSSING PERIOD
   P2 PH5        ;180 DEG REFOCUSSING PULSE
   D2            ;SECOND REFOCUSSING PERIOD
11 GO=2 PH6      ;ACQUIRE FID USING PHASE PROGRAM TO CANCEL
                 ;UNWANTED SINGLE QUANTUM SIGNALS.
                 ;ONE CYCLE OF 32 TRANSIENTS
12 D4 CPD S1         ;SWITCH BACK TO LOWER POWER
13 WR #1         ;SAVE FID AFTER EACH 32 TRANSIENT CYCLE
14 IPHA          ;INCREMENT ALL PHASE PROGRAMS
15 LO TO 2 TIMES 4      ;LOOP FOR 4 CYCLES (CYCLOPS) FOR
                        ;TOTAL OF 4*32=128 TRANSIENTS
16 IN=2          ;REPEAT CYCLE OF 128 NE TIMES
17 D4 CPD S1
18 EXIT          ;EXIT WITH LOW POWER DEC.

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1

PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2

PH3=A0 A1 A2 A3

PH4=A0
PH5=A0 A0 A2 A2

PH6=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;D1 = 1-5*T1 X-NUCLEUS
;P1,P2 = 90,180 X PULSE
;D3 = 3E-6 TO ALLOW PHASE SWITCHING
;D4 = 5 MSEC FOR DEC. POWER SWITCHING
;S1 = CA. 0.5 WATT, LOW-POWER FOR NOE
;S2 = NORMAL POWER FOR OPTIMAL DEC., IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING D2.

;D2 = (2N+1)/(4J) WHERE N=0,1,2,... TO CREATE OPTIMUM
;  DBL. QUANTUM COHERENCE (ANTI-PHASE J(XX) DOUBLETS)
;  THE ADDITIONAL ECHO SEQUENCE FOLLOWING STEP 10 ALLOWS
;  ANTI-PHASE MULTIPLETS TO BECOME IN-PHASE, WHICH MAY BE USEFUL
;  WHEN OBSERVING SMALL COUPLINGS.  NB: MAGNETIZATION IS BEING
;  LOST DURING THIS TIME BY T2 RELAXATION.

;RD=PW=0
;NS=32 , DS=4
;NE DEFINES NO. OF SUPERCYCLES
;TOTAL TRANSIENTS = 128*NE
```

INADSYM.AUR

```
; Symmetrized Inadequate 2-D (Ernst-Type Phase Cycle) Using Split T1 Domain
  And Ca. 120 Deg Conversion Pulse To Give Cosy-Like Symmetry Representation
;      D.L.TURNER, J.MAGN.RES. 49, 175 (1982)
;      T.H.MARECI AND R.FREEMAN, J.MAGN.RES., 48, 158 (1982)

;  D1-90-D2-180-D2-90-D0-125-D0-FID

; F2 DOMAIN: X-NUCLEUS SHIFTS AND J(XX)
; F1 DOMAIN: ONE-HALF SHIFTS AND COUPLINGS
; CORRELATIONS APPEAR AS OFF-DIAGONAL PEAKS AS IN COSY.

1 ZE
2 D1 CPD S1      ;RELAX, LOW POWER DEC. FOR NOE
3 D3 S2          ;SWITCH TO HIGHER DEC. POWER
4 P1 PH1         ;90 DEG PULSE
5 D2             ;SPIN-ECHO PERIOD= (2N+1)/4J(XX)
6 P2 PH2         ;180 DEG
7 D2             ;SECOND ECHO PERIOD
8 P1 PH3         ;CREATE DBL. QUANTUM COHERENCE
9 D0             ;EVOLUTION
10 P3 PH4        ;CONVERT DBL. QUANTUM TO SINGLE, SELECT COHERENCE
                 ;  TRANSFER ECHO.
11 D0            ;EVOLUTION
12 GO=2 PH5      ;DETECTION, 32-PHASE CYCLE FOR DBL. QUANTUM
13 D3 S1         ;LOWER DEC. POWER
   IPHA                  ;INCREMENT ALL PHASES BY 90 DEG
   LO TO 2 TIMES C       ;LOOP FOR NEXT PHASE CYCLE
14 WR #1         ;STORE FID (SERIES FILE)
15 IF #1         ;INCREMENT FILE EXTENSION
16 IN=1          ;LOOP FOR NEXT EXPERIMENT
17 D2 DO         ;GATE DEC. OFF
18 EXIT

PH1=A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3
    A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1 A2 A3 A0 A1
PH2=A0 A1 A2 A3 A1 A2 A3 A0 A2 A3 A0 A1 A3 A0 A1 A2
PH3=A0 A1 A2 A3
PH4=A0
PH5=R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0
    R2 R0 R2 R0 R0 R2 R0 R2 R2 R0 R2 R0 R0 R2 R0 R2

;PROGRAM REQUESTS FILENAME FOR .SER FILE
;NS=32 FOR DBL. QUANTUM PHASE CYCLE
;VC LIST ENTRY =1,2, OR 4 (COMPLETE 128-PHASE CYCLE)
;NE DEFINES NO. OF FIDS = TD1, TOTAL SCANS NS*C PER FID
;RD=PW=0
;P1,P2= 90,180 DEG PULSES
;P3 = 120 DEG FOR MAX. DBL.QUANTUM SIGNAL, =135 DEG FOR
;      BETTER F1 QUAD IMAGE SUPPRESSION.
;D1=1-5*T1 , S1 = CA. 0.5 WATT FOR NOE
;S2 = OPTIMAL DEC. POWER , IMPORTANT TO MINIMIZE
;      LOSS OF TRANSVERSE MAGNETIZATION DURING D2.
;D2=(2N+1)/4J WHERE N=0,1,2,3,...
;D3=5 MSEC
;D0=2E-6
;IN=0.25/SW1,  SW1=SW/4, ND0=2
;SYMMETRIZATION OF MATRIX IS POSSIBLE
; BY ZERO-FILLING SO THAT SI1=0.5*SI, I2D NEED NOT BE 1.

;RECOMMEND:  SI=TD SO THAT HZ/PT2=< 0.5*J(XX)
;            SI1=SI/2,  NE=TD1=SI/4
```

INEPINAD.AU

```
; INEPT-INADEQUATE
;O.W. Sorensen, R. Freeman, T.A. Frenkiel, T.H. Mareci & R. Schuck,
;    J. Magn. Reson. 46,180 (1982)


1 ZE
2 D1 S1 DO                      ;relaxation delay
  P1:D PH4                      ;90 deg H-1 pulse
  D2                            ;1/(4J)CH
  (P2 PH5):D (P4 PH6)           ;180 deg H-1 and C-13 pulse
  D2
  (P1 PH7):D (P3 PH1)           ;90 deg H-1 and C-13 pulse
  D3 S2                         ;1/(3J)CH
  D6 CPD                        ;D6=D7-D3
  P4 PH2                        ;180 deg C-13 pulse
  D7                            ;1/(4J)CC
  P3 PH9                        ;90 deg C-13 pulse
  D5                            ;10 usec
  P3 PH3                        ;90 deg C-13 pulse
  GO=2 PH8
  WR #1
  D2 DO
EXIT

PH1= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3
PH2= 0 2 0 2 0 2 0 2 0 2 0 2 0 2 0 2
     1 3 1 3 1 3 1 3 1 3 1 3 1 3 1 3
     2 0 2 0 2 0 2 0 2 0 2 0 2 0 2 0
     3 1 3 1 3 1 3 1 3 1 3 1 3 1 3 1
PH3= 0 0 2 2 3 3 1 1 2 2 0 0 1 1 3 3
     2 2 0 0 1 1 3 3 0 0 2 2 3 3 1 1
PH4= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH5= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH7= 1
PH9= 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
```

```
PH8=R0 R0 R2 R2 R1 R1 R3 R3


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(4J)CH
;P3,P4 : 90, 180 deg C-13 pulse
;S2 : decoupler power level for CPD
;P9 : 90 deg H-1 pulse at power level S2
;D3 : 1/(3J)CH for all multiplicities (like in INEPTRD.AU)
;D6 : D7 - D3
;D7 : 1/(4J)CC
;D5 = 10 usec
;NS : 64 * n , 256 for full phase cycle
;DS : 8
```

INEPREL1.AU

; C-Relayed H,C-INEPT - For Correlation Of Quaternary Carbons With Neighboured Protonated Carbons Via Carbon-Carbon Double Quantum Coherence - No Suppression Of Signals Of Protonated Carbons - For 2D Version See HCCCOSY.AU

;H. Kessler, W. Bermel & C.Griesinger, J. Magn. Reson. 62, 573 (1985)

```
1 ZE
2 D1 S1 DO               ;relaxation delay
3 P1:D PH1               ;90 deg H-1 pulse
4 D2                     ;1/(4J)CH
5 (P2 PH2):D (P4 PH4)    ;180 deg H-1 and C-13 pulse
6 D2                     ;1/(4J)CH
7 (P1 PH3):D (P3 PH5)    ;90 deg H-1 and C-13 pulse
8 D3 S2                  ;1/(3J)CH
9 D6 CPD                 ;D3+D6=D7  , D7=1/4J(CC)
10 P4 PH6                ;180 deg C-13 pulse
11 D7                    ;1/(4J)CC
12 P3 PH7                ;90 deg C-13 pulse
13 GO=2 PH8
14 WR #1
15 D2 DO
16 EXIT

PH1= 0 2
PH2= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH3= 1 3 3 1 3 1 1 3
PH4= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH5= 0 0 0 0 2 2 2 2
PH6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH7= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH8= R0 R0 R2 R2


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;P3,P4: 90, 180 deg C-13 pulse
;D2 : 1/(4J)CH
;D3 : 1/(3J)CH for all multiplicities (like in INEPTRD.AU)
;S2 : decoupler power level for CPD
;P9 : 90 deg H-1 pulse at power level S2
;D6 : D7 - D3
;D7 : 1/(4J)CC
;NS : 128 * n
;DS : 4
```

INEPREL2.AU

; C-Relayed H,C-INEPT - For Correlation Of Quaternary Carbons With Neighboured Protonated Carbons Via Carbon-Carbon Double Quantum Coherence -

With Suppression Of Signals Of Protonated Carbons - For 2D Version See HCCCOSY.AU

```
;H. Kessler, W. Bermel & C.Griesinger, J. Magn. Reson. 62, 573 (1985)

1 ZE
2 D1 S1 DO                ;relaxation delay
3 P1:D PH1                ;90 deg H-1 pulse
4 D2                      ;1/(4J)CH
5 (P2 PH2):D (P4 PH4)     ;180 deg H-1 and C-13 pulse
6 D2                      ;1/(4J)CH
7 (P1 PH3):D (P3 PH5)     ;90 deg H-1 and C-13 pulse
8 D3 S2                   ;1/(3J)CH
9 D6 CPD                  ;D3+D6=D7  , D7=1/4J(CC)
10 P4 PH6                 ;180 deg C-13 pulse
11 D7                     ;1/(4J)CC
12 P3 PH7                 ;90 deg C-13 pulse
13 D5 DO                  ;1/(2J)CH    ;to suppress signals of protonated
 carbons
14 GO=2 PH8 CPD
15 WR #1
16 D2 DO
17 EXIT

PH1= 0 2
PH2= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH3= 1 3 3 1 3 1 1 3
PH4= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH5= 0 0 0 0 2 2 2 2
PH6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH7= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
     2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH8= R0 R0 R2 R2

;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;P3,P4: 90, 180 deg C-13 pulse
;D2 : 1/(4J)CH
;D3 : 1/(3J)CH for all multiplicities (like in INEPTRD.AU)
;S2 : decoupler power level for CPD
;P9 : 90 deg H-1 pulse at power level S2
;D6 : D7 - D3
;D7 : 1/(4J)CC
;D5 : 1/(2J)CH
;NS : 128 * n
;DS : 4
```

INEPT.AUR

; INEPT For Non-Selective Polarization Transfer From 1H To X-Nuclei Via J(XH). Basic Sequence For Coupled Spectra. This Is The Shortest Polarization Transfer Sequence And Is Recommended When T2 Relaxation Times Are Short.

```
;  1H: D1 - 90 - D2 - 180 - D2 - 90
;   X:                180        90 - FID
```

```
1 ZE
2 D1 S1 DO       ;RELAXATION DELAY FOR 1H, PREPARE DEC. POWER
                 ;FOR PULSING
3 (P1 PH1 D2):D          ;90 DEG PULSE FOR 1H, THEN DELAY
                         ;FOR EVOLUTION OF SHIFTS AND COUPLINGS
4 (P2 PH2 D2):D  P4 PH4 ;SIMULTANEOUS 180 DEG PULSES TO 1H AND X,
                         ;TO REFOCUS SHIFTS BUT COUPLING EVOLVES
                         ;FURTHER TO GIVE ANTI-PARALLEL 1H
                         ;DOUBLETS.
5 P1:D PH3  P3 PH5       ;90 DEG 1H PULSE (90 DEG PHASE SHIFT)
                         ;CAUSES POLARIZATION TRANSFER, 90 DEG X
                         ;PULSE GENERATES DETECTABLE X,Y-MAGN.
6 GO=2 PH6       ;ACQUIRE X-NUCLEUS FID WITHOUT DECOUPLING,
                 ;NO NET Z-MAGN.
7 EXIT

PH1=B0 B0 B0 B0 B0 B0 B0 B0
    B2 B2 B2 B2 B2 B2 B2 B2
PH2=B0 B2
PH3=B1 B1 B3 B3
PH4=A0 A2
PH5=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH6=R0 R0 R2 R2 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;S1 = 0H NORMALLY FOR MAX. PULSE POWER
;P1,P2 = 90,180 FOR 1H DECOUPLER
;P3,P4 = 90, 180 FOR X-NUCLEUS
;D2 = 0.25/J(XH) FOR MAX. TRANSVERSE POLARIZATION OF 1H

;NS=4*N
;RD=PW=0
```

INEPTP.AUR

```
; INEPT+ For Non-Selective Polarization Transfer From 1H To X-Nuclei Via
  J(XH).  Extended Sequence For Elimination Of Multiplet Anomalies In Coupled
  Spectra.
; O.W.SORENSEN & R.R.ERNST, J.MAGN.RES., 51, 477 (83)

;  1H: D1 - 90 - D2 - 180 - D2 - 90 - D3 - 180 - D3 - 90 -
;   X:                180        90        180           -FID

1 ZE
2 D1 S1 DO      ;RELAXATION DELAY FOR 1H, PREPARE DEC. POWER
                ;FOR PULSING
3 (P1 PH1 D2):D         ;90 DEG PULSE FOR 1H, THEN DELAY
                        ;FOR EVOLUTION OF SHIFTS AND COUPLINGS
4 (P2 PH2 D2):D  P4 PH4 ;SIMULTANEOUS 180 DEG PULSES TO 1H AND X,
                        ;TO REFOCUS SHIFTS BUT COUPLING EVOLVES
                        ;FURTHER TO GIVE ANTI-PARALLEL 1H
                        ;DOUBLETS.
5 (P1 PH3):D  (P3 PH5 D3) ;90 DEG 1H PULSE (90 DEG PHASE SHIFT)
                        ;CAUSES POLARIZATION TRANSFER, 90 DEG X
                        ;PULSE GENERATES DETECTABLE X,Y-MAGN.
                        ;DELAY ALLOWS EVOLUTION OF ANTIPHASE
                        ;X MULTIPLETS.
6 (P2 PH2):D (P4 PH6 D3)  ;180 DEG PULSES FOR 1H AND X TO
                            ;REFOCUS SHIFTS
  P1:D PH1      ;90 DEG PURGE PULSE
7 GO=2 PH7      ;ACQUIRE X-NUCLEUS FID WITHOUT DECOUPLING,
                ;SIGNAL PHASE AND INTENSITY DEPENDS ON J(XH),
                ;CHOICE OF D3, AND X-H MULTIPLICITY.
8 D2 DO
9 EXIT          ;EXIT WITH DEC. OFF

PH1=B0 B0 B0 B0 B0 B0 B0 B0
    B2 B2 B2 B2 B2 B2 B2 B2
PH2=B0 B2
PH3=B1 B1 B3 B3
PH4=A0 A2
PH5=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH6=A0 A2 A0 A2 A1 A3 A1 A3
PH7=R0 R0 R2 R2 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;S1 = 0H NORMALLY FOR MAX. PULSE POWER
;P1,P2 = 90,180 FOR 1H DECOUPLER
;P3,P4 = 90, 180 FOR X-NUCLEUS
;D2 = 0.25/J(XH) FOR MAX. TRANSVERSE POLARIZATION OF 1H

;D3 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. D3 = .125/J  GIVES XH, XH2, XH3 POSITIVE
;           = .25/J   GIVES XH ONLY
;           = .375/J  GIVES XH, XH3 POS. AND XH2 NEG.

;NS=4*N, RD=PW=0
```

INEPTRD.AUR

; INEPT For Non-Selective Polarization Transfer From 1H To X-Nuclei Via J(XH).
With Refocussing For Decoupled Spectra

```
;  1H: D1 - 90 - D2 - 180 - D2 - 90 - D3 - 180 - D3 - BB
;   X:               180        90        180        FID

1 ZE
2 D1 S1 DO      ;RELAXATION DELAY FOR 1H, PREPARE DEC. POWER
                ;FOR PULSING
3 (P1 PH1 D2):D         ;90 DEG PULSE FOR 1H, THEN DELAY
                        ;FOR EVOLUTION OF SHIFTS AND COUPLINGS
4 (P2 PH2 D2):D  P4 PH4 ;SIMULTANEOUS 180 DEG PULSES TO 1H AND X,
                        ;TO REFOCUS SHIFTS BUT COUPLING EVOLVES
                        ;FURTHER TO GIVE ANTI-PARALLEL 1H
                        ;DOUBLETS.
5 (P1 PH3):D  (P3 PH5 D3)   ;90 DEG 1H PULSE (90 DEG PHASE SHIFT)
                        ;CAUSES POLARIZATION TRANSFER, 90 DEG X
                        ;PULSE GENERATES DETECTABLE X,Y-MAGN.
                        ;DELAY ALLOWS EVOLUTION OF ANTIPHASE
                        ;X MULTIPLETS.
6 (P2 PH2):D  (P4 PH6 D3 S2)  ;180 DEG PULSES FOR 1H AND X TO
                              ;REFOCUS SHIFTS, SET DEC. POWER

7 GO=2 PH7 CPD          ;ACQUIRE X-NUCLEUS FID WITH DECOUPLING,
                        ;SIGNAL PHASE AND INTENSITY DEPENDS ON J(XH),
                        ;CHOICE OF D3, AND X-H MULTIPLICITY.
8 D2 DO
9 EXIT                  ;EXIT WITH DEC. OFF

PH1=B0 B0 B0 B0 B0 B0 B0 B0
    B2 B2 B2 B2 B2 B2 B2 B2
PH2=B0 B2
PH3=B1 B1 B3 B3
PH4=A0 A2
PH5=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH6=A0 A2 A0 A2 A1 A3 A1 A3
PH7=R0 R0 R2 R2 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;S1 = 0H NORMALLY FOR MAX. PULSE POWER
;S2 = POWER SETTING FOR GOOD DECOUPLING WITH CPD
;P1,P2 = 90,180 FOR 1H DECOUPLER
;P3,P4 = 90, 180 FOR X-NUCLEUS
;P9 = 90 DEG DEC. PULSE FOR POWER S2
;D2 = 0.25/J(XH) FOR MAX. TRANSVERSE POLARIZATION OF 1H

;D3 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. D3 = .125/J  GIVES XH, XH2, XH3 POSITIVE
;          = .25/J   GIVES XH ONLY
;          = .375/J  GIVES XH, XH3 POS. AND XH2 NEG.

;NS=4*N, RD=PW=0
```

INVD1D9.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence In
Inverse Mode. Decoupling During Acquisition Using GARP1 (see also INVH1.AU)

```
;this requires BSV7 with fast TLO (RCP7) and additional
;   amplifier (e.g. BFX5) gated with RCP3
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)
```

```
  II
1 ZE
2 D1 S1 DO                      ;relaxation delay
  P1:D PH1                      ;90 deg H-1 pulse
  D2                            ;1/(2J)XH
  P3 PH3                        ;90 deg X pulse
  D0                            ;t1/2
  P2:D PH2                      ;180 deg H-1 pulse
  D0                            ;t1/2
  P3 PH4                        ;90 deg X pulse
  D2                            ;1/(2J)XH refocussing delay
  D5                            ;DE/2
3 D5:D PH8
4 D6 ADC                        ;D6 = 2 usec
  P0      :C7
5 P8*0.339:C7:T:C3 PH10         ;GARP1 decoupling
  P8*0.613:C7:T:C3 PH12
  P8*2.864:C7:T:C3 PH10
  P8*2.981:C7:T:C3 PH12
  P8*0.770:C7:T:C3 PH10
  P8*0.691:C7:T:C3 PH12
  P8*0.944:C7:T:C3 PH10
  P8*1.020:C7:T:C3 PH12
  P8*1.494:C7:T:C3 PH10
  P8*2.846:C7:T:C3 PH12
  P8*0.738:C7:T:C3 PH10
  P8*0.510:C7:T:C3 PH12
  P8*0.283:C7:T:C3 PH10
  P8*0.808:C7:T:C3 PH12
  P8*1.328:C7:T:C3 PH10
  P8*1.536:C7:T:C3 PH12
  P8*2.871:C7:T:C3 PH10
  P8*0.721:C7:T:C3 PH12
  P8*0.788:C7:T:C3 PH10
  P8*0.858:C7:T:C3 PH12
  P8*1.091:C7:T:C3 PH10
  P8*1.484:C7:T:C3 PH12
  P8*2.843:C7:T:C3 PH10
  P8*0.729:C7:T:C3 PH12
  P8*0.593:C7:T:C3 PH10
  L0 TO 5 TIMES 2
6 P8*0.339:C7:T:C3 PH12
  P8*0.613:C7:T:C3 PH10
  P8*2.864:C7:T:C3 PH12
  P8*2.981:C7:T:C3 PH10
  P8*0.770:C7:T:C3 PH12
  P8*0.691:C7:T:C3 PH10
  P8*0.944:C7:T:C3 PH12
  P8*1.020:C7:T:C3 PH10
  P8*1.494:C7:T:C3 PH12
  P8*2.846:C7:T:C3 PH10
  P8*0.738:C7:T:C3 PH12
  P8*0.510:C7:T:C3 PH10
  P8*0.283:C7:T:C3 PH12
  P8*0.808:C7:T:C3 PH10
  P8*1.328:C7:T:C3 PH12
  P8*1.536:C7:T:C3 PH10
  P8*2.871:C7:T:C3 PH12
  P8*0.721:C7:T:C3 PH10
  P8*0.788:C7:T:C3 PH12
  P8*0.858:C7:T:C3 PH10
  P8*1.091:C7:T:C3 PH12
  P8*1.484:C7:T:C3 PH10
  P8*2.843:C7:T:C3 PH12
  P8*0.729:C7:T:C3 PH10
  P8*0.593:C7:T:C3 PH12
```

```
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
  P0      :C7
8 RCYC=2 PH5
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH2=0
PH3=0 1 2 3
PH4=0 0 0 0 2 2 2 2
PH5=R0 R3 R2 R1 R2 R1 R0 R3

PH8=0

PH10=0
PH12=2


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;D5 : DE/2
;D6 = 2 usec
;P0 = 5 usec
;P8 : 90 deg pulse for X decoupling
;L2 : L2 * 31.75 * 4 * P9 => AQ
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0 = 2
```

INVD1DP3.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence In Inverse Mode. Phase Sensitive Using TPPI. Decoupling During Acquisition Using GARP1 (see also INVH1.AU).

```
;this requires BSV7 with fast TLO (RCP7) and additional
;   amplifier (e.g. BFX5) gated with RCP3
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)


  II
1 ZE
2 D1 S1 DO                       ;relaxation delay
  P1:D PH1                       ;90 deg H-1 pulse
  D2                             ;1/(2J)XH
  P3 PH3                         ;90 deg X pulse
  D0                             ;t1/2
  P2:D PH2                       ;180 deg H-1 pulse
  D0                             ;t1/2
  P3 PH4                         ;90 deg X pulse
  D2                             ;1/(2J)XH refocussing delay
  D5                             ;DE/2
3 D5:D PH8
4 D6 ADC                         ;D6 = 2 usec
  P0 :C7
5 P7 :C7:T:C3 PH10               ;GARP1 decoupling
  P10:C7:T:C3 PH12
  P28:C7:T:C3 PH10
  P30:C7:T:C3 PH12
  P15:C7:T:C3 PH10
  P11:C7:T:C3 PH12
  P19:C7:T:C3 PH10
  P20:C7:T:C3 PH12
  P24:C7:T:C3 PH10
  P27:C7:T:C3 PH12
  P14:C7:T:C3 PH10
  P8 :C7:T:C3 PH12
  P6 :C7:T:C3 PH10
  P17:C7:T:C3 PH12
  P22:C7:T:C3 PH10
  P25:C7:T:C3 PH12
  P29:C7:T:C3 PH10
  P12:C7:T:C3 PH12
  P16:C7:T:C3 PH10
  P18:C7:T:C3 PH12
  P21:C7:T:C3 PH10
  P23:C7:T:C3 PH12
  P26:C7:T:C3 PH10
  P13:C7:T:C3 PH12
  P9 :C7:T:C3 PH10
  L0 TO 5 TIMES 2
6 P7 :C7:T:C3 PH12
  P10:C7:T:C3 PH10
  P28:C7:T:C3 PH12
  P30:C7:T:C3 PH10
  P15:C7:T:C3 PH12
  P11:C7:T:C3 PH10
  P19:C7:T:C3 PH12
  P20:C7:T:C3 PH10
  P24:C7:T:C3 PH12
  P27:C7:T:C3 PH10
  P14:C7:T:C3 PH12
  P8 :C7:T:C3 PH10
  P6 :C7:T:C3 PH12
  P17:C7:T:C3 PH10
  P22:C7:T:C3 PH12
```

```
  P25:C7:T:C3 PH10
  P29:C7:T:C3 PH12
  P12:C7:T:C3 PH10
  P16:C7:T:C3 PH12
  P18:C7:T:C3 PH10
  P21:C7:T:C3 PH12
  P23:C7:T:C3 PH10
  P26:C7:T:C3 PH12
  P13:C7:T:C3 PH10
  P9 :C7:T:C3 PH12
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
  P0 :C7
8 RCYC=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0

PH8=0

PH10=0
PH12=2


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;D5 : DE/2
;D6 = 2 usec
;P0 = 5 usec
;P9 : 90 deg pulse for X decoupling
;L2 : L2 * 31.75 * 4 * P9 => AQ
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) = (1/2) DW(X)
;ND0 = 4
;MC2 = W
```

INVD1DP9.AUR

```
; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence In
  Inverse Mode.  Phase Sensitive Using TPPI.  Decoupling During Acquisition
  Using GARP1 (see also INVH1.AU).
;this requires BSV7 with fast TLO (RCP7) and additional
;   amplifier (e.g. BFX5) gated with RCP3
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)


  II
1 ZE
2 D1 S1 DO                          ;relaxation delay
  P1:D PH1                          ;90 deg H-1 pulse
  D2                                ;1/(2J)XH
  P3 PH3                            ;90 deg X pulse
  D0                                ;t1/2
  P2:D PH2                          ;180 deg H-1 pulse
  D0                                ;t1/2
  P3 PH4                            ;90 deg X pulse
  D2                                ;1/(2J)XH refocussing delay
  D5                                ;DE/2
3 D5:D PH8
4 D6 ADC                            ;D6 = 2 usec
  P0      :C7
5 P8*0.339:C7:T:C3 PH10             ;GARP1 decoupling
  P8*0.613:C7:T:C3 PH12
  P8*2.864:C7:T:C3 PH10
  P8*2.981:C7:T:C3 PH12
  P8*0.770:C7:T:C3 PH10
  P8*0.691:C7:T:C3 PH12
  P8*0.944:C7:T:C3 PH10
  P8*1.020:C7:T:C3 PH12
  P8*1.494:C7:T:C3 PH10
  P8*2.846:C7:T:C3 PH12
  P8*0.738:C7:T:C3 PH10
  P8*0.510:C7:T:C3 PH12
  P8*0.283:C7:T:C3 PH10
  P8*0.808:C7:T:C3 PH12
  P8*1.328:C7:T:C3 PH10
  P8*1.536:C7:T:C3 PH12
  P8*2.871:C7:T:C3 PH10
  P8*0.721:C7:T:C3 PH12
  P8*0.788:C7:T:C3 PH10
  P8*0.858:C7:T:C3 PH12
  P8*1.091:C7:T:C3 PH10
  P8*1.484:C7:T:C3 PH12
  P8*2.843:C7:T:C3 PH10
  P8*0.729:C7:T:C3 PH12
  P8*0.593:C7:T:C3 PH10
  L0 TO 5 TIMES 2
6 P8*0.339:C7:T:C3 PH12
  P8*0.613:C7:T:C3 PH10
  P8*2.864:C7:T:C3 PH12
  P8*2.981:C7:T:C3 PH10
  P8*0.770:C7:T:C3 PH12
  P8*0.691:C7:T:C3 PH10
  P8*0.944:C7:T:C3 PH12
  P8*1.020:C7:T:C3 PH10
  P8*1.494:C7:T:C3 PH12
  P8*2.846:C7:T:C3 PH10
  P8*0.738:C7:T:C3 PH12
  P8*0.510:C7:T:C3 PH10
  P8*0.283:C7:T:C3 PH12
  P8*0.808:C7:T:C3 PH10
  P8*1.328:C7:T:C3 PH12
```

```
  P8*1.536:C7:T:C3 PH10
  P8*2.871:C7:T:C3 PH12
  P8*0.721:C7:T:C3 PH10
  P8*0.788:C7:T:C3 PH12
  P8*0.858:C7:T:C3 PH10
  P8*1.091:C7:T:C3 PH12
  P8*1.484:C7:T:C3 PH10
  P8*2.843:C7:T:C3 PH12
  P8*0.729:C7:T:C3 PH10
  P8*0.593:C7:T:C3 PH12
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
  P0      :C7
8 RCYC=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0

PH8=0

PH10=0
PH12=2


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;D5 : DE/2
;D6 = 2 usec
;P0 = 5 usec
;P8 : 90 deg pulse for X decoupling
;L2 : L2 * 31.75 * 4 * P9 => AQ
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) = (1/2) DW(X)
;ND0 = 4
;MC2 = W
```

INVD1DPS.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence In Inverse Mode. Phase Sensitive Using TPPI. Decoupling During Acquisition Using GARP1 (see also INVH1.AU).

```
;this requires BSV7 with fast TLO (RCP7) and additional amplifier (e.g. BFX5)
 gated with RCP3;  with water suppression using presaturation this requires
 fast switching decoupler
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)

  II
1 ZE
2 D8:D PH8
  D8 S3 DO
  D1 HG                             ;relaxation delay
  D8 DO S1
  P1:D PH1                          ;90 deg H-1 pulse
  D2                                ;1/(2J)XH
  P3 PH3                            ;90 deg X pulse
  D0                                ;t1/2
  P2:D PH2                          ;180 deg H-1 pulse
  D0                                ;t1/2
  P3 PH4                            ;90 deg X pulse
  D2                                ;1/(2J)XH refocussing delay
  D5                                ;DE/2
3 D5:D PH8
4 D6 ADC                            ;D6 = 2 usec
  P0      :C7
5 P8*0.339:C7:T:C3 PH10             ;GARP1 decoupling
  P8*0.613:C7:T:C3 PH12
  P8*2.864:C7:T:C3 PH10
  P8*2.981:C7:T:C3 PH12
  P8*0.770:C7:T:C3 PH10
  P8*0.691:C7:T:C3 PH12
  P8*0.944:C7:T:C3 PH10
  P8*1.020:C7:T:C3 PH12
  P8*1.494:C7:T:C3 PH10
  P8*2.846:C7:T:C3 PH12
  P8*0.738:C7:T:C3 PH10
  P8*0.510:C7:T:C3 PH12
  P8*0.283:C7:T:C3 PH10
  P8*0.808:C7:T:C3 PH12
  P8*1.328:C7:T:C3 PH10
  P8*1.536:C7:T:C3 PH12
  P8*2.871:C7:T:C3 PH10
  P8*0.721:C7:T:C3 PH12
  P8*0.788:C7:T:C3 PH10
  P8*0.858:C7:T:C3 PH12
  P8*1.091:C7:T:C3 PH10
  P8*1.484:C7:T:C3 PH12
  P8*2.843:C7:T:C3 PH10
  P8*0.729:C7:T:C3 PH12
  P8*0.593:C7:T:C3 PH10
  L0 TO 5 TIMES 2
6 P8*0.339:C7:T:C3 PH12
  P8*0.613:C7:T:C3 PH10
  P8*2.864:C7:T:C3 PH12
  P8*2.981:C7:T:C3 PH10
  P8*0.770:C7:T:C3 PH12
  P8*0.691:C7:T:C3 PH10
  P8*0.944:C7:T:C3 PH12
  P8*1.020:C7:T:C3 PH10
  P8*1.494:C7:T:C3 PH12
  P8*2.846:C7:T:C3 PH10
  P8*0.738:C7:T:C3 PH12
  P8*0.510:C7:T:C3 PH10
```

```
  P8*0.283:C7:T:C3 PH12
  P8*0.808:C7:T:C3 PH10
  P8*1.328:C7:T:C3 PH12
  P8*1.536:C7:T:C3 PH10
  P8*2.871:C7:T:C3 PH12
  P8*0.721:C7:T:C3 PH10
  P8*0.788:C7:T:C3 PH12
  P8*0.858:C7:T:C3 PH10
  P8*1.091:C7:T:C3 PH12
  P8*1.484:C7:T:C3 PH10
  P8*2.843:C7:T:C3 PH12
  P8*0.729:C7:T:C3 PH10
  P8*0.593:C7:T:C3 PH12
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
  P0      :C7
8 RCYC=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0
PH8=0
PH10=0
PH12=2

;D8 : > 400 usec
;S3 : power level for presaturation
;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;D5 : DE/2
;D6 = 2 usec
;P0 = 5 usec
;P8 : 90 deg pulse for X decoupling
;L2 : L2 * 31.75 * 4 * P9 => AQ
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) = (1/2) DW(X)
;ND0 = 4
;MC2 = W
```

INVD1MLP.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using MLEV17 For Homonuclear Hartman-Hahn. Mixing In Inverse Mode. Phase Sensitive Using TPPI. No Decoupling During Acquisition (see also INVH1.AU)

```
;L. Lerner and A. Bax, J, Magn. Reson. 69, 375-380 (1986)


  II
1 ZE
2 D1 S1 DO                         ;relaxation delay
3 P1:D PH1                         ;90 deg H-1 pulse
4 D2                               ;1/(2J)XH
5 P3 PH3                           ;90 deg X pulse
  D0                               ;t1/2
```

```
6 P2:D PH2
  D0
7 P3 PH4
  D2 S2 DO                         ;1/(2J)XH refocussing delay
                                   ;switch decoupler power for spinlock
  (P7 PH16):D                      ;trim pulse
8 (P8 PH12 P9 PH13 P8 PH12):D      ;MLEV y-spinlock
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P9 PH13):D                      ;180 deg H-1 pulse
  L6 TO 8 TIMES UPR
  (P7 PH16):D                      ;trim pulse
9 GO=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT


PH1=0
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0

PH12=0
PH13=1
PH14=2
PH15=3
PH16=1



;D1 * 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse at power level S1
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;S2 : power level for spinlock (90 deg pulse 25 - 35 usec)
;P7 = 2.5 msec
;P8,P9 : 90, 180 deg H-1 pulse at power level S2
;L6 : spinlock time = (L6 * 66 * P8) + (2 * P7)
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
;MC2 = W
```

INVD1MPS.AUR

```
; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence
  Using MLEV17 For Homonuclear Hartman-Hahn.  Mixing In Inverse Mode.  Phase
  Sensitive Using TPPI.  No Decoupling During Acquisition (see also INVH1.AU)
;Water Suppression Using Presaturation
;This Requires Fast Switching Decoupler


  II
1 ZE
2 D8:D PH8
  D8 S3 DO
  D1 HG                            ;relaxation delay
  D8 DO S1
3 P1:D PH1                         ;90 deg H-1 pulse
4 D2                               ;1/(2J)XH
5 P3 PH3                           ;90 deg X pulse
  D0                               ;t1/2
6 P2:D PH2
  D0
7 P3 PH4
  D2 S2 DO                         ;1/(2J)XH refocussing delay
                                   ;switch decoupler power for spinlock
  (P7 PH16):D                      ;trim pulse
8 (P8 PH12 P9 PH13 P8 PH12):D      ;MLEV y-spinlock
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P9 PH13):D                      ;180 deg H-1 pulse
  L6 TO 8 TIMES UPR
  (P7 PH16):D                      ;trim pulse
9 GO=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT


PH1=0
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0

PH8=0

PH12=0
PH13=1
PH14=2
PH15=3
PH16=1
```

```
;D8 : > 400 usec
;S3 : power level for presaturation
;D1 * 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse at power level S1
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;S2 : power level for spinlock (90 deg pulse 25 - 35 usec)
;P7 = 2.5 msec
;P8,P9 : 90, 180 deg H-1 pulse at power level S2
;L6 : spinlock time = (L6 * 66 * P8) + (2 * P7)
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
;MC2 = W
```

INVD1PH.AU

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Inverse Mode. No Decoupling During Acquisition. Phase Sensitive Using TPPI (see also INVH1.AU)

```
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)


1 ZE
2 D1 S1 DO                    ;relaxation delay
3 P1:D PH1                    ;90 deg H-1 pulse
4 D2                          ;1/(2J)XH
5 P3 PH3                      ;90 deg X pulse
  D0                          ;t1/2
6 P2:D PH2                    ;180 deg H-1 pulse
  D0                          ;t1/2
7 P3 PH4                      ;90 deg X pulse
  D2                          ;1/(2J)XH refocussing delay
9 GO=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH2=0
PH3=0 2 0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
;MC2= W
```

INVD2D.AU

```
; 2D H-1/X Correlation Via Heteronuclear Zero Quantum Coherence Using Inverse
  Mode.  No Decoupling During Acquisition (see also INVH1.AU).  Gives
  Splitting In F1 With Respect To Passive XH Coupling.
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)


1 ZE
2 D1 S1 DO                      ;relaxation delay
3 P1:D PH1                      ;90 deg H-1 pulse
4 D2                            ;1/(2J)XH
5 P3 PH2                        ;90 deg X pulse
6 D0                            ;t1
7 P3 PH3                        ;90 deg X pulse
8 GO=2 PH4
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH2=0 1 2 3
PH3=0 0 0 0 2 2 2 2
PH4=R0 R3 R2 R1 R2 R1 R0 R3


;D1 : 1-5 T1
;S1 = 0H
;P1 : 90 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN = 1 / (SW(X) + SW(H))
;ND0= 1
```

INVD2D2.AU

; 2D H-1/X Correlation Via Heteronuclear Double Quantum Coherence Using Inverse Mode. No Decoupling During Acquisition (see also INVH1.AU). Gives Splitting In F1 With Respect To Passive XH Coupling.

;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson 55, 301 (1983)

```
1 ZE
2 D1 S1 DO                      ;relaxation delay
3 P1:D PH1                      ;90 deg H-1 pulse
4 D2                            ;1/(2J)XH
5 P3 PH2                        ;90 deg X pulse
6 D0                            ;t1
7 P3 PH3                        ;90 deg X pulse
8 GO=2 PH4
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH2=0 1 2 3
PH3=0 0 0 0 2 2 2 2
PH4=R0 R1 R2 R3 R2 R3 R0 R1


;D1 : 1-5 T1
;S1 = 0H
;P1 : 90 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN = 1 / (SW(X) + SW(H))
;ND0= 1
```

INVDEPT.AU

; Inverse DEPT Using Inverse Mode No Decoupling During Acquisition

```
;  H-1:                  P0        180        FID
;    X:  D1 - 90 - D2 - 180 - D2 -  90 - D2 -

;(see also INVH1.AU)
;M.R. Bendall, D.T. Pegg, D.M. Doddrell & J. Field, J. Magn. Reson.
;   51, 520 - 526 (1983)

1 ZE
2 D1 S1  DO                        ;relaxation delay
3 (P3 PH1)                         ;90 deg X pulse
  D2                               ;1/(2J)XH
4 (P4 PH2)  (P0 PH4):D             ;180 deg X pulse, 90 deg H-1 pulse
  D2
5 (P3 PH3)  (P2 PH5):D             ;90 deg X pulse, 180 deg H-1 pulse
  D2
6 GO=2 PH7
7 EXIT

PH1=0
PH2=0 2 1 3
PH3=1 1 1 1 3 3 3 3
PH4=0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1
    2 2 2 2 2 2 2 2 3 3 3 3 3 3 3 3
PH5=0 2 0 2 0 2 0 2 1 3 1 3 1 3 1 3
PH7=R0 R0 R2 R2 R2 R2 R0 R0 R1 R1 R3 R3 R3 R3 R1 R1
    R2 R2 R0 R0 R0 R0 R2 R2 R3 R3 R1 R1 R1 R1 R3 R3


;D1 : 1-5 * T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;P3,P4 : 90, 180 deg X pulse
;P0 : variable pulse depending on multiplicity selection
;        45 deg    XH, XH2, XH3    all positive
;        90 deg    XH              only
;       135 deg    XH, XH3         positive
;                  XH2             negative
;D2 : 1/(2J)XH
;DS = 2 or 4
;NS = 32 * n
```

INVDQ.AU

; 1D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Inverse Mode. No Decoupling During Acquisition. Giving Anti-Phase Signals (see also INVH1.AU)

```
1 ZE
2 D1 S1 DO              ;relaxation delay
3 P1:D PH1              ;90 deg H-1 pulse
4 D2                    ;1/(2J)XH
5 P3 PH2                ;90 deg X pulse
6 D3
9 P3 PH3                ;90 deg X pulse
10 GO=2 PH4
14 EXIT

PH1=0 0 0 0 2 2 2 2
PH2=0 2
PH3=0 0 2 2
PH4=R0 R2 R2 R0 R2 R0 R0 R2


;D1 : 1-5 T1
;S1 = 0H
;P1 : 90 deg H-1 pulse
;P3 : 90 deg X pulse
;D2 : 1/(2J)XH
;D3 = 3 usec
;DS : 2 or 4
;NS : 8 * n
```

INVDQR.AU

; 1D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Inverse Mode. No Decoupling During Acquisition. Giving In-Phase Signals (see also INVH1.AU)

```
1 ZE
2 D1 S1 DO                  ;relaxation delay
3 P1:D PH1                  ;90 deg H-1 pulse
4 D2                        ;1/(2J)XH
5 P3 PH3                    ;90 deg X pulse
6 P2:D PH2                  ;180 deg H-1 pulse
7 P3 PH4                    ;90 deg X pulse
8 D2
9 GO=2 PH5
  EXIT

PH1=0 0 0 0 1 1 1 1
PH2=0 2 0 2 1 3 1 3
    1 3 1 3 2 0 2 0
PH3=0
PH4=0 0 2 2
PH5=R0 R0 R2 R2 R1 R1 R3 R3 R2 R2 R0 R0 R3 R3 R1 R1


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;DS : 2 or 4
;NS : 16 * n
```

INVDQRG.AU

; 1D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Inverse Mode. CPD Decoupling During Acquisiation Using GARP1 (see also INVH1.AU).

```
;   A.J. Shaka, P.B. Barker & R. Freeman, J. Magn. Reson. 64,
;    547 - 552 (1985)


1 ZE
2 D1 S1 DO TH                    ;relaxation delay,
                                 ;switch to high power output
  P1:D PH1                       ;90 deg H-1 pulse
  D2                             ;1/(2J)XH
  P3 PH3                         ;90 deg X pulse
  P2:D PH2                       ;180 deg H-1 pulse
  P3 PH4                         ;90 deg X pulse
  D2 TL                          ;1/(2J)XH, switch to low power output
  D5                             ;DE/2
3 D5:D PH9                       ;reset phase before acquisition
4 D6 ADC                         ;D6 = 2 USEC
5 (P7  PH10 P10 PH12 P28 PH10 P30 PH12 P15 PH10 P11 PH12)
  (P19 PH10 P20 PH12 P24 PH10 P27 PH12 P14 PH10 P8  PH12)
  (P6  PH10 P17 PH12 P22 PH10 P25 PH12 P29 PH10 P12 PH12)
  (P16 PH10 P18 PH12 P21 PH10 P23 PH12 P26 PH10 P13 PH12)
  (P9  PH10)
  L0 TO 5 TIMES 2
6 (P7  PH12 P10 PH10 P28 PH12 P30 PH10 P15 PH12 P11 PH10)
  (P19 PH12 P20 PH10 P24 PH12 P27 PH10 P14 PH12 P8  PH10)
  (P6  PH12 P17 PH10 P22 PH12 P25 PH10 P29 PH12 P12 PH10)
  (P16 PH12 P18 PH10 P21 PH12 P23 PH10 P26 PH12 P13 PH10)
  (P9  PH12)
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
8 RCYC=2 PH5                     ;loop for NS scans
  D2 DO TH                       ;switch to high power output
EXIT

PH1=0 0 0 0 1 1 1 1
PH2=0 2 0 2 1 3 1 3 3 1 3 1 2 0 2 0
PH3=0
PH4=0 0 2 2
PH5=R0 R0 R2 R2 R1 R1 R3 R3 R2 R2 R0 R0 R3 R3 R1 R1

PH9=0

PH10=0
PH12=2


;P6 = 25.5 degree pulse at low power output
;P7 = 30.5                        P19= 85.0
;P8 = 45.9                        P20= 91.8
;P9 = 53.4                        P21= 98.2
;P10= 55.2                        P22=119.5
;P11= 62.2                        P23=133.6
;P12= 64.9                        P24=134.5
;P13= 65.6                        P25=138.2
;P14= 66.4                        P26=255.9
;P15= 69.3                        P27=256.1
;P16= 70.9                        P28=257.8
;P17= 72.7                        P29=258.4
;P18= 77.2                        P30=268.3
;L2=31.75 * 4 * (90 degree pulse length) => AQ

;D1 : 1-5 T1
```

```
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;D5 : DE/2
;D6 = 2 usec
;DS : 8
;NS : 16 * n
```

INVDR2D.AU

```
; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence
  Using Inverse Mode.  No Decoupling During Acquisition (see also INVH1.AU)
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)


1 ZE
2 D1 S1 DO                    ;relaxation delay
3 P1:D PH1                    ;90 deg H-1 pulse
4 D2                          ;1/(2J)XH
5 P3 PH3                      ;90 deg X pulse
  D0                          ;t1/2
6 P2:D PH2                    ;180 deg H-1 pulse
  D0                          ;t1/2
7 P3 PH4                      ;90 deg X pulse
9 GO=2 PH5
  WR #1
  IF #1
  IN=1
EXIT

PH1=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH2=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH3=0 1 2 3 0 1 2 3 2 3 0 1 2 3 0 1
PH4=0 0 0 0 2 2 2 2 2 2 2 2 0 0 0 0
PH5=R0 R3 R2 R1 R2 R1 R0 R3
    R2 R1 R0 R3 R0 R3 R2 R1


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0 = 2
```

INVDR2D2.AU

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Inverse Mode. No Decoupling During Acquisition (see also INVH1.AU)

```
1 ZE
2 D1 S1 DO                        ;relaxation delay
3 P1:D PH1                        ;90 deg H-1 pulse
  D3                              ;1/(4j)XH
  (P2 PH6):D (P4 PH7)             ;180 deg H-1 and X pulse
4 D3                              ;1/(4J)XH
5 (P1 PH8):D (P3 PH3)             ;90 deg H-1 and X pulse
  D0                              ;t1/2
6 P2:D PH2                        ;180 deg H-1 pulse
  D0                              ;t1/2
7 P3 PH4                          ;90 deg X pulse
9 GO=2 PH5
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH8=2
PH2=0
PH3=0 1 2 3
PH4=0 0 0 0 2 2 2 2
PH6=0
PH7=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH5=R0 R3 R2 R1 R2 R1 R0 R3


;D1 : 1-5 T1
;S1 = 0H
;P1.P2 : 90, 180 deg H-1 pulse
;D3 : 1/(4J)XH
;P3,P4 : 90, 180 deg X pulse
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0= 2
```

INVDR2DG.AU

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Cogerence Using Inverse Mode. CPD Decoupling During Acquisition Using GARP1 (see also INVH1.AU)

```
;   A.J. Shaka, P.B. Barker & R. Freeman, J. Magn. Reson. 64,
;   547 - 552 (1985)
;A. Bax, R.H. Griffey & B.L. Hawkins, J. Magn. Reson. 55, 301 (1983)


1 ZE
2 D1 S1 DO TH                       ;relaxation delay
                                    ;switch to high power output
  P1:D PH1                          ;90 deg H-1 pulse
  D2                                ;1/(2J)XH
  P3 PH3                            ;90 deg X pulse
  D0                                ;t1/2
  P2:D PH2                          ;180 deg H-1 pulse
  D0                                ;t1/2
  P3 PH4                            ;90 deg X pulse
  D2 TL                             ;1/(2J)XH, switch to low power output
  D5                                ;DE/2
3 D5:D PH9                          ;reset phase before acquisition
4 D6 ADC                            ;D6 = 2 usec
5 (P7  PH10 P10 PH12 P28 PH10 P30 PH12 P15 PH10 P11 PH12)
  (P19 PH10 P20 PH12 P24 PH10 P27 PH12 P14 PH10 P8  PH12)
  (P6  PH10 P17 PH12 P22 PH10 P25 PH12 P29 PH10 P12 PH12)
  (P16 PH10 P18 PH12 P21 PH10 P23 PH12 P26 PH10 P13 PH12)
  (P9  PH10)
  L0 TO 5 TIMES 2
6 (P7  PH12 P10 PH10 P28 PH12 P30 PH10 P15 PH12 P11 PH10)
  (P19 PH12 P20 PH10 P24 PH12 P27 PH10 P14 PH12 P8  PH10)
  (P6  PH12 P17 PH10 P22 PH12 P25 PH10 P29 PH12 P12 PH10)
  (P16 PH12 P18 PH10 P21 PH12 P23 PH10 P26 PH12 P13 PH10)
  (P9  PH12)
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
8 RCYC=2 PH5                        ;loop for NS scans
  WR #1
  IF #1
  IN=1
  D2 DO TH
EXIT

PH1=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH2=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH3=0 1 2 3 0 1 2 3 2 3 0 1 2 3 0 1
PH4=0 0 0 0 2 2 2 2 2 2 2 2 0 0 0 0
PH5=R0 R3 R2 R1 R2 R1 R0 R3
    R2 R1 R0 R3 R0 R3 R2 R1

PH9=0

PH10=0
PH12=2


;P6 = 25.5 degree pulse at low power output
;P7 = 30.5
;P8 = 45.9
;P9 = 53.4
;P10= 55.2
;P11= 62.2
;P12= 64.9
;P13= 65.6
;P14= 66.4
```

```
;P15= 69.3
;P16= 70.9
;P17= 72.7
;P18= 77.2
;P19= 85.0
;P20= 91.8
;P21= 98.2
;P22=119.5
;P23=133.6
;P24=134.5
;P25=138.2
;P26=255.9
;P27=256.1
;P28=257.8
;P29=258.4
;P30=268.3
;L2= 31.75 * 4 * (90 degree pulse length) => AQ

;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3 : 90 deg X pulse
;D5 : DE/2
;D6 = 2 usec
;DS : 8
;NS : 8 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0= 2
```

INVDR2DP.AU

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Inverse Mode. Phase Sensitive Using TPPI. No Decoupling During Acqusition (see also INVH1.AU)

```
1 ZE
2 D1 S1 DO                      ;relaxation delay
3 P1:D PH1                      ;90 deg H-1 pulse
4 D3                            ;1/(4J)XH
5 (P2 PH2):D (P4 PH5)           ;180 deg H-1 and X pulse
  D3                            ;1/(4J)XH
  (P1 PH3):D (P3 PH6)           ;90 deg H-1 and X pulse
  D0                            ;t1/2
6 P2:D PH4                      ;180 deg H-1 pulse
  D0                            ;t1/2
7 P3 PH7                        ;90 deg X pulse
9 GO=2 PH8
  WR #1
  IF #1
  IP6                           ;TPPI
  IN=1
EXIT

PH1=0
PH2=0
PH3=2
PH4=0
PH5=0 0 0 0 2 2 2 2
PH6=0 2
PH7=0 0 2 2
PH8=R0 R2 R2 R0


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D3 : 1/(4J)XH
;P3,P4 : 90, 180 deg X pulse
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) = DW(X)/2
;ND0= 4
```

INVDR2LP.AU

```
; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence
  Using Inverse Mode.  Optimized On Long Range Couplings With Low Pass J-
  Filter To Suppress One-Bond Correlations (see also INVH1.AU)
;A. Bax & M.F. Summers, J. Am. Chem. Soc 108, 2093 - 2094 (1986)


1 ZE
2 D1 S1 DO                        ;relaxation delay
3 P1:D PH1                        ;90 deg H-1 pulse
  D2                              ;1/(2J)XH
  P3 PH3                          ;90 deg X pulse
4 D4
5 P3 PH4                          ;90 deg X pulse
  D0                              ;t1/2
6 P2:D PH2                        ;180 deg H-1 pulse
  D0                              ;t1/2
7 P3 PH5                          ;90 deg X pulse
9 GO=2 PH6
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH2=0
PH3=0 0 2 2
PH4=0 2 0 2 3 1 3 1
PH5=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH6=R0 R2 R0 R2 R1 R3 R1 R3 R2 R0 R2 R0 R3 R1 R3 R1


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH  - one-bond coupling
;P3 : 90 deg X pulse
;D4 - evolution of XH long range coupling
;DS : 4
;NS : 16 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0 = 2
```

INVGATE.AU

; Inverse Gated Het.-Nuclear Decoupling 1H-Decoupled Spectrum Without NOE

```
1 ZE            ;ZERO MEMORY
2 D1 DO S2      ;POWER S2, GATED OFF FOR D1
3 GO=2 CPD      ;CPD DEC. DURING AQ. (RD=0)
4 D2 DO         ;LEAVE DEC. GATED OFF
5 EXIT

;D1 MUST BE 5-10 TIMES T1 (OR AQ) FOR NOE SUPPRESSION
;S2 = OPTIMUM POWER SETTING FOR CPD (P9)
;D2=5 MSEC
;RD=0
```

INVH1.AU

; 1D Proton Spectrum Using Inverse Mode

```
1 ZE
2 D1 S1 DO           ;relaxation delay
  P1:D PH1           ;excitation pulse
3 GO=2 PH2           ;acquisition
  D2 DO
EXIT

PH1=0 2 2 0 1 3 3 1
PH2=R0 R2 R2 R0 R1 R3 R3 R1

;PR = x 1 (x depending on X-nucleus)
;O1 - center of X-nucleus spectrum
;O2 - center of proton spectrum
;SW - spectral width of proton spectrum
;QN, OP for receiver phase
;S1 = 0H
;D2 = 5 msec
```

INVH1G.AU

```
; 1D Proton Spectrum Using Inverse Mode With Decoupling During Acquisition
  Using GARP1: (see also INVH1.AU)
;   A.J. Shaka, P.B. Barker & R. Freeman, J. Magn. Reson 64,
;    547 - 552 (1985)


1 ZE
  D2 S1 TL                          ;switch to low power output
2 D1                                ;relaxation delay
  P1:D PH1                          ;excitation pulse
  D5                                ;DE/2
3 D5:D PH9                          ;reset phase before acquisition
4 D6 ADC                            ;D6 = 2 usec
5 (P7  PH10 P10 PH12 P28 PH10 P30 PH12 P15 PH10 P11 PH12)
  (P19 PH10 P20 PH12 P24 PH10 P27 PH12 P14 PH10 P8  PH12)
  (P6  PH10 P17 PH12 P22 PH10 P25 PH12 P29 PH10 P12 PH12)
  (P16 PH10 P18 PH12 P21 PH10 P23 PH12 P26 PH10 P13 PH12)
  (P9  PH10)
  L0 TO 5 TIMES 2
6 (P7  PH12 P10 PH10 P28 PH12 P30 PH10 P15 PH12 P11 PH10)
  (P19 PH12 P20 PH10 P24 PH12 P27 PH10 P14 PH12 P8  PH10)
  (P6  PH12 P17 PH10 P22 PH12 P25 PH10 P29 PH12 P12 PH10)
  (P16 PH12 P18 PH10 P21 PH12 P23 PH10 P26 PH12 P13 PH10)
  (P9  PH12)
  L1 TO 6 TIMES 2
7 L2 TO 5 TIMES UPR
8 RCYC=2 PH2                        ;loop for NS scans
  D2 DO TH                          ;switch to high power output
EXIT

PH1=0 2 2 0 1 3 3 1
PH2=R0 R2 R2 R0 R1 R3 R3 R1

PH9=0

PH10=0
PH12=2


;P6 = 25.5 degree pulse at low power output
;P7 = 30.5 deg
;P8 = 45.9 deg
;P9 = 53.4 deg
;P10= 55.2 deg
;P11= 62.2 deg
;P12= 64.9 deg
;P13= 65.6 deg
;P14= 66.4 deg
;P15= 69.3 deg
;P16= 70.9 deg
;P17= 72.7 deg
;P18= 77.2 deg
;P19= 85.0 deg
;P20= 91.8 deg
;P21= 98.2 deg
;P22=119.5 deg
;P23=133.6 deg
;P24=134.5 deg
;P25=138.2 deg
;P26=255.9 deg
;P27=256.1 deg
;P28=257.8 deg
;P29=258.4 deg
;P30=268.3 deg
```

```
;D2 = 5 msec
;S1 = 0H
;D5 = DE/2
;D6 = 2 usec
;L2 = 31.75 * 4 * (90 degree pulse length)  => AQ
```

INVINEPR.AU

; Inverse INEPT With Refocussing Using Inverse Mode. No Decoupling During Acquisition. Giving In-Phase Signals (see also INVH1.AU)

```
;  H-1:                180        90        180      - FID
;    X: D1 - 90 - D2 - 180 - D2 - 90 - D3 - 180 - D3 -

;M.R. Bendall, D.T. Peg, D.M. Doddrell & J. Field, J. Magn. Reson.
;   51, 520 - 526 (1983)


1 ZE
2 D1 S1 DO                         ;relaxation delay
3 (P3 PH1)                         ;90 deg X pulse
  D2
4 (P4 PH2)  (P2 PH4):D             ;180 deg X and H-1 pulse
  D2
5 (P3 PH3)  (P1 PH5):D             ;90 deg X and H-1 pulse
  D3                               ;1/(4J)XH
6 (P4 PH2)  (P2 PH6):D             ;180 deg X and H-1 pulse
  D3 DO
7 GO=2 PH7
8 D2 DO
9 EXIT

PH1=0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH2=0 2
PH3=1 1 3 3 3 3 1 1
PH4=0 2
PH5=0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1
    2 2 2 2 2 2 2 2 3 3 3 3 3 3 3 3
PH6=0 2 0 2 0 2 0 2 1 3 1 3 1 3 1 3
PH7=R0 R0 R2 R2 R2 R2 R0 R0 R1 R1 R3 R3 R3 R3 R1 R1


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;P3,P4 : 90, 180 deg X pulse
;D2 : 1/(6J)XH for all multiplicities
;D3 : 1/(4J)XH
;DS : 2 or 4
;NS : 32 * n
```

INVINEPT.AU

; Inverse INEPT Using Inverse Mode. No Decoupling During Acquisition. Giving Anti-Phase Signals. (see also INVH1.AU)

```
;  H-1:                180        90 - FID
;    X: D1 - 90 - D2 - 180 - D2 - 90

;M.R. Bendall, D.T. Pegg, D.M. Doddrell & J. Field, J. Magn. Reson.
;   51, 520 - 526 (1983)


1 ZE
2 D1 S1 DO                    ;relaxation delay
3 (P3 PH1)                    ;90 deg X pulse
  D2
4 (P4 PH2)  (P2 PH4):D        ;180 deg X and H-1 pulse
  D2
5 (P3 PH3)  (P1 PH5):D        ;90 deg X and H-1 pulse
6 GO=2 PH6
7 EXIT

PH1=0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH2=0 2
PH3=1 1 3 3 3 3 1 1
PH4=0 2
PH5=0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1
    2 2 2 2 2 2 2 2 3 3 3 3 3 3 3 3
PH6=R0 R0 R2 R2 R2 R2 R0 R0 R1 R1 R3 R3 R3 R3 R1 R1


;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;P3,P4 : 90, 180 deg X pulse
;D2 : 1/(6J)XH for all multiplicities
;DS : 2 or 4
;NS : 32 * n
```

INVMLEV.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Bird Sequence And MLEV17 For Homonuclear Hartman-Hahn Mixing In Inverse Mode. No Decoupling During Acquisition (see also INVH1.AU)

;L. Lerner and A. Bax, J, Magn. Reson. 69, 375-380 (1986)

```
  II
1 ZE
2 D1 S1 DO                        ;relaxation delay
  P1:D PH1                        ;90 deg H-1 pulse
  D2                              ;1/(2J)XH
  (P2 PH1):D (P4 PH7)             ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                              ;recovery delay
  P3 PH10                         ;90 deg X pulse
3 P1:D PH1                        ;90 deg H-1 pulse
4 D2                              ;1/(2J)XH
5 P3 PH3                          ;90 deg X pulse
  D0                              ;t1/2
6 P2:D PH2
  D0
7 P3 PH4
  D2 S2 DO                        ;1/(2J)XH refocussing delay
                                  ;switch decoupler power for spinlock
  (P7 PH16):D                     ;trim pulse
8 (P8 PH12 P9 PH13 P8 PH12):D     ;MLEV y-spinlock
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P9 PH13):D                     ;180 deg H-1 pulse
  L6 TO 8 TIMES UPR
  (P7 PH16):D                     ;trim pulse
9 GO=2 PH5
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH6=1
PH7=0
PH9=2
PH10=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH2=0
PH3=0 1 2 3
PH4=0 0 0 0 2 2 2 2
PH5=R0 R3 R2 R1 R2 R1 R0 R3

PH12=0
PH13=1
PH14=2
```

```
PH15=3
PH16=1


;D1 * 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse at power level S1
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;S2 : power level for spinlock (90 deg pulse 25 - 35 usec)
;P7 = 2.5 msec
;P8,P9 : 90, 180 deg H-1 pulse at power level S2
;L6 : spinlock time = (L6 * 66 * P8) + (2 * P7)
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0 = 2
```

INVMLEVP.AUR

; 2D H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Bird Sequence And Mlev17 For Homonuclear Hartman-Hahn Mixing In Inverse Mode. Phase Sensitive Using TPPI. No Decoupling During Acquisition (see also INVH1.AU)

```
;L. Lerner and A. Bax, J, Magn. Reson. 69, 375-380 (1986)


  II
1 ZE
2 D1 S1 DO                         ;relaxation delay
  P1:D PH1                         ;90 deg H-1 pulse
  D2                               ;1/(2J)XH
  (P2 PH1):D (P4 PH7)              ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                               ;recovery delay
  P3 PH10                          ;90 deg X pulse
3 P1:D PH1                         ;90 deg H-1 pulse
4 D2                               ;1/(2J)XH
5 P3 PH3                           ;90 deg X pulse
  D0                               ;t1/2
6 P2:D PH2
  D0
7 P3 PH4
  D2 S2 DO                         ;1/(2J)XH refocussing delay
                                   ;switch decoupler power for spinlock
  (P7 PH16):D                      ;trim pulse
8 (P8 PH12 P9 PH13 P8 PH12):D      ;MLEV y-spinlock
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH12 P9 PH13 P8 PH12):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P8 PH14 P9 PH15 P8 PH14):D
  (P9 PH13):D                      ;180 deg H-1 pulse
  L6 TO 8 TIMES UPR
  (P7 PH16):D                      ;trim pulse
9 GO=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT


PH1=0
PH6=1
PH7=0
PH9=2
PH10=0 0 0 0 2 2 2 2
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0
```

```
PH12=0
PH13=1
PH14=2
PH15=3
PH16=1



;D1 : 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse at power level S1
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;S2 : power level for spinlock (90 deg pulse 25 - 35 usec)
;P7 = 2.5 msec
;P8,P9 : 90, 180 deg H-1 pulse at power level S2
;L6 : spinlock time = (L6 * 66 * P8) + (2 * P7)
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
;MC2 = W
```

INVP90D4.AU

; Determination Of 90 Degree X Pulse By Observation Of H-1 Using Fast TLO Output (RCP7) And Additional Amplifier (e.g. BFX5, Gated With RCP3)

```
1 ZE
2 D1 S1 DO                      ;relaxation delay
3 P1:D PH1                      ;proton excitation pulse
4 D2                            ;1/(2J)XH
  P0:C7
5 P3:C7:T:C3 PH2                ;X pulse
  P0:C7
6 GO=2 PH3
EXIT

PH1=0 2 2 0 1 3 3 1
PH2=0 2 2 0 1 3 3 1
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;D1 : 5 - 10 T1
;P1 : 90 degree proton pulse
;D2 : 1/(2J)XH
;P0 = 5 usec
;P3 :  0 degree for phase correction of satellite signals
;               (anti-phase dublett)
;     90 degree gives minimum of satellite signals
;    180 degree gives inverted spectrum of satellite signals
;               (anti-phase dublett)
```

INVP90DE.AU

; Determination Of 90 Degree X Pulse By Observation Of H-1

```
1 ZE
2 D1 S1 DO                      ;relaxation delay
3 P1:D PH1                      ;proton excitation pulse
4 D2                            ;1/(2J)XH
5 P3 PH2                        ;X pulse
6 GO=2 PH3
EXIT

PH1=0 2 2 0 1 3 3 1
PH2=0 2 2 0 1 3 3 1
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;D1 : 5 - 10 T1
;P1 : 90 degree proton pulse
;D2 : 1/(2J)XH
;P3 :  0 degree for phase correction of satellite signals
;               (anti-phase dublett)
;     90 degree gives minimum of satellite signals
;    180 degree gives inverted spectrum of satellite signals
;               (anti-phase dublett)
```

INVREC.AUR

```
; Inversion-Recovery T1 With Delay List Cycling
;   D1 - 180 - VD - 90 - FID

1 ZE
2 WR #1         ;PREPARE A SET OF ZEROED DISK FILES
3 IF #1
4 LO TO 2 TIMES C       ;FIRST C = NO. OF FILES = NE
  VC            ;SELECT SECOND 'C'
5 RF #1.001     ;RESET FILE EXTENSION, BEGIN CYCLE
6 RE #1         ;READ CURRENT FID
7 D1            ;RELAXATION DELAY FOR EQUILIBRIUM
8 P2            ;180 DEG PULSE
9 VD            ;VARIABLE DELAY (TAKEN FROM CURRENT 'VD' LIST)
10 GO=7         ;ACQUIRE DATA AFTER 90 DEG PULSE, LOOP TO 7
11 WR #1        ;STORE CURRENT FID
12 IF #1        ;INCREMENT FILE EXTENSION
13 IN=6         ;LOOP TO 6 AND INCREMENT VD LIST POINTER
14 LO TO 5 TIMES C      ;REPEAT CYCLE THROUGH DELAY LIST
                        ;C IS SECOND IN VC LIST
15 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES THE NUMBER OF TAU VALUES IN THE VD LIST, THE NUMBER
;OF FIDS STORED.
;THE CURRENT VD LIST MUST CONTAIN THE SET OF RECOVERY DELAYS
;TO BE USED (IN ANY ORDER).
;D1 = 5*T1
;P2 = 180 PULSE (CONSTANT PHASE)
;RD=0, PW= 90 PULSE
;NS = MULTIPLE OF 8, DS=0
;CURRENT VC LIST MUST CONTAIN AN ENTRY WHICH DEFINES THE NUMBER
;OF FILES (=NE) AND A SECOND ENTRY WHICH DEFINES THE NUMBER
;OF CYCLES TO BE MADE THROUGH THE VD LIST FOR LONG-TERM AVERAGING.
;TOTAL TRANSIENTS PER FILE = C*NS.
```

INVRECX.AUR

```
; Inversion-Recovery T1 For X-Nuclei With 1H Decoupling Using Delay List
  Cycling And Power Gating.
; 1H:  BB(S1)- - - - - S2 - - - - - -
;  X:  D1 - 180 - VD - D2 -90 - FID

1 ZE
2 WR #1         ;PREPARE A SET OF ZEROED DISK FILES
3 IF #1
4 LO TO 2 TIMES C       ;FIRST C = NO. OF FILES = NE
  VC            ;SELECT SECOND 'C'
5 RF #1.001     ;RESET FILE EXTENSION, BEGIN CYCLE
6 RE #1         ;READ CURRENT FID
7 D1 CPD S1     ;RELAXATION DELAY FOR EQUILIBRIUM
                ;MINIMAL DECOUPLING POWER FOR NOE
8 P2            ;180 DEG PULSE
9 VD            ;VARIABLE DELAY (TAKEN FROM CURRENT 'VD' LIST)
  D2 S2         ;INCREASE DEC. POWER FOR GOOD DECOUPLING
10 GO=7         ;ACQUIRE DATA AFTER 90 DEG PULSE, LOOP TO 7
   D2 CPD S1    ;REDUCE DEC. POWER
11 WR #1        ;STORE CURRENT FID
12 IF #1        ;INCREMENT FILE EXTENSION
13 IN=6         ;LOOP TO 6 AND INCREMENT VD LIST POINTER
14 LO TO 5 TIMES C      ;REPEAT CYCLE THROUGH DELAY LIST
                        ;C IS DEFINED IN VC LIST
15 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES THE NUMBER OF TAU VALUES IN THE VD LIST, THE NUMBER
;  OF FIDS STORED.
;THE CURRENT VD LIST MUST CONTAIN THE SET OF RECOVERY DELAYS
;  TO BE USED (IN ANY ORDER). ACTUAL TAU VALUE IS VD+D2.
;D1 = 5*T1
;D2 = 2 MSEC TO CHANGE DEC. POWER
;S1 = CA. 0.4 WATT FOR NOE
;S2 = POWER FOR OPTIMUM DECOUPLING WITH MOD=0 OR 1
;  FOR MOD=1 SET P9 = 90 DEC. PULSE FOR POWER S2.
;P2 = 180 PULSE (CONSTANT PHASE)
;RD=0, PW= 90 PULSE
;NS = MULTIPLE OF 8, DS=0
;CURRENT VC LIST MUST CONTAIN AN ENTRY WHICH DEFINES THE NUMBER
;OF FILES TO BE STORED AND A SECOND ENTRY DEFINING THE NUMBER
;OF CYCLES TO BE MADE THROUGH THE VD LIST FOR LONG-TERM AVERAGING.
;TOTAL TRANSIENTS PER FILE = C*NS.
```

INVREL2D.AUR

; 2D Relayed H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Bird Sequence In Inverse Mode. No Decoupling During Acquisition (see also INVH1.AU)

;L. Lerner and A. Bax, J, Magn. Reson. 69, 375-380 (1986)

```
  II
1 ZE
2 D1 S1 DO                          ;relaxation delay
  P1:D PH1                          ;90 deg H-1 pulse
  D2                                ;1/(2J)XH
  (P2 PH1):D (P4 PH7)               ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                                ;recovery delay
  P3 PH10                           ;90 deg X pulse
3 P1:D PH1                          ;90 deg H-1 pulse
4 D2                                ;1/(2J)XH
5 P3 PH3                            ;90 deg X pulse
  D0                                ;t1/2
6 P2:D PH2
  D0
7 P3 PH4
  D2                                ;1/(2J)XH  refocussing delay
  P1:D PH6                          ;90 deg H-1 y pulse for polarisation
                                    ;transfer
9 GO=2 PH5
  WR #1
  IF #1
  IN=1
EXIT

PH1=0
PH6=1
PH7=0
PH9=2
PH10=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH2=0
PH3=0 1 2 3
PH4=0 0 0 0 2 2 2 2
PH5=R0 R3 R2 R1 R2 R1 R0 R3


;D1 * 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;DS : 2 or 4
;NS : 8 * n
;D0 = 3 usec
;IN : 1 / 2SW(X) = DW(X)
;ND0 = 2
```

INVREL2P.AUR

; 2D Relayed H-1/X Correlation Via Heteronuclear Zero And Double Quantum Coherence Using Bird Sequence In Inverse Mode. Phase Sensitive Using TPPI. No Decoupling During Acquisition (see also INVH1.AU)

```
;L. Lerner and A. Bax, J, Magn. Reson. 69, 375-380 (1986)


  II
1 ZE
2 D1 S1 DO                        ;relaxation delay
  P1:D PH1                        ;90 deg H-1 pulse
  D2                              ;1/(2J)XH
  (P2 PH1):D (P4 PH7)             ;180 deg H-1 and X pulse
  D2
  P1:D PH9
  D4                              ;recovery delay
  P3 PH10                         ;90 deg X pulse
3 P1:D PH1                        ;90 deg H-1 pulse
4 D2                              ;1/(2J)XH
5 P3 PH3                          ;90 deg X pulse
  D0                              ;t1/2
6 P2:D PH2
  D0
7 P3 PH4
  D2                              ;1/(2J)XH  refocussing delay
  P1:D PH6                        ;90 deg H-1 y pulse for polarisation
                                  ;transfer
9 GO=2 PH5
  WR #1
  IF #1
  IP3
  IN=1
EXIT

PH1=0
PH6=1
PH7=0
PH9=2
PH10=0 0 0 0 2 2 2 2
PH2=0
PH3=0 2
PH4=0 0 2 2
PH5=R0 R2 R2 R0


;D1 * 1-5 T1
;S1 = 0H
;P1,P2 : 90, 180 deg H-1 pulse
;D2 : 1/(2J)XH
;P3,P4 : 90, 180 deg X pulse
;D4 : optimize to give null for protons bound to C-12
;DS : 2 or 4
;NS : 4 * n
;D0 = 3 usec
;IN : 1 / 4SW(X) =(1/2) DW(X)
;ND0 = 4
;MC2 = W
```

INVT1.AUR

; Determination Of X-Nucleus T1 Values By Observation Of Proton Signals In Inverse Mode (see also INVH1.AU)

```
  RVD VDLIST
1 ZE
2 D5 S2 DO
  D1 BB
  P4 PH1
  VD
  D5 CW
  D5 DO S1
  P3 PH2
  D3
  (P4 PH6) (P2 PH7):D
  D3
  (P3 PH3) (P1 PH4):D
  GO=2 PH5
  WR #1
  IF #1
  IN=1
EXIT


PH1=0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
PH2=0 2
PH3=1 1 1 1 1 1 1 1 3 3 3 3 3 3 3 3
PH4=0 0 1 1 2 2 3 3
PH5=R0 R2 R1 R3 R2 R0 R3 R1 R2 R0 R3 R1 R0 R2 R1 R3
PH6=1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
    1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
    3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3
    3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3
PH7=0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
    2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2
    2 2 2 2 2 2 2 2 2 2 2 2 2 2 2 2


;D5 : > 400 usec
;S2 : power level for proton saturation
;D1 : 3-5 T1 for X-nucleus
;S1 = 0H
;P1, P2 : 90, 180 deg H-1 pulse at power level S1
;D3 : 1/(4J)XH
;P3, P4 : 90, 180 deg X pulse
;DS : 4
;NS : 64 * n
;VD : variable delay according to VDLIST
;NE = number of values in VDLIST
```

JMODXH.AUR

```
; J-Modulated Spin-Echo For X-Nuclei Coupled To 1H.  Can Be Used To Determine
  The Number Of Attached Protons.
;   1H: BB - DO ------ BB ----------
;    X: D1 - 90 - VD - 180 - VD - FID

1 ZE
2 D1 S1 CPD     ;RELAXATION DELAY FOR X-NUCLEUS,
                ;LOW-POWER DEC. TO HOLD NOE
3 D2 DO S2      ;GATE DEC. OFF, PREPARE DEC. POWER FOR GOOD DEC.
4 P1 PH1        ;90 DEG PULSE X-NUCLEUS
5 VD            ;EVOLUTION PERIOD WITH J-MOD.
6 P2 PH2 CPD    ;180 DEG PULSE, TURN ON BB DEC.
7 VD            ;REFOCUS SHIFTS WITHOUT J-MOD.
8 GO=2 PH3      ;ACQUIRE ECHO FID WITH DECOUPLING, LOOP TO 2
9 D2 S1 CPD     ;TURN DOWN DEC. POWER
10 WR #1        ;STORE FID
11 IF #1        ;INCREMENT FILE EXTENSION
12 IN=1         ;LOOP FOR NEXT EXPERIMENT,
                ;INCREMENT VD LIST POINTER
13 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;EXORCYCLE
    A2 A2 A2 A2 A3 A3 A3 A3
PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A1 A3
PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME FOR FIDS
;NE DEFINES NUMBER OF EXPERIMENTS, NO. OF VD VALUES
;D1 = DELAY FOR RELAXATION OF X-NUCLEUS
;D2 = 5 MSEC TO SWITCH DEC. POWER
;P1 = 90 DEG PULSE FOR X
;P2 = 180 DEG PULSE FOR X
;RD=PW=0,
;S1 = LOW POWER FOR NOE
;S2 = HIGHER POWER FOR GOOD DEC. WITH CPD, IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING
;     SECOND VD DELAY.

;CHOICE OF VD ALLOWS FOR A PHASE SELECTION OF DIFFERENT
;MULTIPLICITIES. E.G. VD = 1/J(XH) GIVES SINGLET AND
;TRIPLET MULTIPLICITIES POSITIVE, DOUBLET AND QUARTET
;TYPES ARE NEG. VD = 0.5/J(XH) GIVES QUATERNARY X ONLY.
```

JRES.AUR

```
; Homonuclear J-Resolved 2-D NMR Using The Hahn Spin-Echo.

;   D1 - 90 - D0 - 180 - D0 - FID

;  F2 DOMAIN (AFTER TILT)= CHEM. SHIFT AND HETERONUC. J
;  F1 DOMAIN = HOMONUC. J
;  ARTEFACTS OCCUR WHEN SPIN SYSTEM IS NOT PURE 1ST-ORDER

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG PULSE
4 D0            ;FIRST HALF OF EVOLUTION PERIOD
5 P2 PH2        ;180 DEG PULSE
6 D0            ;SECOND HALF OF EVOLUTION, REFOCUS SHIFTS
                ;BUT J-MODULATION CONTINUES.
7 GO=2 PH3      ;ACQUIRE FID
8 WR #1         ;STORE FID (SERIES FILE)
9 IF #1         ;INCREMENT FILE NUMBER
10 IN=1         ;INCREMENT D0 BY 'IN', LOOP FOR NEXT EXPER.
11 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;EXORCYCLE
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE = NUMBER OF FIDS = TD1
;NS CAN BE 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1,P2 = 90,180 DEG PULSES
;D0 = 3E-6 AS INITIAL DELAY
;IN = 0.25/SW1
;SW1 > HALF THE WIDTH OF LARGEST MULTIPLET
;ND0 = 2

;NB: FOR TILT, I2D MUST BE = 1,2,4,8,16,... WITHIN ALLOWED
;  ROUND-OFF ERROR OF CA. 0.5%
```

JRESX.AUR

; Homonuclear J-Resolved 2-D NMR For X-Nuclei With Power-Gated 1H Decoupling Using The Hahn Spin-Echo.

```
; 1H:   BB(S1)-S2- - - - - - - - - - - - - -
;  X:   D1 -   D2 - 90 - D0 - 180 - D0 - FID

;  F2 DOMAIN (AFTER TILT)= CHEM. SHIFT AND (NON-1H) HETERONUC. J
;  F1 DOMAIN = HOMONUC. J
;  ARTEFACTS OCCUR WHEN SPIN SYSTEM IS NOT PURE 1ST-ORDER

1 ZE
2 D1 CPD S1     ;RELAXATION, MINIMAL DECOUPLING FOR NOE
  D2 S2         ;SWITCH TO OPTIMAL DECOUPLING POWER
3 P1 PH1        ;90 DEG PULSE
4 D0            ;FIRST HALF OF EVOLUTION PERIOD
5 P2 PH2        ;180 DEG PULSE
6 D0            ;SECOND HALF OF EVOLUTION, REFOCUS SHIFTS
                ;BUT J-MODULATION CONTINUES.
7 GO=2 PH3      ;ACQUIRE FID
  D2 S1         ;REDUCE DEC. POWER
8 WR #1         ;STORE FID (SERIES FILE)
9 IF #1         ;INCREMENT FILE NUMBER
10 IN=1         ;INCREMENT D0 BY 'IN', LOOP FOR NEXT EXPER.
11 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1     ;EXORCYCLE
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE = NUMBER OF FIDS = TD1
;NS CAN BE 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1  , S1 = CA. 0.5 WATT TO HOLD NOE
;D2 = 5 MSEC
;S2 = OPTIMAL DEC. POWER FOR CPD MODE (P9), IMPORTANT TO
;     MINIMIZE LOSS OF TRANSVERSE MAGNETIZATION DURING EVOLUTION.
;P1,P2 = 90,180 DEG PULSES
;D0 = 3E-6 AS INITIAL DELAY
;IN = 0.25/SW1
;SW1 > HALF THE WIDTH OF LARGEST MULTIPLET
;ND0 = 2

;NB: FOR TILT, I2D MUST BE = 1,2,4,8,16,... WITHIN ALLOWED
;  ROUND-OFF ERROR OF CA. 0.5%
```

JSCALE.AUR

```
; Acquisition Of X-Nucleus Spectrum With Uniform Scaling Of X-H Couplings,
  Using Interrupted Waltz-8 Decoupling
; G.A.MORRIS, ET AL, J.MAGN.RES. 58, 155 (1984)

1 ZE
2 D1 BB S2       ;RELAXATION DELAY WITH DEC. FOR NOE
3 P5:A           ;TRANSM. PULSE (RECEIVER BLANKED)
4 D5             ;DE/2 (RECEIVER STILL OFF)
5 D5 PH9 CW      ;SET REFERENCE PHASE FOR DETECTION AND
                 ; OPEN RECEIVER GATE, SET CW MODE
6 D6 ADC         ;D6=2 USEC, 'ADC' OPENS REC. AND STARTS DIGITIZER
                 ; TAKE TD DATA POINTS USING DWELL TIME DW
7 (P2 PH2 P4 PH0 P2 PH2 P3 PH0 P1 PH2):D  ;ELEMENT 'K'
8 (P2 PH0 P4 PH2 P2 PH0 P3 PH2 P1 PH0):D  ;ELEMENT 'K-BAR'
9 L0 TO 8 TIMES 2                         ;REPEAT 'K-BAR'
  (P2 PH2 P4 PH0 P2 PH2 P3 PH0 P1 PH2):D  ;ELEMENT 'K'
  D7 DO                          ;INTERRUPT DECOUPLING
  L1 TO 7 TIMES UPR              ;REPEAT WALTZ-8 SEQUENCE
10 RCYC=2 PH8                    ;LOOP FOR NS SCANS
  EXIT

PH0=0   ;DECOUPLER PHASES
PH1=1
PH2=2
PH3=3
PH8=R0 R0 R2 R2 R1 R1 R3 R3
PH9=0 0 0 0 3 3 3 3             ;REFERENCE PHASE FOR DETECTION

;PERFORMS DATA ACQUISITION IN A MANNER IDENTICAL TO 'GO' (QP)
; D1 IS EQUIVALENT TO 'RD'
; S2 DEFINES DECOUPLER POWER
; P5 IS EQUIVALENT TO PW
; 2*D5 IS EQUIVALENT TO DE
; D6=2 USEC FOR ADC COMMAND
; D7 ADJUST FOR SCALING: J(RED)=J*D7/(D7 + 48*P1)
; P1=90 DEG 1H DEC. PULSE AT POWER SETTING S2
; P2,P3,P4=180,270,360 DEG DEC. PULSE

;L1 = LOOP COUNTER, SET SO THAT L1*(48*P1 + D7) => AQ.
```

MCHSAMA.AU

```
; Manual Sample Change

1 SXM
2 RJX
3 LOPO
4 ROT
5 LOCK
6 RJXS
7 AU @
  JOUA
8 IJX
9 LO TO 1 TIMES 10000
10 EXIT
```

MCHSAMP.AU

```
; Manual Sample Change


1 JOUP
2 RJX
3 RJXS
4 AU @
5 LO TO 1 TIMES 1000
6 EXIT
```

MLEV17PC.AUR

```
; Homonuclear Hartmann-Hahn Transfer Using MLEV17 Sequence For Mixing.  This
  Sequence Is Sensitive To Errors In Quad-Adjustment.  The Use Of MLEV17PH.AUR
  Is Recommended.
;For O1/O2-Coherence
;This Requires A Special Directional Coupler (10 Db Loss On F2)
;Phase Sensitive Using Tppi
;Lit: A. Bax & D.G. Davis, J. Magn. Reson. 65, 355 (1985)

  II
1 ZE
2 D1 S1 DO                ;relaxation delay
  P1 PH1                  ;90 deg transmitter H-1 pulse
  D0                      ;t1
  D2 PH5
  (P3 PH5):D              ;trim pulse (decoupler)
3 D2 PH2                  ;MLEV17 y-spinlock (decoupler)
  (P5 PH2^):D
  D2 PH3
  (P6 PH3^):D
  D2 PH4
  (P5 PH4^):D
  LO TO 3 TIMES 16
  D2 PH5
  (P7 PH5):D              ;60-180 degree pulse to remove effects of
                          ;pulse imperfections during MLEV16-part
                          ;(decoupler)
  L6 TO 3 TIMES UPR       ;repeat sequence to get appropiate
                          ;length of mixing time
  D2 PH5
  (P4 PH5):D              ;trim pulse (decoupler)
  GO=2 PH6
  WR #1
  IF #1
  IP1
  IN=1
EXIT

PH1=0 2
PH2=0 0 2 2 2 0 0 2 2 2 0 0 0 2 2 0
PH3=1 1 3 3 3 1 1 3 3 3 1 1 1 3 3 1
PH4=0 0 2 2 2 0 0 2 2 2 0 0 0 2 2 0
PH5=1
PH6=R0 R2


;D1 : 1-5 T1
;S1 : power level for spinlock (90 deg pulse : 25 - 35 usec)
;P1 : 90 deg H-1 transmiter pulse
;D0 = 3 usec
;D2 = 2 usec
;P5, P6 : 90, 180 deg H-1 decoupler pulse at power level S1
;P3,P4 = ca. 2.5 msec
;P7 : 60 - 180 deg H-1 decoupler pulse
;L6 has to be optimized depending whether direct or remote
;   connectivities should be observed
;   one MLEV17-cycle has the length of 66 times the length of P1,
;   for direct connectivities the length of the mixing time
;   should be 0.1/J(HH), therefore
;   0.1/J(HH) = ca L6 * 66 * P1
;   to observe remote connectivities the length of the mixing
;   time has to be increased.
;NS : 2 * n
;DS : 2 or 4
;MC2 = W
```

```
;ND0 = 2
;IN : DW (H-1)

;this sequence is sensitive to errors in quad-adjustment
;the use of MLEV17PH.AUR is recommended
```

MLEV17PH.AUR

; Homonuclear Hartmann-Hahn Transfer With Mixing By Composite Pulse Cycle Using Inverse Mode, Phase Sensitive (TPPI)

```
;Lit: Ad Bax and Donald G.Davis J.Magn.Res. 65, 355 (1985)

1 ZE
2 D1 S1 DO                    ;relaxation time, set decoupler for
                              ;pulsing
  (P1 PH1):D                  ;preparation pulse
  D0                          ;evolution time
  (P3 PH6):D                  ;trim pulse to defocus magnetization
                              ;not parallel to the X-axis
3 (P1 PH2 P2 PH3 P1 PH2):D  ;start of MLEV16 cycle for
4 (P1 PH4 P2 PH5 P1 PH4):D  ;net magnetization transfer
  LO TO 4 TIMES 2
  (P1 PH2 P2 PH3 P1 PH2):D
5 (P1 PH4 P2 PH5 P1 PH4):D
  LO TO 5 TIMES 2
6 (P1 PH2 P2 PH3 P1 PH2):D
  LO TO 6 TIMES 2
  (P1 PH4 P2 PH5 P1 PH4):D
7 (P1 PH2 P2 PH3 P1 PH2):D
  LO TO 7 TIMES 2
  (P1 PH4 P2 PH5 P1 PH4):D
8 (P1 PH2 P2 PH3 P1 PH2):D
  LO TO 8 TIMES 2
9 (P1 PH4 P2 PH5 P1 PH4):D
  LO TO 9 TIMES 2
  (P2 PH3):D                   ;180 degree pulse to remove effects of
                               ;pulse imperfections during MLEV16-part
  L6 TO 3 TIMES UPR            ;repeat sequence to get appropiate
                               ;length of mixing time
  (P4 PH6):D                   ;trim pulse
  GO=2 PH7                     ;acquisition
  WR #1                        ;store .SER file on disk
  IF #1                        ;increment file extension
  IP1                          ;increment phase 1 by 90 degree (TPPI)
  IN=1                         ;increment D0 and loop for next
                               ;experiment
EXIT

PH1=0 2 2 0 1 3 3 1
PH2=3 1 3 1 0 2 0 2
PH3=0 2 0 2 1 3 1 3
PH4=1 3 1 3 2 0 2 0
PH5=2 0 2 0 3 1 3 1
PH6=0 2 0 2 1 3 1 3
PH7=R0 R2 R2 R0 R1 R3 R3 R1

;RD = PW = 0
;P1 = 90 degree 1H dec. at power S1
;P2 = 180 degree
;P3 = P4 ca 2.5 msec
;D0 = 1 usec
;D1 = relaxation delay >T1
;S1 = decoupler power for P1 and spin locking (MLEV-CYCLE)
;   ca 2 to 4 watt
;L0 = 1
;L6 has to be optimized depending whether direct or remote
;   connectivities should be observed
;   one MLEV17-cycle has the length of 66 times the length of P1,
;   for direct connectivities the length of the mixing time
;   should be 0.1/J(HH), therefore
;   0.1/J(HH) = ca L6 * 66 * P1
```

```
;   to observe remote connectivities the length of the mixing
;   time has to be increased.

;MC2 = W, ND0 = 2
;SW1 = SW/2
```

NOEDIFF.AUR

; NOE Difference Spectroscopy Using One Freq. List To Define A Series Of Irradiation Points (On-Resonance) And One Control (Off-Resonance). The Individual FIDs Are Stored. For Long-Term Averaging The Routine Cycles Through The Freq. List And Fids Several Times. Also Can Be Used For Pseudo-Indor.

```
1 ZE
2 WR #1          /DEFINE FID
                 ;PREPARE A SET OF ZEROED FILES ON DISK
3 IF #1
4 LO TO 2 TIMES C  ;C= NO. OF FIDS TO BE STORED
  FL #2          /DEFINE FREQ. LIST
                 ;READ IN DESIRED FREQ. LIST
5 RF #1.001      ;RESET FILE EXTENSION TO .001, BEGIN CYCLE
6 RE #1          ;READ CURRENT FID FILE
7 D3 O2 S3       ;SET DEC. FREQ. O2 FROM CURRENT FL LIST
8 D1 DO          ;RELAX. TIME WITH DEC. GATED OFF
9 D2 HG          ;IRRAD. TIME (CA. T1) USING POWER S3
10 GO=8 DO       ;ACQUIRE DATA WITH DEC. OFF, LOOP TO 8
11 WR #1         ;STORE CURRENT ACCUMULATED FID
12 IF #1         ;INCREMENT FID EXTENSION
13 LO TO 6 TIMES C      ;LOOP TO 6 FOR EACH FREQ. IN FL LIST
14 IN=5          ;LOOP FOR ANOTHER CYCLE
                 ;NE=NUMBER OF CYCLES THROUGH LIST
15 EXIT

;PROGRAM REQUESTS FILENAME #1 FOR FIDS, #2 FOR FREQ. LIST.
;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS ONE O2
;ENTRY FOR EACH DESIRED IRRAD. POINT PLUS ONE OFF-RES. CONTROL
;VALUE FOR O2 WHICH SHOULD BE WITHIN THE SW REGION (E.G. AT ONE
;EDGE OF THE SPECTRUM). THE NUMBER OF FREQ. IN THE LIST MUST BE
;DEFINED BY AN ENTRY IN A 'VC' LIST, WHICH ALSO DEFINES THE
;NUMBER OF FIDS TO BE STORED.

;NS DEFINES THE NO. OF TRANSIENTS PER CYCLE FOR EACH O2 VALUE
;AND SHOULD BE A MULTIPLE OF 8.

;NE DEFINES THE NO. OF CYCLES TO BE MADE THROUGH COMPLETE LIST.
;TOTAL TRANSIENTS PER FID = NE*NS.
;USE 2-4 DUMMY SCANS FOR STEADY-STATE!
;RD=0
;D3 = 0.1 SEC TO SET O2
;D1+AQ = 2-4*T1 FOR TRUNCATED NOE APPLICATIONS WHERE NO SECONDARY
;  OR STEADY-STATE EFFECTS (SPIN-DIFFUSION) ARE DESIRED.
;D2 = CA. T1 FOR SMALL MOLECULES (EXTREME NARROWING LIMIT)
;   = 50-200 MSEC FOR LARGE MOLECULES (CROSS-RELAXATION).

;S3 DEFINES DEC. POWER TYPICALLY 35-55L DEPENDING ON REQUIRED
;IRRAD. BANDWIDTH.
```

NOEMULT.AUR

; NOE Difference Spectroscopy Using A Series Of Freq. Lists To Define Multiple Irradiation Points For Each On-Resonance Site And One Control (Off-Resonance). The Individual Fids Are Stored. For Long-Term Averaging The Routine Cycles Through The Freq. List And Fids Several Times. This Technique Allows Use Of Lower Power And Avoids Indor Effects.

```
; D.NEUHAUS, J.MAGN.RES. 53, 109 (1983)
; M.KINNS & J.K.M.SANDERS, J.MAGN.RES. 56, 518 (1984)

1 ZE
2 WR #1         /DEFINE FID
                ;PREPARE A SET OF ZEROED FILES ON DISK
3 IF #1
4 LO TO 2 TIMES C  ;C= NO. OF FIDS TO BE STORED

5 RF #1.001     ;RESET FILE EXTENSION TO .001, BEGIN CYCLE
  RF #2.001     /DEFINE FREQ. LIST
6 RE #1         ;READ CURRENT FID FILE
  FL #2         ;READ CURRENT FREQ. LIST
  VC            ;SELECT SECOND 'C' FROM LIST
7 D3 O2 S3      ;SET DEC. FREQ. O2 FROM CURRENT FL LIST
8 D1 DO         ;RELAX. TIME WITH DEC. GATED OFF
9 D5 HG O2      ;TIME TO SET O2 VALUE (5 MSEC)
  D2
  LO TO 9 TIMES C      ;IRRAD. C*(D2+D5) SEC
10 GO=8 DO      ;ACQUIRE DATA WITH DEC. OFF, LOOP TO 8
11 WR #1        ;STORE CURRENT ACCUMULATED FID
12 IF #1        ;INCREMENT FID EXTENSION
   IF #2        ;INCREMENT FREQ. LIST EXTENSION
   VC           ;SELECT FIRST 'C' IN LIST
13 LO TO 6 TIMES C      ;LOOP TO 6 FOR EACH FREQ. IN FL LIST
14 IN=5         ;LOOP FOR ANOTHER CYCLE
                ;NE=NUMBER OF CYCLES THROUGH LIST
15 EXIT

;PROGRAM REQUESTS FILENAME #1 FOR FIDS, #2 FOR FL LISTS
;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS THE O2 VALUES FOR
; FOR EACH  IRRAD. POINT IN A MULTIPLET.
; THE LAST LIST CONTAINS ONE OFF-RES. CONTROL
;VALUE FOR O2 WHICH SHOULD BE WITHIN THE SW REGION (E.G. AT ONE
;EDGE OF THE SPECTRUM). THE NUMBER OF DIFFERENT LISTS MUST BE
;DEFINED BY THE FIRST ENTRY IN A 'VC' LIST, THE SECOND ENTRY
;DEFINES THE NUMBER OF LOOPS FOR IRRADIATION.

;NB: LONGEST FL LIST SHOULD BE THE FIRST ONE AND IN MEMORY
; BEFORE STARTING AU!!

;NS DEFINES THE NO. OF TRANSIENTS PER CYCLE FOR EACH FID
;AND SHOULD BE A MULTIPLE OF 8.

;NE DEFINES THE NO. OF CYCLES TO BE MADE THROUGH COMPLETE SET
;  OF LISTS. TOTAL TRANSIENTS PER FID=NE*NS
;USE 2-4 DUMMY SCANS FOR STEADY-STATE!
;RD=0
;D3 = 0.1 SEC TO SET O2
;D1+AQ = 2-4*T1 FOR TRUNCATED NOE APPLICATIONS WHERE NO SECONDARY
;  OR STEADY-STATE EFFECTS (SPIN-DIFFUSION) ARE DESIRED.
;SET D2+D5 AND VC COUNTER FOR 'LO TO 9' TO GIVE TOTAL DESIRED
; IRRAD. TIME, WHEREBY MINIMUM VALUE FOR D5 IS CA. 5 MSEC.

;S3 DEFINES DEC. POWER TYPICALLY 40-60L DEPENDING ON REQUIRED
;IRRAD. BANDWIDTH.
```

NOEPHHG.AUR

; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI) Mode With Pre-Saturation Of Solvent. Dipolar Coupling May Be Due To NOE Or Chemical Exchange.

```
;  D1 - 90 - D0 - 90 - D9 - 90 - FID

; G.BODENHAUSEN, H.KOGLER, R.R.ERNST, J.MAGN.RES. 58,370 (1984)

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; ZERO-QUANTUM SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED
; BY RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1 HG S3      ;RELAXATION, PRE-SAT.
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P1 PH2        ;MIXING PULSE, 90 DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P1 PH3        ;DETECTION PULSE, 90 DEG
8 GO=2 PH4 DO   ;ACQUIRE FID, WITHOUT DEC.
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
   IP1          ;INCREMENT PHASE PROGRAM PH1 (TPPI METHOD)
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0 A2
PH2=A0 A0 A0 A0 A0 A0 A0 A0
    A2 A2 A2 A2 A2 A2 A2 A2
PH3=A0 A0 A2 A2 A1 A1 A3 A3
PH4=R0 R2 R2 R0 R1 R3 R3 R1
    R2 R0 R0 R2 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;S3 = MINIMAL DEC. POWER FOR PRE-SATURATION
;P1 = 90 DEG

;D0 = 3E-6 INITIAL DELAY
;IN = DW, ND0=2, MC2=W FOR TPPI MODE  (SEE COSYPH.AU)
;SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: CAUSES RANDOM VARIATION OF MAX. +/- V9% FOR D9 TO CANCEL
;  SCALAR CORRELATION EFFECTS. SET V9 TO GIVE CA. 20 MSEC
;  VARIATION TO CANCEL ZERO-QUANTUM COHERENCE BETWEEN SPINS WITH
;  SHIFT DIFFERENCE >50 HZ.

;MC2=W, REV=Y, REDF=N
;TYPICALLY USE TD = SI2, NO ZERO-FILLING IN F2
;              NE = TD/4, ZERO-FILL IN F1, SI1=SI2/2
;MATRIX CAN ONLY BE SYMMETRIZED ABOUT DIAGONAL IF SI2=SI1.

;TO DEFINE PHASE CORRECTION: TAKE FIRST .SER FILE, TRANSFORM (FT)
;  WITH DESIRED WINDOW FUNCTION, AND PHASE CORRECT IN EP SO THAT
;  SPECTRUM HAS PURE NEGATIVE PHASE.
```

```
;  EXAMINE CONSTANTS IN PARAMETER DISPLAY OR WITH 'TY'.
;  'XFB' WILL APPLY THESE CONSTANTS IN F2 DOMAIN AND ZERO IN F1.
;  DIAGONAL PEAKS, NEG. NOE, AND CHEM. EXCHANGE WILL BE NEGATIVE,
;  POS. NOE WILL BE POSITIVE, J-CORRELATIONS MAY INTRODUCE
;  MIXED POS/NEG COMPONENTS.
```

NOESPHPC.AUR

; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI)
; Mode (See COSYPH.AU). Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

```
;  D1 - 90 - D0 - 90 - D9 - 90 - FID

; G.BODENHAUSEN, H.KOGLER, R.R.ERNST, J.MAGN.RES. 58, 370 (1984)

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; ZERO-QUANTUM SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED
; BY RANDOM VARIATION OF THE MIXING TIME D9.
;with presaturation using O1/O2-coherence

1 ZE
2 D2:T PH8
  D1 S3 HG            ;RELAXATION
  D2 DO
3 P1 PH1              ;90 DEG EXCITATION PULSE
4 D0                  ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P1 PH2              ;MIXING PULSE, 90 DEG
6 D9 HG               ;MIXING TIME FOR Z-MAGN. EXCHANGE
  D2 DO
7 P1 PH3              ;DETECTION PULSE, 90 DEG
8 GO=2 PH4            ;ACQUIRE FID
9 WR #1               ;STORE FID
10 IF #1              ;INCREMENT FILE NUMBER
   IP1                ;INCREMENT PHASE PROGRAM PH1 (TPPI METHOD)
11 IN=1               ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0 A2
PH2=A0 A0 A0 A0 A0 A0 A0 A0
    A2 A2 A2 A2 A2 A2 A2 A2
PH3=A0 A0 A2 A2 A1 A1 A3 A3
PH4=R0 R2 R2 R0 R1 R3 R3 R1
    R2 R0 R0 R2 R3 R1 R1 R3
PH8=0


;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG
;D2 : > 400 usec
;S3 : power level for presaturation

;D0 = 3E-6 INITIAL DELAY
;IN = DW, ND0=2, MC2=W FOR TPPI MODE  (SEE COSYPH.AU)
;SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: CAUSES RANDOM VARIATION OF MAX. +/- V9% FOR D9 TO CANCEL
;  SCALAR CORRELATION EFFECTS. SET V9 TO GIVE CA. 20 MSEC
;  VARIATION TO CANCEL ZERO-QUANTUM COHERENCE BETWEEN SPINS WITH
;  SHIFT DIFFERENCE >50 HZ.

;MC2=W, REV=Y, REDF=N
```

```
;TYPICALLY USE TD = SI2, NO ZERO-FILLING IN F2
;              NE = TD/4, ZERO-FILL IN F1, SI1=SI2/2
;MATRIX CAN ONLY BE SYMMETRIZED ABOUT DIAGONAL IF SI2=SI1.

;TO DEFINE PHASE CORRECTION: TAKE FIRST .SER FILE, TRANSFORM (FT)
;  WITH DESIRED WINDOW FUNCTION, AND PHASE CORRECT IN EP SO THAT
;  SPECTRUM HAS PURE NEGATIVE PHASE.
;  EXAMINE CONSTANTS IN PARAMETER DISPLAY OR WITH 'TY'.
;  'XFB' WILL APPLY THESE CONSTANTS IN F2 DOMAIN AND ZERO IN F1.
;  DIAGONAL PEAKS, NEG. NOE, AND CHEM. EXCHANGE WILL BE NEGATIVE,
;  POS. NOE WILL BE POSITIVE, J-CORRELATIONS MAY INTRODUCE
;  MIXED POS/NEG COMPONENTS.
```

NOESY.AUR

; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode). Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

```
;  D1 - 90 - D0 - 90(OR 45) - D9 - 90(OR 45) - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED BY
; RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P2 PH2        ;MIXING PULSE, 90 (OR 45) DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P3 PH3        ;DETECTION PULSE, 90 (OR 45) DEG
8 GO=2 PH4      ;ACQUIRE FID
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0                           ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=A0 A2 A1 A3                  ;SCANS 3-4 GIVE F1 QUAD (N-TYPE)
PH3=A0 A0 A1 A1 A2 A2 A3 A3      ;SCANS 5-8 SUPPRESS DBL. QUANTUM
    A1 A1 A2 A2 A3 A3 A0 A0
PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R1 R3 R3 R1 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG, P2 AND P3 = NORMALLY 90 DEG BUT CAN BE 45 DEG
;   TO GIVE REPRESENTATION LIKE COSY-45.
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: D9 WILL BE VARIED RANDOMLY BY MAX. +/- V9 % OF ITS VALUE
;  TO SUPPRESS ZERO-QUANTUM J-CROSS PEAKS (COSY); CHOOSE V9
;  SO THAT D9 IS VARIED BY CA. +/- 20 MSEC TO SUPPRESS J-CROSS
;  PEAKS BETWEEN SPINS WHOSE SHIFTS DIFFER BY >50 HZ.
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL
```

NOESYHG.AUR

; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode) With Pre-Saturation Of Solvent. Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

```
;  D1 - 90 - D0 - 90(OR 45) - D9 - 90(OR 45) - FID
```

```
; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED BY
; RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1 HG S3      ;RELAXATION WITH PRE-SATURATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P2 PH2        ;MIXING PULSE, 90 (OR 45) DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P3 PH3        ;DETECTION PULSE, 90 (OR 45) DEG
8 GO=2 PH4 DO   ;ACQUIRE FID WITH DEC. GATED OFF
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=A0 A2 A1 A3                 ;SCANS 3-4 GIVE F1 QUAD (N-TYPE)
PH3=A0 A0 A1 A1 A2 A2 A3 A3     ;SCANS 5-8 SUPPRESS DBL. QUANTUM
    A1 A1 A2 A2 A3 A3 A0 A0
PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R1 R3 R3 R1 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;S3 = DEC. POWER FOR PRE-SATURATION, SHOULD BE AS LOW AS POSSIBLE
;     TO AVOID BLOCH-SIEGERT EFFECTS (30-40L).
;P1 = 90 DEG, P2 AND P3 = NORMALLY 90 DEG BUT CAN BE 45 DEG
;   TO GIVE REPRESENTATION LIKE COSY-45.
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: D9 WILL BE VARIED RANDOMLY BY MAX. +/- V9 % OF ITS VALUE
;  TO SUPPRESS ZERO-QUANTUM J-CROSS PEAKS (COSY); CHOOSE V9
;  SO THAT D9 IS VARIED BY CA. +/- 20 MSEC TO SUPPRESS J-CROSS
;  PEAKS BETWEEN SPINS WHOSE SHIFTS DIFFER BY >50 HZ.
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL
```

NOESYPH.AUR

; Homonuclear Dipolar-Correlated 2-D NMR In Phase-Sens. (TPPI) Mode (See COSYPH.AU). Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

```
;  D1 - 90 - D0 - 90 - D9 - 90 - FID

; G.BODENHAUSEN, H.KOGLER, R.R.ERNST, J.MAGN.RES. 58, 370 (1984)

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; ZERO-QUANTUM SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED
; BY RANDOM VARIATION OF THE MIXING TIME D9.
```

```
1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P1 PH2        ;MIXING PULSE, 90 DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P1 PH3        ;DETECTION PULSE, 90 DEG
8 GO=2 PH4      ;ACQUIRE FID
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
   IP1          ;INCREMENT PHASE PROGRAM PH1 (TPPI METHOD)
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0 A2
PH2=A0 A0 A0 A0 A0 A0 A0 A0
    A2 A2 A2 A2 A2 A2 A2 A2
PH3=A0 A0 A2 A2 A1 A1 A3 A3
PH4=R0 R2 R2 R0 R1 R3 R3 R1
    R2 R0 R0 R2 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1 = 90 DEG

;D0 = 3E-6 INITIAL DELAY
;IN = DW, ND0=2, MC2=W FOR TPPI MODE  (SEE COSYPH.AU)
;SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: CAUSES RANDOM VARIATION OF MAX. +/- V9% FOR D9 TO CANCEL
;  SCALAR CORRELATION EFFECTS. SET V9 TO GIVE CA. 20 MSEC
;  VARIATION TO CANCEL ZERO-QUANTUM COHERENCE BETWEEN SPINS WITH
;  SHIFT DIFFERENCE >50 HZ.

;MC2=W, REV=Y, REDF=N
;TYPICALLY USE TD = SI2, NO ZERO-FILLING IN F2
;              NE = TD/4, ZERO-FILL IN F1, SI1=SI2/2
;MATRIX CAN ONLY BE SYMMETRIZED ABOUT DIAGONAL IF SI2=SI1.

;TO DEFINE PHASE CORRECTION: TAKE FIRST .SER FILE, TRANSFORM (FT)
;  WITH DESIRED WINDOW FUNCTION, AND PHASE CORRECT IN EP SO THAT
;  SPECTRUM HAS PURE NEGATIVE PHASE.
;  EXAMINE CONSTANTS IN PARAMETER DISPLAY OR WITH 'TY'.
;  'XFB' WILL APPLY THESE CONSTANTS IN F2 DOMAIN AND ZERO IN F1.
;  DIAGONAL PEAKS, NEG. NOE, AND CHEM. EXCHANGE WILL BE NEGATIVE,
;  POS. NOE WILL BE POSITIVE, J-CORRELATIONS MAY INTRODUCE
;  MIXED POS/NEG COMPONENTS.
```

NOESYX.AUR

; Homonuclear Dipolar-Correlated 2-D NMR (Magnitude Mode) For X-Nuclei With Power-Gated 1H Decoupling. Dipolar Coupling May Be Due To Noe Or Chemical Exchange.

```
;  D1 - 90 - D0 - 90(OR 45) - D9 - 90(OR 45) - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED BY
; RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1 CPD S1     ;RELAXATION WITH DECOUPLING
  D2 S2         ;SWITCH TO OPTIMAL DEC. POWER
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P2 PH2        ;MIXING PULSE, 90 (OR 45) DEG
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P3 PH3        ;DETECTION PULSE, 90 (OR 45) DEG
8 GO=2 PH4      ;ACQUIRE FID
  D2 S1         ;REDUCE DEC. POWER
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=A0 A2 A1 A3                 ;SCANS 3-4 GIVE F1 QUAD (N-TYPE)
PH3=A0 A0 A1 A1 A2 A2 A3 A3     ;SCANS 5-8 SUPPRESS DBL. QUANTUM
    A1 A1 A2 A2 A3 A3 A0 A0
PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R1 R3 R3 R1 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;D2 = 5 MSEC TO CHANGE DEC. POWER
;S1 = DEC. TO MAINTAIN HETERONUC. NOE
;S2 = OPTIMAL CPD DEC. POWER DURING PULSE SEQUENCE (P9)
;P1 = 90 DEG, P2 AND P3 = NORMALLY 90 DEG BUT CAN BE 45 DEG
;   TO GIVE REPRESENTATION LIKE COSY-45.
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1, SW1=SW/2

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9: D9 WILL BE VARIED RANDOMLY BY MAX. +/- V9 % OF ITS VALUE
;  TO SUPPRESS ZERO-QUANTUM J-CROSS PEAKS (COSY); CHOOSE V9
;  SO THAT D9 IS VARIED BY CA. +/- 20 MSEC TO SUPPRESS J-CROSS
;  PEAKS BETWEEN SPINS WHOSE SHIFTS DIFFER BY >50 HZ.
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
;MATRIX CAN BE SYMMETRIZED ABOUT DIAGONAL
```

NQR.AUR

```
; Cross-Polarization With Hartmann-Hahn Spin-Lock And Dipolar Dephasing For
  Non Quaternary Carbon Suppression To Be Used With AR Amplifier.
; ************** USE TLO ***************

;  1H: D1 - 90(+/-Y) - P2(X) - D5 - CW
;   X:                 P2(X) - D5 - FID(+/-)

1 ZE
2 D1 DO                  ;1H RELAXATION, DEC. OFF
  (P4):C8                ;UNBLANK AR
3 (P1 PH1):D:E (P1):C8   ;90 DEG 1H PULSE (POWER S1), BLANK RECEIVER
4 (P2 PH3):D (P2 PH2):T:E:C8     ;SPIN-LOCK 1H WITH PHASE 0
                         ;CROSS-POLARIZE TO X
                         ;VIA HARTMANN-HAHN MATCH.
  D5                     ;DIPOLAR DEPHASING
5 GO=2 PH4 CW            ;ACQUIRE X-NUCLEUS FID WITH CW 1H DECOUPLING
6 D3 DO
7 EXIT                   ;EXIT WITH DEC. GATED OFF

PH1=1 3
PH2=0 0 2 2 1 1 3 3
PH3=0
PH4=R0 R2 R2 R0 R1 R3 R3 R1

;NORMALLY 1H-5H WITH A MAX. OF CA. 16W TO FEED HP 1H ENDSTAGE.
;RD=PW=0
;NS=8*N
;D1 = 1-5*T1 FOR 1H BUT > 20*(AQ+P2) TO AVOID EXCESSIVE
;HEATING.
;D3 = 3 MSEC TO ENSURE DEC. IS OFF BEFORE EXITING.
;P1 = 90 DEG 1H PULSE = 90 DEG X PULSE AT HARTMANN-HAHN CONDITION
;P2 = SPIN-LOCK TIME (E.G. 0.5-5 MSEC)
;SET D5 BETWEEN 30 AND 100US
```

P1331.AUR

```
; Water Suppression With 1-3-3-1 Pulse Sequence.
; HORE, J.MAGN.RES. 54, 539 (1983).
;                   55, 283 (1983).
;         --                   --
; P1- D2 - P3 - D2 - P3 - D2 - P1 - FID

;P1 SET FOR 11.25 DEG GIVES AN EFFECTIVE 90 DEG FLIP
; USE ATTENUATOR TO REDUCE HIGH-POWER
; PULSES 10-20 DB SO THAT 90 DEG = 20-50 USEC.
; FOR BSV-7 TRANSMITTER USE TLO.
;SET OFFSET ON WATER, FURTHER NULLS OCCUR AT INTERVALS 1/D2
;FROM TRANSMITTER. THERE ARE NO ADJUSTABLE PARAMETERS.

1 ZE
2 D1              ;RELAXATION
3 P1 PH1          ;'1' PULSE
4 D2
5 P3 PH2          ;'3' PULSE
6 D2
7 P3 PH1
8 D2
9 P1 PH2
10 GO=2 PH3
   EXIT

PH1=A0 A2 A2 A0 A1 A3 A3 A1
PH2=A2 A0 A0 A2 A3 A1 A1 A3
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;RD=PW=0
;D1=1-5*T1
;P3=3*P1
;O1 ON SOLVENT RESONANCE (NULL)
;D2 <= 2/SW (NULLS AT 1/D2 INTERVALS FROM O1)
;LEFT AND RIGHT HALVES OF SPECTRUM HAVE OPPOSITE PHASE.
```

PENLOCK.AU

; For use with Barcode Pen EXE files with LOCKING removed from AUTOSAMA.AU. This removes the LOCK searching for second and subsequent experiments on a sample. As well as step 1, steps 3-5 must be deactivated in AUTOSAMA.AU.

```
1 LOPO
2 ROT
3 LOCK
4 EXIT
```

PENPXB.AU

; Used with Barcode Pen EXE files which have the option of manually selected expansion plots after AUTOMATION process-ing. This is done using with a Microprogram because of problems with 'PXB NAME.EXT' in foreground.

```
1 PXB INT1
2 EXIT
```

PENSAMA.AU

; For Use With Light-Pen

```
1 RJX
2 RJXS
3 AU @
4 JOUA
6 IJX
7 LO TO 1 TIMES 10000
8 EXIT
```

PENSAMP.AU

; For Use With Light-Pen

```
1 JOUP
2 RJX
3 AU @
4  RJXS
5 LO TO 1 TIMES 1000
6 EXIT
```

POMMIE.AUR

```
; DEPT Polarization Transfer From 1H To X-Nuclei For Refocussed Decoupled
  Spectra Using Max. MQ Coherence And Phase-Shifted Read Pulse.
; J.M.BULSING ET AL, J.MAGN.RES. 56, 167 (1984).
; 1-3 EXPERIMENTS CAN BE DONE AND STORED.

;  1H: D1 - 90 - D2 - 180 - D2 - 90-90 - D2 - BB
;   X:                 90         180         FID

1 ZE
2 WR #1                ;CREATE ZEROED FILES
  IF #1
  LO TO 2 TIMES C       ;'C' DEFINES NUMBER OF EXPERIMENTS
3 RF #1.001
5 RE #1
6 D1 S1  DO                     ;1H RELAXATION, SET DEC. POWER
                                ;FOR PULSING
  (P1 PH1 D2):D                 ;90 DEG 1H PULSE, SHIFTS AND
                                ;J(XH) EVOLVE
  (P2 PH2):D  (P3 PH5 D2)       ;180 DEG 1H PULSE TO REFOCUS
                                ;SHIFTS, 90 DEG X PULSE FOR DBL Q
                                ;COHERENCE
  (P1 PH3 P1 PH4):D (P4 PH6 D2 S2)
                         ;GENERATE MQ COHERENCE AND RECONVERT
            ;WITH PHASE-SHIFTED PULSE FOR POLARIZATION TRANSFER
            ;180 X PULSE TO REFOCUS X SHIFTS, SET DEC. POWER
7 GO=6 PH7 CPD           ;ACQUIRE FID WITH DEC.
8 D2 DO
  WR #1              ;STORE FID
  IF #1
  IP4                ;INCREMENT PHASE OF READ PULSE BY 45 DEG.
9 LO TO 5 TIMES C    ;LOOP FOR NEXT EXPERIMENT
10 IP4
   LO TO 10 TIMES 5
   IN=3               ;CYCLE THROUGH ALL EXPERIMENTS NE TIMES
   EXIT               ;EXIT WITH DEC. OFF

PH1=0                ;DECOUPLER PHASES
PH2=0 2 1 3
PH3=0 0 0 0 2 2 2 2
PH4=(8) 1 1 1 1 5 5 5 5

PH5= 0 0 0 0 0 0 0 0    ;TRANSMITTER PHASES
     1 1 1 1 1 1 1 1
     2 2 2 2 2 2 2 2
     3 3 3 3 3 3 3 3
PH6= 0 2 0 2 0 2 0 2
     1 3 1 3 1 3 1 3

PH7= R0 R0 R2 R2 R2 R2 R0 R0
     R1 R1 R3 R3 R3 R3 R1 R1
     R2 R2 R0 R0 R0 R0 R2 R2
     R3 R3 R1 R1 R1 R1 R3 R3

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;PH4 IS VARIABLE AND DEFINES HOW MQ COHERENCE IS READ AND
;  HOW MULTIPLICITIES WILL BE SELECTED:
;  E.G. PH4 = 45 DEG GIVES XH, XH3 POSITIVE, XH2 NEGATIVE
;           = 90 DEG GIVES XH ONLY (DEPENDS ON ACCURACY OF PH4)
;           = 135 DEG GIVES ALL XH, XH2, XH3 POS.
```

```
; NS=4*N (32 = COMPLETE PHASE CYCLE, 4= MINIMUM)
;RD=PW=0
```

POWGATE.AUR

; Power Gated Het.-Nuclear CPD Decoupling To Minimize Dielectric Heating

```
1 ZE            ;ZERO MEMORY
2 D1 CPD S1     ;BB DEC. WITH POWER S1 DURING D1
3 D2 S2         ;SWITCH TO POWER S2
4 GO=2          ;AQ. WITH DEC. POWER S2
5 D2 S1         ;LEAVE DEC. AT POWER S1 FOR NOE
6 EXIT

;S1 CA. 0.5 WATT OR AS NEEDED FOR NOE GENERATION
;S2 SET AS NEEDED FOR GOOD DECOUPLING
;D1 TYP. 1-5 TIMES AQ. AS DESIRED TO MINIMIZE AVERAGE POWER
;D2 TYP. 5-10 MSEC TO ALLOW POWER SWITCHING
;RD=0
;OPTIMUM EFFICIENCY: PW=90 DEG, D1+AQ=1.25*T1

;P9 DEFINES 90 DEG. DEC. PULSE FOR POWER S2 (SEE CPDCHECK.AU)
;S2 = CA. 1 W AND P9= CA. 100 USEC ARE TYPICAL (10MM).
```

PRESAT.AU

; Homo-Nuclear Presaturation (Solvent Suppression)

```
1 ZE            ;ZERO MEMORY
2 D1 HG S3      ;APPLY CW DEC. AT FREQ. O2, POWER S3, DURING D1
3 GO=2 DO       ;GATE DEC. OFF DURING AQ.
4 EXIT          ;EXIT WITH DEC. OFF

;RD=0
;D1 TYPICALLY 1-3 TIMES T1
;S3 TYP. 20-30L
```

PRESATHD.AUR

; A Series Of Homodecouplings With Solvent Suppression

```
1 ZE
2 FL #1         /DEFINE FREQ. LIST
                ;READ FREQ. LIST WITH FILENAME #1
3 IF #1         ;INCREMENT FREQ. LIST EXTENSION
4 D1 HG O2 S3   ;SOLVENT SUPPRESSION FOR TIME D1 USING
                ;THE FIRST FREQ. IN LIST AND POWER S3
5 D2 HD O2 S4   ;SWITCH TO HOMODEC. USING SECOND FREQ.
                ;IN LIST AND POWER S4
6 GO=4          ;ACQUIRE DATA AND LOOP TO 4
7 WR #2         /DEFINE FID
                ;STORE FID
8 IF #2         ;INCREMENT FID FILE EXTENSION
9 IN=1          ;LOOP FOR NEXT EXPERIMENT
10 D2 DO
11 EXIT         ;EXIT WITH DEC. OFF

;PROGRAM REQUESTS FILENAME FOR FREQ. LISTS (#1) AND FIDS (#2)
;NE DEFINES NUMBER OF EXPERIMENTS, IE. NO. OF DIFFERENT FREQ.
;LISTS. EACH LIST HAS TWO ITEMS: O2 FOR SOLVENT PEAK FOLLOWED
;BY THE O2 FOR DECOUPLING.
;D1 = TIME FOR SOLVENT SUPPRESSION, E.G. 1-3*T1
;S3 = POWER FOR SOLVENT SUPPRESSION, E.G. 20-30L
;D2 = 5-50 MSEC TO SWITCH TO HOMODEC. AND SET O2, LONGER TIMES
;  MINIMIZE TRANSIENT EFFECTS AT DECOUPLING SITE BUT ALLOW
;  MORE SOLVENT SIGNAL TO APPEAR.
;S4 = POWER FOR HOMODEC.
;RD=0, DS=4
```

PRESATM.AUR

; Multiple Peak Suppression By Pre-Saturation

```
1 ZE
  D3 DO S3      ;SET DECOUPLER POWER TO S3
2 D1 HG O2      ;TURN ON HOMO-GATED DEC.
                ;AND SET O2 FREQ. FROM CURRENT FL LIST
  D2            ;IRRADIATE AT THIS FREQ.
3 LO TO 2 TIMES C       ;AFTER TIME D1+D2 SET O2 TO NEXT VALUE
                        ;IN LIST, 'C' FROM VC LIST
4 GO=2 DO       ;ACQUIRE DATA WITH DEC. GATED OFF
5 EXIT

;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS THE O2 VALUES
;FOR EACH OF THE PEAKS TO BE SATURATED.
;D1 = TIME TO SWITCH THE DECOUPLER  O2 VALUE (>=5 MSEC)
;D1+D2 = IRRADIATION TIME AT ONE FREQ.  E.G. 20-200 MSEC.
;D3=1 SEC, S3=20-30L
;A 'VC' LIST MUST BE DEFINED WITH THE VALUE OF 'C' FOR STEP 3
;  THE TOTAL PRE-SAT. TIME IS THEN C*(D1+D2), E.G. 3-5 SEC.
;RD=0, DS=4
```

PRESATPC.AU

; Homo-Nuclear Presaturation (Solvent Suppression). This Requires O1/O2-Coherence

```
1 ZE                ;ZERO MEMORY
```

```
2 D2:T PH1           ;RESET PHASE FOR PRESAT. TO ZERO
  D1 HG S3           ;APPLY CW DEC. AT FREQ. O1, POWER S3, DURING D1
  D2 DO              ;SWITCH DEC. OFF
3 GO=2
4 EXIT

PH1=0

;RD=0
;D1 TYPICALLY 1-3 TIMES T1
;S3 TYP. 20-30L
;D2 = 1 MSEC
```

QUADECHO.AU

; Quadrupolar Echo Sequence To Be Used With The Normal Low-Power Transmitter

```
1 ZE
2 D8
3 D1
4 (P1 PH1 D6 P1 PH2 D6)
5 D3
  GO=2 PH3
EXIT

PH1 = 0 2
PH2 = 1
PH3 = R0 R2

; SET D6 LONGER THAN PROBE DEADTIME
```

QUADECHP.AU

??

QUAT.AUR

; Sequence Gives 1H-Coupled Spectrum For Only X-Nuclei That Are Not Protonated (Quaternary)

```
; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 272 (1983)

1 ZE            ;ZERO MEMORY
2 D1 CPD S2     ;RELAXATION WITH DECOUPLING FOR NOE
4 P3 PH3 DO     ;TURN DECOUPLER OFF, 90 DEG X PULSE
5 D2 S1         ;FIRST PART OF SPIN-ECHO PERIOD 1/(2J)
                ;  SET DEC. POWER FOR PULSE
6 (P4 PH4) (P1 PH1):D   ;180 DEG X PULSE, 90 DEG 1H PULSE
7 D2 S2         ;REFOCUSSING PERIOD, SET DEC. POWER
8 GO=2 PH5      ;ACQUIRE SPIN-ECHO FID WITHOUT DECOUPLING
9 EXIT

PH1=B0 B0 B0 B0 B2 B2 B2 B2
PH3=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH4=A0 A2 A1 A3 A1 A3 A2 A0     ;EXORCYCLE FOR SPIN-ECHO
    A1 A3 A0 A2 A0 A2 A1 A3
PH5=R0 R0 R2 R2 R1 R1 R3 R3

;NS=4*N
;RD=PW=0
;D1=1-5*T1 FOR X NUCLEUS
;S1 = 0H FOR PULSE
;S2 = CPD DEC. POWER FOR NOE
;D2 = 1/(2J) WHERE J IS AVERAGE XH COUPLING AT PROTONATED SITES,
;     IE. SAME D2 AS FOR DEPT.
;P1 = 90 1H DEC. PULSE AT POWER S1
;P3,P4 = 90,180 X-NUCLEUS PULSE.

;NB: RESIDUAL PROTONATED SIGNALS APPEAR WHEN P1 OR D2 IS NOT
;    OPTIMUM.
```

QUATD.AUR

```
; Sequence Gives 1H-Decoupled Spectrum For Only X-Nuclei That Are Not
  Protonated (Quaternary)
; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 272 (1983)

1 ZE            ;ZERO MEMORY
2 D1 CPD S2     ;RELAXATION WITH DECOUPLING FOR NOE
4 P3 PH3 DO     ;TURN DECOUPLER OFF, 90 DEG X PULSE
5 D2 S1         ;FIRST PART OF SPIN-ECHO PERIOD 1/(2J)
                ;  SET DEC. POWER FOR PULSE
6 (P4 PH4) (P1 PH1):D   ;180 DEG X PULSE, 90 DEG 1H PULSE
7 D2 S3         ;REFOCUSSING PERIOD, SET  POWER FOR DEC.
8 GO=2 PH5 CPD  ;ACQUIRE SPIN-ECHO FID WITH DECOUPLING
  D2 S2         ;TURN DOWN DEC. POWER
9 EXIT

PH1=B0 B0 B0 B0 B2 B2 B2 B2
PH3=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3
PH4=A0 A2 A1 A3 A1 A3 A2 A0     ;EXORCYCLE FOR SPIN-ECHO
    A1 A3 A0 A2 A0 A2 A1 A3
PH5=R0 R0 R2 R2 R1 R1 R3 R3

;NS=4*N
;RD=PW=0
;D1=1-5*T1 FOR X NUCLEUS
;S1 = 0H FOR PULSE
;S2 = DEC. POWER FOR NOE
;S3 = OPTIMUM DEC. POWER
;D2 = 1/(2J) WHERE J IS AVERAGE XH COUPLING AT PROTONATED SITES,
;     IE. SAME D2 AS FOR DEPT.
;P1 = 90 1H DEC. PULSE AT POWER S1
;P3,P4 = 90,180 X-NUCLEUS PULSE.

;NB: RESIDUAL PROTONATED SIGNALS APPEAR WHEN P1 OR D2 IS NOT
;    OPTIMUM.
```

RECOSY.AUR

```
; 1-Step Relayed Cosy For AMX Systems (Magnitude Mode) With Incremented Mixing
  Period To Cover A Wide Range Of Couplings (Cf. COSYRCT2.AU)
; HOMANS ET AL, PNAS USA 81, 6286 (84)

; 90-D0-90-D2-(D5)*N-180-D2-(D5)*N-90-FID   N=1,2,...NE

; CORRELATION CROSS-PEAKS CAN BE OBTAINED FROM SPINS A AND X
; IN AN AMX SYSTEM WHEN J(AX) IS TOO SMALL.

1 ZE
2 D1            ;RELAXATION DELAY
3 P1 PH1        ;90 DEG PULSE CREATES XY-MAGN.
4 D0            ;EVOLUTION OF SHIFTS
5 P1 PH2        ;COMPLETE FIRST COHERENCE TRANSFER, E.G.
                ; SPIN A TO M DEPENDS ON SIN(PI*J(AM)*D0)
6 D2            ;SECOND COHERENCE PERIOD
7 D5                    ;MIXING PERIOD INCREMENTS
8 L0 TO 7 TIMES UPR     ;WITH EXPERIMENT COUNTER
9 P2 PH2        ;REFOCUS CHEMICAL SHIFTS
10 D2
11 D5
12 L0 TO 11 TIMES UPR
13 P1 PH3       ;COMPLETE SECOND TRANSFER (EG. M TO X)
14 GO=2 PH4     ;ACQUIRE FID
15 WR #1        ;STORE FID IN .SER FILE
16 IF #1
   IU0
17 IN=1         ;LOOP FOR NEXT EXPERIMENT
18 EXIT

PH1=A0 A0 A0 A0 A0 A0 A0 A0     ;SCANS 1-2 SUPPRESS AXIAL PEAKS
    A1 A1 A1 A1 A1 A1 A1 A1     ;SCANS 3-4 FOR F1 QUAD (N-TYPE)
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH2=A0 A0 A1 A1 A2 A2 A3 A3     ;SCANS 5-8 SUPPRESS NOESY PEAKS
    A1 A1 A2 A2 A3 A3 A0 A0     ;FURTHER CYCLING FOR F2 QUAD
    A2 A2 A3 A3 A0 A0 A1 A1
    A3 A3 A0 A0 A1 A1 A2 A2
PH3=A0 A2 A1 A3 A0 A2 A1 A3
    A1 A3 A2 A0 A1 A3 A2 A0
    A2 A0 A3 A1 A2 A0 A3 A1
    A3 A1 A0 A2 A3 A1 A0 A2

PH4=R0 R0 R2 R2 R0 R0 R2 R2
    R1 R1 R3 R3 R1 R1 R3 R3
    R2 R2 R0 R0 R2 R2 R0 R0
    R3 R3 R1 R1 R3 R3 R1 R1

;COHERENCE TRANSFER PERIOD IS TWICE D2+N*D5 (N=1,2,...NE)
; TO COVER A RANGE OF J COUPLINGS, DEFINE D2=0.25/J(MAX)
; AND D5 <= ( (0.25/J(MIN)) - D2 )/NE.
; EG. FOR J(MAX) = 12, D2=21 MSEC
;     FOR J(MIN) = 2, D5 <= 104/NE MSEC
; NB. THE USEFUL UPPER LIMIT FOR D5 WILL DEPEND ON THE T2
; RELAXATION TIMES.

;NS=8*N
;P1=90, P2=180
;  OTHERWISE PARAMETERS AS FOR COSY.
;L0 = 1
;See Also 'COSYRCT.AUR'
```

RECOSY2.AUR

; COSY With 2-Step Relayed Coherence Transfer (Magnitude Mode) For AMQX Spin Systems Using Incremented Mixing Periods To Cover A Wider Range Of Couplings (Cf. COSYRCT3.AU)

```
; HOMANS ET AL, PNAS USA 81, 6286 (84)

;  90-D0-90-D2-(D5)*N-180-D2-(D5)*N-
;     90-D3-(D6)*N-180-D3-(D6)*N-90-FID    N=1,2,...NE

; CORRELATION CROSS-PEAKS CAN BE OBTAINED FOR SPIN A
; FROM SPINS M,Q,X IN AN AMQX SPIN SYSTEM.

1 ZE
2 D1            ;RELAXATION DELAY
3 P1 PH1        ;90 DEG PULSE CREATES XY-MAGN.
4 D0            ;EVOLUTION OF SHIFTS
5 P1 PH2        ;COMPLETE FIRST COHERENCE TRANSFER, E.G.
                ; SPIN A TO M DEPENDS ON SIN(PI*J(AM)*D0)
6 D2            ;SECOND COHERENCE PERIOD
7 D5                    ;MIXING PERIOD INCREMENTS WITH
8 L0 TO 7 TIMES UPR     ;EXPERIMENT COUNTER
9 P2 PH2        ;REFOCUS CHEMICAL SHIFTS
10 D2
11 D5
12 L0 TO 11 TIMES UPR
13 P1 PH3        ;COMPLETE SECOND TRANSFER (EG. M TO Q)
14 D3
15 D6
16 L0 TO 15 TIMES UPR
17 P2 PH2
18 D3
19 D6
20 L0 TO 19 TIMES UPR
21 P1 PH4       ;THIRD TRANSFER FROM Q TO X
22 GO=2 PH5     ;ACQUIRE FID
23 WR #1        ;STORE FID IN .SER FILE
24 IF #1
   IU0
25 IN=1         ;LOOP FOR NEXT EXPERIMENT
26 EXIT

PH1=A0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
                                ;SCANS 3-4 FOR F1 QUAD (N-TYPE)
                                ;SCANS 5-8 SUPPRESS NOESY PEAKS
PH2=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3

PH3=A0 A0 A2 A2 A1 A1 A3 A3

PH4=A0 A2 A0 A2 A1 A3 A1 A3

PH5=R0 R0 R0 R0 R2 R2 R2 R2

;FOR LINEAR SPIN SYSTEM AMQX, THE TRANSFER FUNCTION IS
; SIN(PI*J(AM)*2D2')SIN(PI*J(MQ)*2D2')*
; SIN(PI*J(MQ)*2D3')SIN(PI*J(QX)*2D3')
;   WHERE D2'=D2+N*D5,  D3'=D3+N*D6    N=1,2,...NE
;SET D2 = CA. 0.25/J(MAX), WHERE J(MAX) = MAX.
; EXPECTED VALUE OF J(AM), J(MQ)
;SET D3 = CA. 0.25/J(MAX), WHERE J(MAX) = MAX. EXPECTED
; VALUE OF J(MQ), J(QX).
;SET D5 <= ( (0.25/J(MIN)) - D2 )/NE, WHERE J(MIN)
; IS THE MIN. EXPECTED VALUE FOR J(AM), J(MQ)
;SET D6 <= ( (0.25/J(MIN)) - D3 )/NE, WHERE J(MIN)
```

```
; IS THE MIN. EXPECTED VALUE FOR J(MQ), J(QX)

;NS=16*N
;P1=90, P2=180
;  OTHERWISE PARAMETERS AS FOR COSY.
;L0 = 1

;SEE ALSO COSYRCT2.AUR
```

RED.AUR

; Redfield Pulse Sequence For Water Suppression Using Low-Power Transmitter With Precision Attenuator. The Excitation Pulse Has Segments With Lengths In The Approx. Ratio 2-1-4-1-2' , Where Individual Adjustment Of Each Segment Is Made To Optimize Suppression (see program 'REDSET.AU'). The Asymmetry Introduced By The Segment 2' Improves Suppression.

```
1 ZE
2 D1            ;RELAX
3 P2 PH1        ;FIRST PULSE SEGMENT OF LENGTH 2
4 P1 PH2        ;PULSE SEGMENT LENGTH 1, 180 PHASE SHIFT
5 P4 PH1        ;SEGMENT LENGTH 4
6 P1 PH2        ;SEGMENT LENGTH 1, 180 PHASE SHIFT
7 P3 PH1        ;SEGMENT LENGTH 2' (ASYMMETRY)
8 GO=2 PH3      ;ACQUIRE FID
9 EXIT

PH1=A0 A2 A2 A0 A1 A3 A3 A1
PH2=A2 A0 A0 A2 A3 A1 A1 A3
PH3=R0 R2 R2 R0 R1 R3 R3 R1

;SET O1 NEAR THE CENTER OF THE REGION OF INTEREST.
;RD=PW=0
;P1 = 0.1/(FREQ. DIFF. BETWEEN WATER AND O1)
;P2,P3,P4 = APPROX. 2*P1, 2'*P1, 4*P1
;D1+AQ = CA. T1 FOR 1H

;TRANSMITTER POWER MUST BE ADJUSTED TO GIVE CA. 45-60 DEG
;FLIP ANGLE ON-RESONANCE WITH THE SPECIFIED PULSE LENGTH.
;SEE SOFTWARE MANUALS FOR FURTHER DESCRIPTION OF TECHNIQUE
;AND CALIBRATION PROCEDURES.
```

REDNOESY.AUR

```
; Homonuclear Dipolar-Correlated 2-D NMR Dipolar Coupling May Be Due To NOE Or
  Chemical Exchange.  Using Two Soft 90 Deg Pulses And Redfield Pulse.
;  D1 - 90 - D0 - 90 - D9 - 90 - FID

; SYMMETRIC MATRIX WITH SHIFTS AND COUPLINGS IN F1, F2
; OFF-DIAGONAL PEAKS CORRELATE SPINS WHICH SHARE A
; DIPOLAR COUPLING.
; SCALAR COUPLING CORRELATIONS ARE STRONGLY REDUCED BY
; RANDOM VARIATION OF THE MIXING TIME D9.

1 ZE
2 D1            ;RELAXATION
3 P8 PH1        ;SOFT 90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
5 P8 PH2        ;SOFT MIXING PULSE
6 D9            ;MIXING TIME FOR Z-MAGN. EXCHANGE
7 P2 PH3        ;ASYMMETRIC REDFIELD DETECTION PULSE
  P1 PH5
  P4 PH3
  P1 PH5
  P3 PH3
8 GO=2 PH4      ;ACQUIRE FID
9 WR #1         ;STORE FID
10 IF #1        ;INCREMENT FILE NUMBER
11 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
12 EXIT

PH1=A0                          ;N-TYPE PEAK SELECTION
PH2=A0 A2 A1 A3
PH3=A0 A0 A1 A1 A2 A2 A3 A3
    A1 A1 A2 A2 A3 A3 A0 A0
PH5=A2 A2 A3 A3 A0 A0 A1 A1
    A3 A3 A0 A0 A1 A1 A2 A2
PH4=R0 R2 R2 R0 R2 R0 R0 R2
    R1 R3 R3 R1 R3 R1 R1 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4, 8 OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;P1,P2,P3,P4 OPTIMIZED FOR REDFIELD WATER SUPPRESSION
;P8 = SOFT 90 DEG PULSE OPTIMIZED FOR WATER SUPPRESSION
;D0 = 3E-6 INITIAL DELAY
;IN = 0.5/SW1 = 2*DW
;ND0 = 1
;I2D = 1

;D9 = MIXING TIME = CA. T1 FOR SMALL MOLECULES (EXTREME
;  NARROWING LIMIT) OR CA. 50-200 MSEC FOR LARGE MOLECULES
;  WITH CROSS-RELAXATION (SPIN-DIFFUSION).
;V9 SET TO CAUSE CA. 20 MSEC RANDOM VARIATION OF D9 TO CANCEL
;   SCALAR CORRELATION EFFECTS.
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI/4, ZERO-FILL IN F1
```

REDSET.AUR

```
; Redfield Pulse Sequence For Water Suppression (Set-Up) Using Low-Power
  Transmitter With Precision Attenuator.  The Excitation Pulse Has Segments
  With Lengths In The Approx. Ratio 2-1-4-1-2' , Where Individual Adjustment
```

Of Each Segment Is Made To Optimize Suppression. The Asymmetry Introduced By The Segment 2' Improves Suppression.

```
1 ZE
2 D1             ;RELAX
3 P2 PH1         ;FIRST PULSE SEGMENT OF LENGTH 2
4 P1 PH2         ;PULSE SEGMENT LENGTH 1, 180 PHASE SHIFT
5 P4 PH1         ;SEGMENT LENGTH 4
6 P1 PH2         ;SEGMENT LENGTH 1, 180 PHASE SHIFT
7 P3 PH1         ;SEGMENT LENGTH 2' (ASYMMETRY)
8 GS=2 PH3       ;ACQUIRE FID IN SET-UP MODE
9 EXIT

PH1=A0
PH2=A2
PH3=R0

;SET O1 NEAR CENTER OF REGION OF INTEREST.
;D1 = 1 SEC
;START SEQUENCE WITH PULSES SET FOR NOMINAL 2-1-4-1-2
;ADJUST O1 FOR BEST SUPPRESSION, THEN ADJUST P4,P2,P3 FOR
;FURTHER IMPROVEMENT, SMALL CHANGES IN P1 MAY ALSO HELP.
;DEVIATIONS FROM IDEAL REDFIELD PULSE SEGMENTS MAY BE POS.
;OR NEG. AND ARE INSTRUMENT-SPECIFIC. CHANGES IN P2,P3 ARE
;USUALLY SUCH THAT P2+P3 IS CONSTANT.

;RD=PW=0
;P1 = 0.1/(FREQ. DIFF. BETWEEN WATER AND O1)
;P2,P3,P4 = APPROX. 2*P1, 2'*P1, 4*P1

;TRANSMITTER POWER MUST BE ADJUSTED TO GIVE CA. 45-60 DEG
;FLIP ANGLE ON-RESONANCE WITH THE SPECIFIED PULSE LENGTH.
;SEE SOFTWARE MANUALS FOR FURTHER DESCRIPTION OF TECHNIQUE
;AND CALIBRATION PROCEDURES.
```

RELAY.AUR

```
; Relayed (H-H-X) Coherence Transfer 2-D.  In Addition To X-H Shift
  Correlation Via 1J(XH), Correlations From More Distant Protons Via J(HH)
  Appear. This Gives Information Also On The X-Nucleus (e.g. Carbon)
  Connectivities, But Unlike Inadequate, It Uses Only The Protonated X-Nuclei.
;P.H.BOLTON, J.MAGN.RES. 48, 336 (1982).
;A.BAX, J.MAGN.RES. 53, 149 (1983).

;  1H: DO-90-D0-   -D0-90-D2-180-D2-D3-90-   BB
;   X: D1      -180-                    90-D4-FID
; F2 DOMAIN: X-NUCLEUS SHIFTS, X-X CONNECTIVITES APPEAR IN ROWS
; F1 DOMAIN: 1H SHIFTS AND J(HH), X-H AND H-H CONNECTIVITIES APPEAR IN
 COLUMNS.

1 ZE
2 D1 S1 DO       ;RELAX, PREPARE DEC. FOR PULSING
3 P1:D PH1       ;90 DEG 1H PULSE
4 D0             ;EVOLUTION OF 1H SHIFTS AND COUPLINGS J(HH)
5 P4 PH5         ;180 DEG X, DECOUPLE X FROM 1H, REFOCUS J(XH)
6 D0             ;FURTHER EVOLUTION
7 P1:D PH2       ;90 DEG 1H, GENERATE COHERENCE BETWEEN
                 ; COUPLED PROTONS
8 D2             ;MIXING TIME, TO OPTIMIZE MODULATIONS CAUSED
                 ; BY NEXT NEAREST PROTONS
9 P2:D PH3       ;180 DEG 1H, REFOCUS 1H SHIFTS AND J(XH)
10 D2            ;WAIT FOR REFOCUSSING
11 D3            ;=1/(2J(XH)), WAIT FOR OPTIMUM POLARIZATION
12 P1:D PH4  P3 PH6   ;90 DEG X AND 1H, POLARIZATION TRANSFER
13 D4 S2         ;WAIT FOR REFOCUSSING, SET DEC. POWER
14 GO=2 PH7 CPD  ;RD=PW=0, ACQUIRE WITH CPD DEC
15 D4 DO         ;GATE DEC. OFF
16 WR #1         ;STORE FID (SERIES FILE)
17 IF #1         ;INCREMENT FILE NUMBER
18 IN=1          ;LOOP FOR NEXT EXPERIMENT
19 EXIT

PH1=B0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=B0 B2 B1 B3                 ;SCANS 3-4 GIVE F1 QUAD (COHERENCE
                                ; TRANSFER ECHO)
PH3=B0 B2 B1 B3 B2 B0 B3 B1     ;SCANS 5-8 PHASE ALTERNATE 180 PULSES
PH4=B0 B2 B1 B3 B0 B2 B1 B3     ;SCANS 9-16 PHASE ALTERNATE
    B2 B0 B3 B1 B2 B0 B3 B1     ;  POLARIZATION PULSE.
PH5=A0 A0 A0 A0 A2 A2 A2 A2
PH6=A0
PH7=R0 R2 R1 R3 R0 R2 R1 R3
    R2 R0 R3 R1 R2 R0 R3 R1

;D1>T1(H)
; 2*D2+D3 =CA. 1/(4J(HH)AVE.), IE. FOR H-H-C RELAY D2=16-25 MSEC
;     OR  =    1/(2J(HH)MAX.)
;D3=1/(2J(XH))
;D4=1/(4J(XH)) FOR ALL MULTIPLICITIES
;  =1/(2J(XH)) FOR DOUBLET MULTIP. ONLY
;D0=3E-6,  S1=0H,  S2=DESIRED DEC POWER
;P1,P2 = 90,180 1H,   P3,P4 = 90, 180 FOR X
;NS=4*N
;ND0=2, IN=0.25/SW1,  SW1= 0.5*(1H SHIFT RANGE)
```

RELAY2.AUR

; Relayed (H-H-X) Coherence Transfer 2-D, With Additional Composite X Refocussing Pulse.

```
;P.H.BOLTON, J.MAGN.RES. 48, 336 (1982).
;A.BAX, J.MAGN.RES. 53, 149 (1983).
;H.KESSLER ET AL., JACS, IN PRESS (1984).
;IN ADDITION TO X-H SHIFT CORRELATION VIA 1J(XH), CORRELATIONS
;FROM MORE DISTANT PROTONS VIA J(HH) APPEAR. THIS GIVES
;INFORMATION ALSO ON THE X-NUCLEUS (E.G. CARBON)
;CONNECTIVITIES, BUT UNLIKE INADEQUATE, IT USES ONLY THE
;PROTONATED X-NUCLEI.

;  1H: DO-90-D0-   -D0-90 -D2- 180 -     D2 -       -90-   BB
;   X: D1      -180-               -D5-180-(CA.D3/2)-90-D4-FID

; F2 DOMAIN: X-NUCLEUS SHIFTS, X-X CONNECTIVITES APPEAR IN ROWS
; F1 DOMAIN: 1H SHIFTS AND J(HH), X-H AND H-H CONNECTIVITIES
;            APPEAR IN COLUMNS.

1 ZE
2 D1 S1 DO      ;RELAX, PREPARE DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS J(HH)
5 P3 PH5        ;COMPOSITE 180 DEG X (90-240-90)
  P5 PH6        ; DECOUPLE X FROM 1H, REFOCUS J(XH)
  P3 PH5
6 D0            ;FURTHER EVOLUTION
7 P1:D PH2      ;90 DEG 1H, GENERATE COHERENCE BETWEEN
                ; COUPLED PROTONS
8 D2            ;MIXING TIME, TO OPTIMIZE MODULATIONS CAUSED
                ; BY NEXT NEAREST PROTONS
9 P2:D PH3      ;180 DEG 1H, REFOCUS 1H SHIFTS
10 (D2) (D5 P3 PH5 P5 PH6 P3 PH5)
                ;WAIT FOR REFOCUSSING OF 1H SHIFTS,
                ;COMP. 180 X PULSE  INSURES ANTI-PARALLEL J(XH)
                ;AND J(HH) VECTORS FOR OPTIMUM POLARIZATION.
12 P1:D PH4  P3 PH7   ;90 DEG X AND 1H, POLARIZATION TRANSFER
13 D4 S2        ;WAIT FOR REFOCUSSING, SET DEC. POWER
14 GO=2 PH8 CPD     ;RD=PW=0, ACQUIRE WITH CPD DEC
15 D4 DO        ;GATE DEC. OFF
16 WR #1        ;STORE FID (SERIES FILE)
17 IF #1        ;INCREMENT FILE NUMBER
18 IN=1         ;LOOP FOR NEXT EXPERIMENT
19 EXIT

PH1=B0                         ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=B0 B2 B1 B3                ;SCANS 3-4 GIVE F1 QUAD (COHERENCE
                               ; TRANSFER ECHO)
PH3=B0 B2 B1 B3 B2 B0 B3 B1    ;SCANS 5-8 PHASE ALTERNATE 180 PULSES
PH4=B0 B2 B1 B3 B0 B2 B1 B3    ;SCANS 9-16 PHASE ALTERNATE
    B2 B0 B3 B1 B2 B0 B3 B1    ;  POLARIZATION PULSE.
PH5=A0 A0 A0 A0 A2 A2 A2 A2
PH6=A1 A1 A1 A1 A3 A3 A3 A3
PH7=A0
PH8=R0 R2 R1 R3 R0 R2 R1 R3
    R2 R0 R3 R1 R2 R0 R3 R1

;D1>T1(H)
; 2*D2 =CA. 1/(4J(HH)AVE.), IE. FOR H-H-C RELAY D2=16-25 MSEC
;  OR  =    1/(2J(HH)MAX.)
;D3=1/(2J(XH))
;D4=1/(4J(XH)) FOR ALL MULTIPLICITIES
;  =1/(2J(XH)) FOR DOUBLET MULTIP. ONLY
;D5=D2 - D3/2 -(2*P3 + P5)
;D0=3E-6
```

```
;S1=0H,  S2=DESIRED DEC POWER FOR CPD
;P1,P2 = 90,180 1H,   P3,P4 = 90, 180 FOR X
;P5 = 240 DEG X-PULSE FOR COMPOSITE INVERSION PULSE.
;NS=4*N
;ND0=2, IN=0.25/SW1,  SW1= 0.5*(1H SHIFT RANGE)
```

REVCOR.AUR

; Reverse X-H Shift Correlation 2-D Observe 1H Coupled To X-Nucleus Using BSV-3 BX Heteronuclear Decoupler And Synthesizer With 90 Deg Phase Shifter.
(For F1 Quad Detection)

```
;  TRANSM. 1H: D1-90-D2-     -180-     -FID
;     DEC.  X:         -90-D0-   -D0-90-DO

;  F2 DOMAIN:  1H SHIFTS, J(HH),  J(XH)
;  F1 DOMAIN:  X SHIFTS, J(HH)

1 ZE
2 D1 CW DO      ;RELAX, SET DEC. TO CW STATUS FOR PULSING
3 P1 PH1        ;90 DEG 1H PULSE
4 D2            ;=1(2J(XH)), CREATE ANTI-PHASE XH DOUBLET
5 P3:D PH3      ;90 DEG X PULSE, CREATE DBL. QUANTUM COHERENCE
6 D0            ;EVOLUTION OF ALL SHIFTS AND COUPLINGS
7 P2 PH2        ;180 DEG 1H, TO REFOCUS 1H SHIFTS AND J(XH)
8 D0            ;FURTHER EVOLUTION OF X SHIFTS AND J(HH)
9 P3:D PH4      ;90 DEG X, RECONVERSION
10 GO=2 PH5     ;DETECT 1H, ANTI-PHASE XH DOUBLET
                ; WITHOUT X DEC., MODULATED BY X SHIFTS
11 WR #1        ;STORE FID
12 IF #1        ;INCREMENT FILE NUMBER
13 IN=1         ;LOOP FOR NEXT EXPERIMENT
14 EXIT

PH1=A0
PH2=A0 A0 A0 A0 A2 A2 A2 A2 A1 A1 A1 A1 A3 A3 A3 A3
PH3=B0
PH4=B0 B2 B1 B3
PH5=R0 R2 R1 R3 R0 R2 R1 R3 R2 R0 R3 R1 R2 R0 R3 R1

;PHASE CYCLE REJECTS 1H NOT COUPLED TO X, GIVES QUAD DETECTION
;IN F1 (SYNTH IN MIDDLE OF X SPECTRUM)
; D1 > T1(1H) , D2=1(2J(XH))
;P1,P2 = 90, 180 FOR 1H
;P3=90 FOR X
;NS=4*N
;RD=PW=0
;D0=3E-6
;ND0=2, SW1=0.5*(X SHIFT RANGE), IN=0.25/SW1
;NE=TD1, NUMBER OF FIDS

;USE XHSEL.AU TO CALIBRATE X-NUCLEUS 90 DEG PULSE

;WHEN 90 DEG DEC PHASE N O T AVAILABLE USE FOLLOWING
;PHASES (SET SYNTH. TO HIGH FIELD SIDE OF X SPECTRUM,
;  SW1 = X SHIFT RANGE).
;PH1=A0 A0 A0 A0 A0 A0 A0 A0 A1 A1 A1 A1 A1 A1 A1 A1
;PH2=A0 A0 A2 A2 A1 A1 A3 A3 A1 A1 A3 A3 A2 A2 A0 A0
;PH3=B0
;PH4=B0 B2
;PH5=R0 R2 R0 R2 R2 R0 R2 R0 R1 R3 R1 R3 R3 R1 R3 R1
```

REVCORD.AUR

```
; Reverse X-H Shift Correlation 2-D Observe 1H Decoupled From X-Nucleus Using
  BSV-3 BX Heteronuclear Decoupler And Synthesizer With 90 Deg Phase Shifter
; (For F1 Quad Detection).  Requires Modification For Computer Control Of
  CW/BB In BSV-3 Or Use CPD Dec.

;  TRANSM. 1H: D1-90-D2-     -180-     -FID
;     DEC.  X:         -90-D0-   -D0-90-CPD

;  F2 DOMAIN:  1H SHIFTS, J(HH)
;  F1 DOMAIN:  X SHIFTS, J(HH)

1 ZE
2 D1 CW DO      ;RELAX, SET DEC. TO CW STATUS FOR PULSING
3 P1 PH1        ;90 DEG 1H PULSE
4 D2            ;=1(2J(XH)), CREATE ANTI-PHASE XH DOUBLET
5 P3:D PH3      ;90 DEG X PULSE, CREATE DBL. QUANTUM COHERENCE
6 D0            ;EVOLUTION OF ALL SHIFTS AND COUPLINGS
7 P2 PH2        ;180 DEG 1H, TO REFOCUS 1H SHIFTS AND J(XH)
8 D0            ;FURTHER EVOLUTION OF X SHIFTS AND J(HH)
9 P3:D PH4      ;90 DEG X, RECONVERSION
  D2            ;WAIT FOR REPHASING OF XH DOUBLET
10 GO=2 PH5 CPD      ;DETECT 1H WITH X DECOUPLING
                ; MODULATED BY X SHIFTS
   D2 CW DO     ;GATE DEC. OFF
11 WR #1        ;STORE FID
12 IF #1        ;INCREMENT FILE NUMBER
13 IN=1         ;LOOP FOR NEXT EXPERIMENT
14 EXIT

PH1=A0
PH2=A0 A0 A0 A0 A2 A2 A2 A2 A1 A1 A1 A1 A3 A3 A3 A3
PH3=B0
PH4=B0 B2 B1 B3
PH5=R0 R2 R1 R3 R0 R2 R1 R3 R2 R0 R3 R1 R2 R0 R3 R1

;PHASE CYCLE REJECTS 1H NOT COUPLED TO X, GIVES QUAD DETECTION
;IN F1 (SYNTH IN MIDDLE OF X SPECTRUM)
; D1 > T1(1H) , D2=1(2J(XH))
;P1,P2 = 90, 180 FOR 1H
;P3=90 FOR X = P9 WHEN USING CPD FOR COMP. PULSE DEC.
;NS=4*N
;RD=PW=0
;D0=3E-6
;ND0=2, SW1=0.5*(X SHIFT RANGE), IN=0.25/SW1
;NE=TD1, NUMBER OF FIDS

;USE XHSEL.AU TO CALIBRATE X-NUCLEUS 90 DEG PULSE
;  AT POWER LEVEL USED FOR DECOUPLING

;WHEN 90 DEG DEC PHASE N O T AVAILABLE USE FOLLOWING
;PHASES (SET SYNTH. TO HIGH FIELD SIDE OF X SPECTRUM, SW1 = X SHIFT RANGE).
;PH1=A0 A0 A0 A0 A0 A0 A0 A0 A1 A1 A1 A1 A1 A1 A1 A1
;PH2=A0 A0 A2 A2 A1 A1 A3 A3 A1 A1 A3 A3 A2 A2 A0 A0
;PH3=B0
;PH4=B0 B2
;PH5=R0 R2 R0 R2 R2 R0 R2 R0 R1 R3 R1 R3 R3 R1 R3 R1
```

ROESYPC.AUR

```
; 2D ROESY With CW Spinlock For Mixing
;For O1/O2-Coherence This Requires Special Directional Coupler (10Db Loss On
 F2)
;Phase Sensitive Using Tppi
```

```
;Lit: A. Bax & D.G. Davis, J. Magn. Reson. 63, 207-213 (1985)

1 ZE
2 D1 S1 DO                  ;relaxation delay
  P1 PH1                    ;90 deg transmitter H-1 pulse
  D0                        ;t1
  D2 PH2
  P2:D PH2                  ;CW spinlock
  GO=2 PH3
  WR #1
  IF #1
  IP1
  IN=1
EXIT

PH1=0 2 2 0 1 3 3 1
PH2=0 2 0 2 1 3 1 3
PH6=R0 R2 R2 R0 R1 R3 R3 R1


;D1 : 1-5 T1
;S1 : power level for spinlock (90 deg pulse : 100 - 125 usec)
;P1 : 90 deg H-1 transmiter pulse
;D0 = 3 usec
;D2 = 2 usec
;P2 : spinlock time (150 - 250 msec)
;NS : 8 * n
;DS : 2 or 4
;MC2 = W
;ND0 = 2
;IN : DW (H-1)
```

SCOSYPH.AUR

```
; Soft 2D-COSY Using Shaped Pulse For Excitation
; O1 In The Middle Of Region To Be Exciteted During T1 And T2

;R. Brueschweiler, J.C. Madsen, C. Griesinger, O.W. Soerensen,
;   R.R. Ernst, J. Magn. Reson. 73, 380 - 385 (1987)

  II
1 WAVE                         ;load waveform to waveform-memory
2 ZE
3 D1                           ;relaxation delay
4 P1:W1 PH1                    ;90 deg H-1 soft pulse
  D0                           ;t1
  P2 PH2                       ;90 deg H-1 pulse (transmitter)
5 GO=3 PH3
  WR #1
  IF #1
  IP1
  IN=1
6 EXIT

PH1=0 0 2 2 1 1 3 3
PH2=0 2 2 0 1 3 3 1
PH3=R0 R0 R2 R2 R1 R1 R3 R3


;define waveform and amplitude in forground (W1, WPn)
;D1 : 1-5 T1
;P1 : 90 deg H-1 soft pulse
;D0 = 3 usec
;P2 : 90 deg H-1 pulse transmitter
;NS : 4 * n
;DS : 2 or 4
;IN = DW
;ND0 = 2

;large phase correction constants required in F1
```

SCOSYPR.AUR

```
; Soft 2D-COSY Using Shaped Pulse For Excitation With Refocussing To Prevent
  Large Phase-Correction Constants In F1
O1 In The Middle Of Region To Be Exciteted During T1 And T2
;R. Brueschweiler, J.C. Madsen, C. Griesinger, O.W. Soerensen,
;   R.R. Ernst, J. Magn. Reson. 73, 380 - 385 (1987)

  II
1 WAVE                          ;load waveform to waveform-memory
2 ZE
3 D1                            ;relaxation delay
4 P1:W1 PH1                     ;90 deg H-1 soft pulse
  D4
  P3 PH4                        ;180 deg H-1 pulse (transmitter)
  D2
  D0                            ;t1
  P2 PH2                        ;90 deg H-1 pulse (transmitter)
5 GO=3 PH3
  WR #1
  IF #1
  IP1
  IP4
  IN=1
6 EXIT

PH1=0 2 0 2 1 3 1 3
PH4=0 2 0 2 1 3 1 3 2 0 2 0 3 1 3 1
PH2=0 2 2 0 1 3 3 1
PH3=R0 R2 R0 R2 R1 R3 R1 R3


;define waveform and amplitude in forground (W1, WPn)
;D1 : 1-5 T1
;P1 : 90 deg H-1 soft pulse
;P3 : 180 deg H-1 transmitter pulse
;D2 = P1/2 + D4
;D0 = 3 usec
;P2 : 90 deg H-1 transmitter pulse
;NS : 4 * n
;DS : 2 or 4
;IN = DW
;ND0 = 2
```

SDDS.AUR

```
; Spin Decoupling Difference Spectroscopy Using A Freq. List To Define Several
  Irradiation Points (On-Resonance) And One Control (Off-Resonance).
; The Individual Fids Are Stored.
; For Long-Term Averaging The Routine Cycles Through The
; Freq. List And Fids Several Times.

1 FL #1         /INPUT FREQ. LIST
                ;READ FREQ. LIST (EXT .001) AND INITIALIZE POINTER
2 ZE
  WR #2         /INPUT FID
                ;PREPARE A SET OF ZEROED FILES ON DISK
3 IF #2
4 LO TO 2 TIMES C  ;C= NO. OF FIDS TO BE STORED

5 RF #2.001     ;RESET FILE EXTENSION TO .001, BEGIN CYCLE
6 RE #2         ;READ CURRENT FID FILE
7 D1 O2 S3      ;SET DEC. FREQ. O2 FROM CURRENT FL LIST,
                ; INCREMENT POINTER, SET POWER S3.
8 D2 DO         ;RELAX. TIME WITH DEC. OFF (FOR NO NOE)
9 D3 HD         ;RELAX. TIME WITH DEC. ON (WITH NOE)
10 GO=8         ;ACQUIRE DATA WITH DEC. ON, LOOP TO 8
11 WR #2        ;STORE CURRENT ACCUMULATED FID
12 IF #2        ;INCREMENT FID EXTENSION
13 LO TO 6 TIMES C      ;LOOP TO 6 FOR EACH FREQ. IN FL LIST
14 IN=5         ;LOOP FOR ANOTHER CYCLE
                ;NE=# OF CYCLES THROUGH LIST
15 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS
;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS ONE O2
;ENTRY FOR EACH DESIRED IRRAD. POINT PLUS ONE OFF-RES. CONTROL
;VALUE FOR O2 WHICH SHOULD BE WITHIN THE SW REGION (E.G. AT ONE
;EDGE OF THE SPECTRUM). THE NUMBER OF FREQ. IN THE LIST MUST BE
;DEFINED BY AN ENTRY IN A 'VC' LIST, WHICH ALSO DEFINES THE
;NUMBER OF FIDS TO BE STORED.

;NS DEFINES THE NO. OF TRANSIENTS PER CYCLE FOR EACH O2 VALUE
;AND SHOULD BE A MULTIPLE OF 8.

;NE DEFINES THE NO. OF CYCLES TO BE MADE THROUGH COMPLETE LIST.
;TOTAL TRANSIENTS PER FID = NE*NS.
;USE 2-4 DUMMY SCANS FOR STEADY-STATE!
;RD=0
;D1 = 0.5 SEC TO SET O2
;D2+D3+AQ IS THE TOTAL RELAXATION TIME. IF STEADY-STATE NOE
;  CAN BE TOLERATED, SET D2=5 USEC AND D3+AQ=1-5*T1;
;  IF DECOUPLING WITH MINIMUM
;  NOE IS DESIRED SET D2=1-5*T1 AND D3 AS SMALL AS POSSIBLE
;  (1-10 MSEC) TO AVOID NOE BUT MINIMIZE TRANSIENT EFFECTS THAT
;  OCCUR WHEN DECOUPLER IS TURNED ON.
;S3 DEFINES DEC. POWER TYPICALLY 10-40L DEPENDING ON REQUIRED
;IRRAD. BANDWIDTH.
```

SECSY.AUR

```
; Homonuclear Spin-Echo Shift-Correlated 2-D Nmr
; K.NAGAYAMA ET AL., J.MAGN.RES. 40, 321 (1980)

;  D1 - 90 - D0 -D2- 90 -D2- D0 - FID

; HORIZONTAL MATRIX WITH SHIFTS AND COUPLINGS IN  F2
; ONE-HALF SHIFT DIFFERENCES AND COUPLINGS IN F1.
; CORRELATION PEAKS LIE ABOVE AND BELOW F1=0 AXIS.
; AN ALTERNATIVE TO COSY; ADVANTAGEOUS ONLY WHEN
; CORRELATED SPINS LIE IN REGIONS MUCH SMALLER THAN TOTAL SW.

1 ZE
2 D1            ;RELAXATION
3 P1 PH1        ;90 DEG EXCITATION PULSE
4 D0            ;EVOLUTION OF SHIFTS AND COUPLINGS
  D2            ;FIXED DELAY TO ENHANCE THE EFFECT OF SMALL J
5 P2 PH2        ;MIXING PULSE, NORMALLY 90  DEG
  D2
6 D0            ;WAIT FOR 'ECHO'
7 GO=2 PH3      ;ACQUIRE FID
8 WR #1         ;STORE FID
9 IF #1         ;INCREMENT FILE NUMBER
10 IN=1         ;INCREMENT D0 AND LOOP FOR NEXT EXPER.
11 EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1
    A2 A2 A2 A2 A3 A3 A3 A3

PH2=A0 A2 A1 A3 A1 A3 A2 A0
    A1 A3 A2 A0 A2 A0 A3 A1

PH3=R0 R0 R2 R2 R1 R1 R3 R3

;PROGRAM REQUESTS FILENAME WITH .SER EXTENSION
;NE DEFINES NUMBER OF FIDS = TD1
;NS = 4,8, OR 16 (COMPLETE PHASE CYCLE)
;DS = 2 OR 4
;RD=PW=0
;D1 = 1-5*T1
;D2 CAN BE 0.05 - 0.1 SEC TO ENHANCE EFFECTS OF SMALL J WHEN
;  HZ/PT > 3, SEE COSYLR.AU.
;P1 = 90 DEG
;P2 = 90 DEG FOR MAX. SENSITIVITY
;D0 = 3E-6 INITIAL DELAY
;IN = 0.25/SW1
;SW1 < SW/2 DEPENDING ON MAX. SHIFT DIFFERENCE FOR
;  COUPLED PAIRS OF SPINS.
;ND0 = 2

;N-TYPE PEAK SELECTION
;TYPICALLY USE TD = SI, NO ZERO-FILLING IN F2
;              NE = SI1/2, ZERO-FILL IN F1
```

SELCOSY.AUR

```
; 1D Cosy Using Selective Excitation With A Shaped Pulse
;C.J. Bauer, R Freeman, T. Frenkiel, J. Keeler & A.J. Shaka,
;   J. Magn. Reson. 58, 442 (1984)
;H. Kessler, H. Oschkinat, C. Griesinger & W. Bermel,
;   J. Magn. Reson. 70, 106 (1986)


  WAVE                   ;load waveform to waveform-memory
1 ZE
2 D1                     ;relaxation delay
  P1:W1 PH1              ;shaped pulse
  D2                     ;evolution time
  P2 PH2                 ;90 deg H-1 pulse
  GO=2 PH3
EXIT

PH1=(360)
;PH1=(1 3 3 1 2 0 0 2) + x

PH2=0 2 0 2 1 3 1 3
PH3=R0 R2 R2 R0 R1 R3 R3 R1


;define waveform (W1) and amplitude (WPn) in forground
;x : phasedifference between transmitter and SEU output
;choose P1 according to desired selectivity
;the flip-angle is determined by the amplitude
;the use of an external attenuator might be necessary
;O1 has to be on resonance on the multiplet to be excited

;D1 : 1-5 T1
;D2 : P1/2 + D2 = 1/(2J)
;P2 : 90 deg H-1 pulse
;NS : 8 * n
;DS : 2 or 4
```

SELCOZF.AUR

```
; 1D COSY With Z-Filter Using Selective Excitation With A Shaped Pulse
;H. Kessler, H. Oschkinat, C. Griesinger & W. Bermel,
;   J. Magn. Reson 70, 106 (1986)

 WAVE                       ;load waveform to waveform-memory
1 ZE
2 D1                        ;relaxation delay
3 P1:W1 PH1                 ;shaped pulse
4 D2                        ;evolution time
5 P2 PH2                    ;90 deg H-1 pulse
  D3                        ;1/(4J)
  P3 PH4                    ;180 deg H-1 pulse
  D3
  P2 PH5                    ;90 deg H-1 pulse
  D9                        ;z-filter
  P2 PH6                    ;90 deg H-1 pulse
6 GO=2 PH3
7 IN=2
EXIT


PH1=(360)
;PH1= (1) + x

PH2=0 2 0 2
PH3=R0 R0 R0 R0 R2 R2 R2 R2 R1 R1 R1 R1 R3 R3 R3 R3
    R2 R2 R2 R2 R0 R0 R0 R0 R3 R3 R3 R3 R1 R1 R1 R1
PH4=0 0 2 2 0 0 2 2 2 2 0 0 2 2 0 0
    0 0 2 2 0 0 2 2 2 2 0 0 2 2 0 0
    1 1 3 3 1 1 3 3 1 1 3 3 1 1 3 3
    1 1 3 3 1 1 3 3 1 1 3 3 1 1 3 3
PH5=1 1 1 1 3 3 3 3 1 1 1 1 3 3 3 3
    1 1 1 1 3 3 3 3 1 1 1 1 3 3 3 3
    3 3 3 3 1 1 1 1 3 3 3 3 1 1 1 1
    3 3 3 3 1 1 1 1 3 3 3 3 1 1 1 1
PH6=0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1
    2 2 2 2 2 2 2 2 3 3 3 3 3 3 3 3

;define waveform (W1) and amplitude (WPn) in forground
;x : phasedifference between transmitter and SEU output
;choose P1 according to desired selectivity
;the flip-angle is determined by the amplitude
;the use of an external attenuator might be necessary
;O1 has to be on resonance on the multiplet to be excited

;D1 : 1-5 T1
;D2 : P1/2 + D2 = 1/(2J)
;P2 : 90 deg H-1 pulse
;D3 : 1/(4J)HH
;P3 : 180 deg H-1 pulse
;D9 : variable delay for z-filter (1 - 10 msec)
;V9 : 50 - 90
;NS : 32 * n
;DS : 2 or 4
;NE : >= 4
```

SELEXC.AUR

```
; Selective Excitation Using A Shaped Pulse
;C.J. Bauer, R Freeman, T. Frenkiel, J. Keeler & A.J. Shaka,
;   J. Magn. Reson. 58, 442 (1984)
;H. Kessler, H. Oschkinat, C. Griesinger & W. Bermel,
;   J. Magn. Reson. 70, 106 (1986)


  WAVE                  ;load waveform to waveform-memory
1 ZE
2 D1                    ;relaxation delay
  P1:W1 PH1             ;shaped pulse
  GO=2 PH2
EXIT

PH1=(360)
;PH1=(0 2 2 0 1 3 3 1) + x

PH2=R0 R2 R2 R0 R1 R3 R3 R1


;define waveform (W1) and amplitude (WPn) in forground
;x : phasedifference between transmitter and SEU output
;choose P1 according to desired selectivity
;the flip-angle is determined by the amplitude
;the use of an external attenuator might be necessary
;O1 has to be on resonance on the multiplet to be excited

;D1 : 1-5 T1
```

SELJRES.AUR

; Selective X-H J-Resolved 2-D. A Low-Power Selective 1h 180 Deg Pulse (Spin-Flip) Is Used To Create A X-H J-Resolved 2-D, Where The J Dimension Shows Only Coupling Effects From The 1h Spin That Is Flipped. When Low-Power Is Used, The One-Bond J Is Also Suppressed. Method Suffers From The Artefacts Of The Spin-Flip Technique, Not Good For Strongly Coupled Protons.

```
; SEE BAX & FREEMAN, JACS 104, 1099 (82).

;       1H: BB -   -DO-   CW-  -CW-DO-   BB
;        X: D1 - P1-D3-D0-D2-P2-D2-D3-D0-FID

;F2 DOMAIN: BB-DECOUPLED X-NUC.
;F1 DOMAIN: SELECTIVE J(XH)

1 ZE
2 D1 S1 O2 CPD   ;RELAX, GENERATE NOE WITH O2 IN MIDDLE OF 1H
3 P1 PH1        ;90 DEG X PULSE, GATE DEC. OFF
4 D3 DO S3 O2   ;FIRST EVOLUTION PERIOD
                ; SET POWER AND O2 FOR SELECTED 1H SIGNAL
  D0
5 D2 CW         ;FIRST HALF OF SELECTIVE 1H PULSE
6 P2 PH2        ;180 X REFOCUSSING PULSE (DEC. STILL ON)
7 D2 CW         ;SECOND HALF OF 1H PULSE
8 D3 DO S2 O2   ;SECOND EVOLUTION PERIOD, DEC. GATED OFF
                ; AND PREPARED FOR DECOUPLING DURING ACQ.
  D0
9 GO=2 PH3 CPD  ;ACQUIRE FID WITH CPD DECOUPLING
10 D3 S1 CPD    ;SWITCH TO LOWER DEC. POWER
11 WR #1        ;WRITE FID TO SERIES FILE
12 IF #1        ;INCREMENT FILE EXTENSION NUMBER
13 IN=1         ;LOOP FOR NEXT FID
14 EXIT         ; EXIT WITH DECOUPLING FOR NOE

PH1=A0 A0 A0 A0 A1 A1 A1 A1 A2 A2 A2 A2 A3 A3 A3 A3
PH2=A0 A2 A1 A3 A1 A3 A2 A0 A1 A3 A2 A0 A2 A0 A3 A1
PH3=R0 R0 R2 R2 R1 R1 R3 R3

;A FREQ. LIST FOR O2 IS REQUIRED WITH 3 ENTRIES:
;       1) O2 FOR NOE GENERATION WITH BB DEC.
;       2) O2 FOR SELECTIVE SPIN-FLIP
;       3) O2 FOR BB DEC. (SAME AS (1))

;D0=3 USEC
;D1= 1-3*T1 FOR X
;P1,P2 = 90,180 DEG FOR X
;D3=1-2 MSEC FOR DECOUPLER SETTINGS
;D2= EG.10 MSEC FOR GAMMA*H2=25 HZ,  D2=0.25/(GAMMA*H2)
;S1,S2 AS FOR POWER-GATED CPD DECOUPLING
;S3 MUST BE CHOSEN TO GIVE SELECTIVE 1H 180 FLIP DURING
;  TIME 2*D2+P2.
;RD=PW=0
;NS=4,8, OR 16*N, DS=2 OR 4
;ND0=2, IN=0.25/SW1, SW1 CHOSEN FOR EXPECTED COUPLINGS.

;TO CALIBRATE DEC. PULSE: SELECT TEST SUBSTANCE WITH KNOWN
;J(XH) FOR A GIVEN X SIGNAL. GAMMA*H2 SHOULD BE CA. 2*J,
; SET D2 ACCORDINGLY, THEN SET D0=(0.25/J)-D2-D3, SET S3 MUCH
; LOWER THAN NEEDED AND PERFORM ONE EXPERIMENT, WHICH SHOULD
; GIVE NORMAL UNMODULATED X SIGNAL. INCREASE S3 UNTIL X SIGNAL
; NULLS IN BB DEC. SPECTRUM OR GIVES ANTIPHASE DOUBLET WITHOUT
; DECOUPLING (OPTIMUM 180 FLIP). FOR GAMMA*H2 = 25HZ A VALUE
; OF 20L-30L FOR S3 WILL BE TYPICAL.

;NB: REPEATED SWITCHING BETWEEN LOW AND HIGH POWER DEC. MODES
```

```
; WILL DECREASE LIFETIME OF RELAYS!
```

SELNOE1D.AUR

```
; 1D NOESY Using Selective Excitation With A Shaped Pulse
;H. Kessler, H. Oschkinat, C. Griesinger & W. Bermel,
;   J. Magn. Reson. 70, 106 (1986)

  WAVE                       ;load waveform to waveform-memory
1 ZE
2 D1                         ;relaxation delay
3 P1:W1 PH1                  ;shaped pulse
4 D2                         ;evolution time
5 P2 PH2                     ;90 deg H-1 pulse
6 D9                         ;mixing time
7 P2 PH3
8 GO=2 PH4
9 IN=2
EXIT

PH1=(360)
;PH1= (0 2) + x

PH2=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
PH3=0 0 2 2 1 1 3 3
PH4=R0 R2 R2 R0 R1 R3 R3 R1
    R2 R0 R0 R2 R3 R1 R1 R3


;define waveform (W1) and amplitude (WPn) in forground
;x : phasedifference between transmitter and SEU output
;choose P1 according to desired selectivity
;the flip-angle is determined by the amplitude
;the use of an external attenuator might be necessary
;O1 has to be on resonance on the multiplet to be excited

;D1 : 1-5 T1
;D2 : optimize for maximum in-phase magnetization
;P2 : 90 deg H-1 pulse
;D9 : mixing time (like in 2D-NOESY)
;NS : 16 * n
;DS : 2 or 4
;NE : >= 4
```

SELNOREL.AUR

```
; 1D Relayed NOESY Using Selective Excitation With A Shaped Pulse
;H. Kessler, H. Oschkinat, C. Griesinger & W. Bermel,
;   J, Magn. Reson. 70, 106 (1986)

  WAVE                        ;load waveform to waveform-memory
1 ZE
2 D1                          ;relaxation delay
3 P1:W1 PH1                   ;shaped pulse
4 D2                          ;evolution time
5 P2 PH2                      ;90 deg H-1 pulse
  D3                          ;1/(4J)
  P3 PH4                      ;180 deg H-1 pulse
  D3
  P2 PH5                      ;90 deg H-1 pulse
  D9                          ;mixing time
  P2 PH6                      ;90 deg H-1 pulse
6 GO=2 PH3
7 IN=2
EXIT

PH1=(360)
;PH1= (1) + x

PH2=0 2 0 2
PH3=R0 R0 R0 R0 R2 R2 R2 R2 R1 R1 R1 R1 R3 R3 R3 R3
    R2 R2 R2 R2 R0 R0 R0 R0 R3 R3 R3 R3 R1 R1 R1 R1
PH4=0 0 2 2 0 0 2 2 2 2 0 0 2 2 0 0
    0 0 2 2 0 0 2 2 2 2 0 0 2 2 0 0
    1 1 3 3 1 1 3 3 1 1 3 3 1 1 3 3
    1 1 3 3 1 1 3 3 1 1 3 3 1 1 3 3
PH5=1 1 1 1 3 3 3 3 1 1 1 1 3 3 3 3
    1 1 1 1 3 3 3 3 1 1 1 1 3 3 3 3
    3 3 3 3 1 1 1 1 3 3 3 3 1 1 1 1
    3 3 3 3 1 1 1 1 3 3 3 3 1 1 1 1
PH6=0 0 0 0 0 0 0 0
    1 1 1 1 1 1 1 1
    2 2 2 2 2 2 2 2
    3 3 3 3 3 3 3 3

;define waveform (W1) and amplitude (WPn) in forground
;x : phasedifference between transmitter and SEU output
;choose P1 according to desired selectivity
;the flip-angle is determined by the amplitude
;the use of an external attenuator might be necessary
;O1 has to be on resonance on the multiplet to be excited

;D1 : 1-5 T1
;D2 : P1/2 + D2 = 1/(2J)
;P2 : 90 deg H-1 pulse
;D3 : 1/(4J)HH
;P3 : 180 deg H-1 pulse
;D9 : mixing time (like in 2D NOESY)
;NS : 16 * n
;DS : 2 or 4
;NE : >= 4
```

SELREL1D.AUR

```
; 1D Relayed COSY Using Selective Excitation With A Shaped Pulse
;H. Kessler, H. Oschkinat, C. Griesinger & W. Bermel,
;   J. Magn. Reson. 70, 106 (1986)


  WAVE                      ;load waveform to waveform-memory
1 ZE
2 D1                        ;relaxation delay
3 P1:W1 PH1                 ;shaped pulse
4 D2                        ;evolution time
5 P2 PH2                    ;90 deg H-1 pulse
  D3                        ;second evolution time
  P3 PH4                    ;180 deg H-1 pulse
  D3
  P2 PH5                    ;90 deg H-1 pulse
6 GO=2 PH3
EXIT


PH1=(360) 295 115 25 205 115 295 205 25
;PH1= (1 3 2 0 3 1 0 2) + x

PH2= 0 0 1 1 2 2 3 3
     2 2 3 3 0 0 1 1
PH3= R0 R2 R1 R3 R2 R0 R3 R1
PH4= 0 0 1 1 2 2 3 3
PH5= 0 0 1 1 2 2 3 3
     0 0 1 1 2 2 3 3
     2 2 3 3 0 0 1 1
     2 2 3 3 0 0 1 1


;define waveform (W1) and amplitude (WPn) in forground
;x : phasedifference between transmitter and SEU output
;choose P1 according to desired selectivity
;the flip-angle is determined by the amplitude
;the use of an external attenuator might be necessary
;O1 has to be on resonance on the multiplet to be excited


;D1 : 1-5 T1
;D2 : P1/2 + D2 = 1/(2J)
;P2 : 90 deg H-1 pulse
;D3 : optimize for second magnetization transfer
;P3 : 180 deg H-1 pulse
;NS : 32 * n
;DS : 2 or 4
```

SEUP90.AU

;Determination Of Amplitude For A 90 Degree Shaped Pulse

```
  RF SEUC.001
1 RC SEUC                    ;read constant file for WP1
  IF SEUC
  WAVE                       ;load waveform to waveform-memory
2 ZE
3 D1                         ;relaxation delay
  P1:W1 PH1                  ;shaped pulse
  GO=3
  WR #1
  IF #1
  IN=1
5 EXIT

PH1=(360) 0

;D1 : 3-5 T1
;NS = 1
;DS = 0
;NE = 10

;set O1 on resonance on a singulet line
;choose P1 according to desired selectivity
;define all parameters for this microprogram
;define W1 with : multiplier 1
;run SEUCONST.EXE to create constants-files (SEUC.*) with
;   WP1 = 10, 20, 30, ..., 100
;select external attenuation of rf power
;start this microprogram
```

SFDEC.AU

; Single Freq. CW Het.-Nuc. Decoupling With Power Gating For Generation Of NOE.

```
1 ZE
2 D2 O2         ;SET DECOUPLER FREQ. O2 FROM CURRENT FL LIST
3 D1 CPD S1     ;TURN ON CPD DEC. WITH POWER S1 FOR NOE
4 D2 CW S2      ;SWITCH TO CW DEC. WITH DESIRED POWER S2
5 GO=3          ;ACQUIRE DATA AND LOOP TO 3
6 WR #1         ;STORE FID
7 IF #1         ;INCREMENT FILE EXTENSION
8 IN=1          ;LOOP FOR NEXT EXPERIMENT
9 D2 DO
10 EXIT         ;EXIT WITH DEC. OFF

;PROGRAM REQUESTS FILENAME AT EXECUTION
;NE DEFINES NUMBER OF ITEMS IN FL LIST =NO. OF EXPERIMENTS
;D1 = TIME TO GENERATE NOE, TYPICALLY 2-4*(AQ OR T1)
;D2 = 5-10 MSEC
;S1 = CA. 0.5 WATT TO GIVE NOE
;S2 = LOWER POWER FOR SELECTIVE DEC., DEPENDS ON J(XH)
;RD=0
```

SFOR.AU

; Single-Freq. Off-Resonance Decoupling Using A Freq. List

```
1 FL #1         /DEFINE FREQ. LIST
                ;READ IN FREQ. LIST
2 ZE            ;ZERO MEMORY
  D1 O2 CW S1   ;TURN ON CW DEC. AT OFFSET O2 FROM FL LIST
                ;USING POWER S1
3 GO=3          ;ACQUIRE DATA
4 WR #2         /DEFINE FID
                ;STORE FID
5 IF #2         ;INCREMENT FILE EXTENSION
6 IN=2          ;LOOP FOR NEXT EXPERIMENT, NE TIMES
7 EXIT          ;EXIT WITH DEC. ON

;PROGRAM ASKS FOR FILENAMES FOR FREQ. LIST (#1) AND FIDS (#2).
;NE DEFINES NUMBER OF ITEMS IN FL = NO. OF EXPERIMENTS
;D1 TYP. 1 SEC TO SET FREQ.
;S1 DEFINES DEC. POWER
;USE RD AS DESIRED
```

SPT.AUR

; Selective Population Transfer (Homo- Or Heteronuc.)

```
; USING A FREQ. LIST TO DEFINE SEVERAL DEC. IRRADIATION POINTS
; (ON-RESONANCE) AND ONE CONTROL (OFF-RESONANCE)
; THE INDIVIDUAL FIDS ARE STORED.
; FOR LONG-TERM AVERAGING THE ROUTINE CYCLES THROUGH THE
; FREQ. LIST AND FIDS SEVERAL TIMES.
; ALSO CAN BE USED FOR PSEUDO-INDOR.

1 ZE
2 WR #1         /DEFINE FID
                ;PREPARE A SET OF ZEROED FILES ON DISK
3 IF #1
4 LO TO 2 TIMES C  ;C= NO. OF FIDS TO BE STORED
  FL #2         /DEFINE FREQ. LIST
                ;READ IN FREQ. LIST
5 RF #1.001     ;RESET FILE EXTENSION TO .001, BEGIN CYCLE
6 RE #1         ;READ CURRENT FID FILE
7 D3 O2 S3      ;SET DEC. FREQ. O2 FROM CURRENT FL LIST
8 D1 DO         ;RELAX. TIME WITH DEC. GATED OFF
9 P1:D          ;PULSE DEC. (CA. 180 DEG FLIP) USING POWER S3
10 GO=8         ;ACQUIRE DATA WITH DEC. OFF, LOOP TO 8
11 WR #1        ;STORE CURRENT ACCUMULATED FID
12 IF #1        ;INCREMENT FID EXTENSION
13 LO TO 6 TIMES C      ;LOOP TO 6 FOR EACH FREQ. IN FL LIST
14 IN=5         ;LOOP FOR ANOTHER CYCLE
                ;NE=NUMBER OF CYCLES THROUGH LIST
15 EXIT

;PROGRAM REQUESTS FILENAME FOR FIDS (#1) AND FREQ. LIST (#2).
;A FREQ. LIST MUST BE DEFINED WHICH CONTAINS ONE O2
;ENTRY FOR EACH DESIRED IRRAD. POINT PLUS ONE OFF-RES. CONTROL
;VALUE FOR O2 WHICH SHOULD BE WITHIN THE SW REGION (E.G. AT ONE
;EDGE OF THE SPECTRUM). THE NUMBER OF FREQ. IN THE LIST MUST BE
;DEFINED BY AN ENTRY IN A 'VC' LIST, WHICH ALSO DEFINES THE
;NUMBER OF FIDS TO BE STORED.

;FOR HOMONUC. APPLICATIONS A CONTROL SPECTRUM MUST BE SUBTRACTED
;TO DISTINGUISH SPT EFFECTS (RESULTS ANALOGOUS TO PSEUDO-INDOR).
;FOR HETERONUC. A CONTROL IS USUALLY NOT NEEDED.

;NS DEFINES THE NO. OF TRANSIENTS PER CYCLE FOR EACH O2 VALUE
;AND SHOULD BE A MULTIPLE OF 8.

;NE DEFINES THE NO. OF CYCLES TO BE MADE THROUGH COMPLETE LIST.
;TOTAL TRANSIENTS PER FID = NE*NS.
;USE 2-4 DUMMY SCANS FOR STEADY-STATE!
;RD=0
;D3 = 0.1 SEC TO SET O2
;D1+AQ = 2-4*T1 FOR 1H
;P1 = CA. 180 DEG FLIP FOR 1H DEC. PULSE
;S3 DEFINES DEC. POWER, TYPICALLY 25-55L DEPENDING ON REQUIRED
; IRRAD. BANDWIDTH. (SEE PULSE-PROG. OR SOFTWARE MANUALS
; FOR TECHNIQUES FOR CALIBRATING DEC. FIELD STRENGTH).
```

STACK.AU

; Stacked Plot Of Files From Disk

```
1 RE #1         ;READ IN DATA FILE (ASSUMES FIRST FILE
                ;HAS EXTENSION .001)
2 IF #1         ;INCREMENT EXTENSION
3 PX            ;PLOT SPECTRUM USING CURRENT PLOT LIMITS
                ;AND DPO DEFINITIONS
4 IPO           ;INCREMENT PLOTTER OFFSET BY VALUES DEFINED BY OP
5 IN=1          ;LOOP FOR NEXT PLOT
6 EXIT

;NE DEFINES THE NUMBER OF PLOT TRACES
;OP DEFINES THE PLOTTER OFFSET INCREMENTS
;WHEN AI=1 IS SET TRACES ARE SCALED ABS. RELATIVE TO THE FIRST
;PLOT LIMITS, SCALE, AND DPO MUST BE DEFINED IN CURRENT JOB
; BEFORE STARTING 'STACK'.
```

TOSS.AUR

; Sideband Suppression For CP-MAS With AR Amplifier

```
; DIXON ET AL. J.MAGN.RES. 49, 341 (1982).
; ***************** USE TLO *****************

; 1H:D1-90(Y)-P2(X)-D5-CW- - - - - - - - - - - - - - - - -
;  X:         P2(X)   -P5-180-P6-180-P7-180-P8-180-P9-FID

1 ZE
2 D1 DO         ;RELAXATION
  (P4):C8
3 (P1 PH1):D:E:C8  ;90 DEG 1H PULSE
4 (P2 PH3):D (P2 PH2):T:E:C8   ;CONTACT TIME FOR SPIN-LOCK AND CP
5 (P5 PH3):D:C8           ;P5(USEC)=(122600/SPIN-RATE IN HZ) MINUS P3/2
6 (P3 PH3):D (P3 PH4):T:C8       ;P3=180 DEG X PULSE
7 (P6 PH3):D:C8           ;P6(USEC)=(77300/SPIN-RATE IN HZ) MINUS P3
8 (P3 PH3):D (P3 PH4):T:C8
8 (P7 PH3):D:C8           ;P7(USEC)=(223600/SPIN-RATE IN HZ) MINUS P3
9 (P3 PH3):D (P3 PH4):T:C8
10 (P8 PH3):D:C8          ;P8(USEC)=(1043300/SPIN-RATE IN HZ) MINUS P3
11 (P3 PH3):D (P3 PH4):T:C8
12 (P9 PH3):D             ;P9(USEC)=(774400/SPIN-RATE IN HZ) MINUS P3/2
13 GO=2 PH5 CW
14 D3 DO
   EXIT

PH1=1 3
PH2=0 0 2 2 1 1 3 3
PH3=0
PH4=0 2 0 2 1 3 1 3
PH5=R0 R2 R2 R0 R1 R3 R3 R1

;SEE CPMAS.AUR FOR GENERAL COMMENTS
;USE TOSSCALC TO CALCULATE DURATIONS AND ENTER THE VALUES FOUND
;FOR D25 TO D29 AS P5 TO P9
```

TOSSNQS.AUR

```
; Sideband Suppression For CP-MAS And Non-Quaternary Carbon Suppression With
  AR Amplifier
; DIXON ET AL. J.MAGN.RES. 49, 341 (1982).
; ************** USE TLO ****************

; 1H:D1-90(Y)-P2(X)-D5-CW- - - - - - - - - - - - - - -D5-CW -
;  X:         P2(X)   -P5-180-P6-180-P7-180-P8-180-P9-D5-FID

1 ZE
2 D1 DO         ;RELAXATION
  (P4):C8
3 (P1 PH1):D:E:C8  ;90 DEG 1H PULSE
4 (P2 PH3):D (P2 PH2):T:E:C8   ;CONTACT TIME FOR SQIN-LOCK AND CP
5 (P5 PH3):D:C8           ;P5(USEC)=(122600/SPIN-RATE IN HZ) MINUS P3/2
6 (P3 PH3):D (P3 PH4):T:C8       ;P3=180 DEG X PULSE
7 (P6 PH3):D:C8           ;P6(USEC)=(77300/SPIN-RATE IN HZ) MINUS P3
8 (P3 PH3):D (P3 PH4):T:C8
8 (P7 PH3):D:C8           ;P7(USEC)=(223600/SPIN-RATE IN HZ) MINUS P3
9 (P3 PH3):D (P3 PH4):T:C8
10 (P8 PH3):D:C8          ;P8(USEC)=(1043300/SPIN-RATE IN HZ) MINUS P3
11 (P3 PH3):D (P3 PH4):T:C8
12 (P9 PH3):D             ;P9(USEC)=(774400/SPIN-RATE IN HZ) MINUS P3/2
   D5                     ;DIPOLAR DEPHASING
13 GO=2 PH5 CW
14 D3 DO
   EXIT

PH1=1 3
PH2=0 0 2 2 1 1 3 3
PH3=0
PH4=0 2 0 2 1 3 1 3
PH5=R0 R2 R2 R0 R1 R3 R3 R1

;SEE CPMAS.AUR FOR GENERAL COMMENTS
;USE TOSSCALC TO CALCULATE P5 TO P9
;SET D5 BETWEEN 30 AND 100US
```

TRPHASE.AUR

; Check Or Calibrate Transmitter Phase Shifts Using Automatic Block Address Advance (ASTI=1).

```
1 ST0           ;SET START ADDRESS TO BEGINNING OF MEMORY REGION
2 ZE            ;ZERO BLOCK
3 ST
4 LO TO 2 TIMES C       ;LOOP TO ZERO ALL BLOCKS
5 ST0                   ;RESET TO FIRST BLOCK
6 P1 PH1        ;TRANSMITTER PULSE
7 GO=6          ;ACQUIRE FID (DETECTOR PHASE=0),LOOP TO 3
                ;  AND AUTOMATICALLY INCREMENT BLOCK ST
8 ST0              ;RESET TO FIRST BLOCK
9 WR #1            ;STORE FID
10 IF #1
11 ST
12 LO TO 9 TIMES C  ;WRITE ALL BLOCKS TO DISK
  EXIT

PH1=(8) 0 1 2 3 4 5 6 7        ;DEFINES 45 DEG PHASE SHIFTS

;NB: USE CP MODE AND SET ASTI=1 TO INCREMENT BLOCK ADDRESS AFTER
;  EACH SCAN!!
;RD=PW=0
;NS=NUMBER OF DIFFERENT FIDS, EG. 8 FOR 8 DIFFERENT PHASES
;NBL = NUMBER OF BLOCKS = NS
;VC LIST HAS ONE ITEM = NS


;PHASE CORRECT FIRST SPECTRUM AND USE 'PK' FOR ALL OTHER
;  SPECTRA.
```

WALTZ.AUR

; Sample Program For Data Acquisition Using Waltz Decoupling Using 'ADC' Command And Normal Dwell Clock. This Is Equivalent To A 'GO' With 'CPD'.

```
1 ZE
2 D1 BB S2       ;RELAXATION DELAY WITH DEC. FOR NOE
3 P5:A           ;TRANSM. PULSE (RECEIVER BLANKED)
4 D5             ;DE/2 (RECEIVER STILL OFF)
5 D5 PH9 CW      ;SET REFERENCE PHASE FOR DETECTION AND
                 ; OPEN RECEIVER GATE, SET CW MODE
6 D6 ADC         ;D6=2 USEC, 'ADC' OPENS REC. AND STARTS DIGITIZER
                 ; TAKE TD DATA POINTS USING DWELL TIME DW
7 (P3 PH2 P4 PH0 P2 PH2 P3 PH0 P1 PH2):D  ;ELEMENT 'Q' BEGINS
  (P2 PH0 P4 PH2 P2 PH0 P3 PH2):D
8 (P3 PH0 P4 PH2 P2 PH0 P3 PH2 P1 PH0):D  ;ELEMENT 'Q-BAR'
  (P2 PH2 P4 PH0 P2 PH2 P3 PH0):D
9 L0 TO 8 TIMES 2                         ;REPEAT 'Q-BAR'
  (P3 PH2 P4 PH0 P2 PH2 P3 PH0 P1 PH2):D  ;ELEMENT 'Q'
  (P2 PH0 P4 PH2 P2 PH0 P3 PH2):D
  L1 TO 7 TIMES UPR             ;REPEAT WALTZ-16 SEQUENCE
10 RCYC=2 PH8                   ;LOOP FOR NS SCANS
  EXIT

PH0=0   ;DECOUPLER PHASES
PH1=1
PH2=2
PH3=3

PH8=R0 R0 R2 R2 R1 R1 R3 R3
PH9=0 0 0 0 3 3 3 3 ;REFERENCE PHASE FOR DETECTION

;PERFORMS DATA ACQUISITION IN A MANNER IDENTICAL TO 'GO'
; D1 IS EQUIVALENT TO 'RD'
; S2 DEFINES DECOUPLER POWER
; P5 IS EQUIVALENT TO PW
; 2*D5 IS EQUIVALENT TO DE
; D6=2 USEC FOR ADC COMMAND
; P1=90 DEG 1H DEC. PULSE AT POWER SETTING S2
; P2,P3,P4=180,270,360 DEG DEC. PULSE

;L1 = LOOP COUNTER, SET SO THAT L1*96*P1 => AQ.
```

XHCORR.AUR

```
; Heteronuc. Shift-Correlated 2-D NMR (CPD Decoupling) Using Polarization
  Transfer From 1H To X Via J(XH).
; A.BAX & G.MORRIS, J.MAGN.RES. 42, 501 (81)

;  1H: DO - 90 - D0 -   - D0 - D3 - 90        BB
;   X: D1           -180-           90 - D4 - FID

; F2 DOMAIN: BB DEC. X-NUCLEUS SPECTRUM
; F1 DOMAIN: X-NUCLEUS DECOUPLED 1H SPECTRUM WITH J(HH)
; J(XH) MUST BE > 1/T2

1 ZE
2 D1 DO S1      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS
5 P4 PH4        ;180 DEG X PULSE TO DECOUPLE X FROM 1H
6 D0            ;FURTHER EVOLUTION
7 D3            ;WAIT FOR OPTIMUM POLARIZATION OF X-H
                ;  1H DOUBLET
8 P1:D PH2  P3 PH3      ;90 DEG 1H PULSE COMPLETES POLAR.
                        ; TRANSFER, 90 DEG X PULSE CREATES
                ;DETECTABLE X,Y-MAGN.
9 D4 S2         ;WAIT FOR ANTI-PHASE X-NUCLEUS MULTIPLETS
                ;  TO REPHASE
10 GO=2 PH5 CPD   ;ACQUIRE BB DEC. X-NUCLEUS FID, MODULATED BY
                ;1H SHIFTS AND J(HH).
11 D4 DO        ;GATE DEC. OFF
12 WR #1        ;STORE FID
13 IF #1        ;INCREMENT FILE NUMBER
14 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
15 EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2

PH5=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

;NS=4*N

;PROGRAM REQUESTS FID FILENAME WITH .SER EXTENSION
;NE DEFINES THE NUMBER OF EXPERIMENTS =TD1 FOR 1H
;D1 = 1-5*T1 FOR 1H
;S1 = 0H, MAX. POWER FOR PULSING
;S2 = NORMAL POWER FOR CPD DECOUPLING
;D0 = 3E-6 INITIAL DELAY
;P1 = 90 DEG 1H PULSE
;P3,P4 = 90,180 X PULSE
;D3 = 0.5/J(XH) FOR MAX. POLARIZATION TRANSFER
;D4 = 0.25/J(XH) TO OBSERVE ALL MULTIPLICITIES
;   = 0.5/J(XH) TO OBSERVE XH DOUBLET MULTIP. ONLY
;RD=PW=0
;ND0 = 2
;IN = 0.25/SW1,   SW1=0.5*(1H SHIFT RANGE)
```

XHCORRC.AUR

```
; XHCORR With Composite Inversion Pulse.  Heteronuc. Shift-Correlated 2-D NMR
  (CPD Decoupling) Using Polarization Transfer From 1H To X Via J(XH).
; A.BAX & G.MORRIS, J.MAGN.RES. 42, 501 (81)

;  1H: DO - 90 - D0 -    - D0 - D3 - 90        BB
;   X: D1           -180-           90 - D4 - FID

; F2 DOMAIN: BB DEC. X-NUCLEUS SPECTRUM
; F1 DOMAIN: X-NUCLEUS DECOUPLED 1H SPECTRUM WITH J(HH)
; J(XH) MUST BE > 1/T2

1 ZE
2 D1 DO S1      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS
5 P3 PH4        ;COMP. 180 DEG X PULSE TO DECOUPLE X FROM 1H
  P5 PH5
  P3 PH4
6 D0            ;FURTHER EVOLUTION
7 D3            ;WAIT FOR OPTIMUM POLARIZATION OF X-H
                ;  1H DOUBLET
8 P1:D PH2  P3 PH3      ;90 DEG 1H PULSE COMPLETES POLAR.
                        ; TRANSFER, 90 DEG X PULSE CREATES
                ;DETECTABLE X,Y-MAGN.
9 D4 S2         ;WAIT FOR ANTI-PHASE X-NUCLEUS MULTIPLETS
                ;  TO REPHASE
10 GO=2 PH6 CPD   ;ACQUIRE BB DEC. X-NUCLEUS FID, MODULATED BY
                ;1H SHIFTS AND J(HH).
11 D4 DO        ;GATE DEC. OFF
12 WR #1        ;STORE FID
13 IF #1        ;INCREMENT FILE NUMBER
14 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
15 EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2
PH5=A1 A1 A1 A1 A3 A3 A3 A3

PH6=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

;NS=4*N

;PROGRAM REQUESTS FID FILENAME WITH .SER EXTENSION
;NE DEFINES THE NUMBER OF EXPERIMENTS =TD1 FOR 1H
;D1 = 1-5*T1 FOR 1H
;S1 = 0H, MAX. POWER FOR PULSING
;S2 = NORMAL POWER FOR CPD DECOUPLING
;D0 = 3E-6 INITIAL DELAY
;P1 = 90 DEG 1H PULSE
;P3,P4 = 90,180 X PULSE
;P5 = 240 DEG X PULSE FOR COMPOSITE INVERSION PULSE
;D3 = 0.5/J(XH) FOR MAX. POLARIZATION TRANSFER
;D4 = 0.25/J(XH) TO OBSERVE ALL MULTIPLICITIES
;   = 0.5/J(XH) TO OBSERVE XH DOUBLET MULTIP. ONLY
;RD=PW=0
```

```
;ND0 = 2
;IN = 0.25/SW1,   SW1=0.5*(1H SHIFT RANGE)
```

XHCORRD.AUR

```
; X-H Shift Correlation With 1H Dec In F1 Domain.  This Is An Extension Of The
  Standard X-H Shift Correlation And Removes J(HH) Coupling From The F1 Domain
  Between Spins Not Attached To The Same X-Nucleus (Assumes J(XH) >> J(HH) ).
  Only Efficient When J(XH) > 1/T2
; BAX, J.MAGN.RES. 53, 517 (1983) AND
; V.RUTAR J.MAGN.RES. 58, 306 (1984)
; J.A.WILDE & P.H.BOLTON, J.MAGN.RES. 59, 343 (1984)

;  1H: DO-90-D0-90-D3-180-D3-90-D0-D3-90    BB
;   X: D1            180            90-D4-FID

1 ZE
2 D1 DO S1      ;RELAX, PREPARE FOR DEC PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;ONE-HALF EVOLUTION PERIOD
5 P1:D PH7      ;90 DEG 1H
6 D3            ;=1/(2J(XH))
7 (P2 PH8):D  (P4 PH4)
                ;180 DEG TO 1H AND X FOR DECOUPLING
8 D3
9 P1:D PH7      ;90 DEG 1H
10 D0           ;COMPLETE EVOLUTION
11 D3           ;WAIT FOR OPTIMUM POLARIZATION
12 P1:D PH2  P3 PH3   ;90 DEG FOR 1H AND X, POLARIZATION TRANSFER
13 D4 S2        ;WAIT FOR REFOCUSSING
14 GO=2 PH6 CPD ;ACQUIRE WITH CPD DEC, FID MODULATED ONLY BY 1H
                ; CHEMICAL SHIFTS
15 D4 DO        ;GATE DEC OFF
16 WR #1        ;STORE FID
17 IF #1        ;INCREMENT FILE NUMBER
18 IN=1         ;LOOP FOR NEXT EXPERIMENT
19 EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2

PH6=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH7=B1 B1 B1 B1 B2 B2 B2 B2     ;PHASE INCREMENTS SUGGESTED BY
    B3 B3 B3 B3 B0 B0 B0 B0     ;BOLTON
PH8=B0 B0 B0 B0 B1 B1 B1 B1
    B2 B2 B2 B2 B3 B3 B3 B3

;USE NS=4*N
;D1=1-5*T1 FOR 1H
;D3=1/(2J(XH))
;D4=1/(4J(XH)) FOR ALL MULTIPLICITIES
;  =1/(2J(XH)) FOR DOUBLET MULTIP. ONLY
;D0=3E-6
;S1=0H,  S2=DESIRED DEC POWER
;P1,P2 = 90,180 FOR 1H
;P3,P4 = 90,180 FOR X
;RD=PW=0
;ND0=2
;IN=0.25/SW1,  SW1=0.5*(1H SHIFT RANGE)
```

XHCORRDC.AUR

```
; XHCORRD Using Composite 180 Deg X Pulse (90-240-90).  X-H Shift Correlation
  With 1H Dec In F1 Domain
; BAX, J.MAGN.RES. 53, 517 (1983) AND
; V.RUTAR J.MAGN.RES. 58, 306 (1984)
; J.A.WILDE & P.H.BOLTON, J.MAGN.RES. 59, 343 (1984)
; THIS IS AN EXTENSION OF THE STANDARD X-H SHIFT CORRELATION AND
; REMOVES J(HH) COUPLING FROM THE F1 DOMAIN
; BETWEEN SPINS NOT ATTACHED TO THE SAME X-NUCLEUS
; (ASSUMES J(XH) >> J(HH) ).
; ONLY EFFICIENT WHEN J(XH) > 1/T2

;  1H: DO-90-D0-90-D3- 180 -D3-90-D0-D3-90    BB
;   X: D1             (180)             90-D4-FID

1 ZE
2 D1 DO S1      ;RELAX, PREPARE FOR DEC PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;ONE-HALF EVOLUTION PERIOD
5 P1:D PH7      ;90 DEG 1H
6 D3            ;=1/(2J(XH))
7 (P2 PH8):D  (P3 PH4 P5 PH5 P3 PH4)
                ;180 DEG TO 1H AND COMPOSITE X FOR DECOUPLING
8 D3
9 P1:D PH7      ;90 DEG 1H
10 D0           ;COMPLETE EVOLUTION
11 D3           ;WAIT FOR OPTIMUM POLARIZATION
12 P1:D PH2  P3 PH3   ;90 DEG FOR 1H AND X, POLARIZATION TRANSFER
13 D4 S2        ;WAIT FOR REFOCUSSING
14 GO=2 PH6 CPD ;ACQUIRE WITH CPD DEC, FID MODULATED ONLY BY 1H
                ; CHEMICAL SHIFTS
15 D4 DO        ;GATE DEC OFF
16 WR #1        ;STORE FID
17 IF #1        ;INCREMENT FILE NUMBER
18 IN=1         ;LOOP FOR NEXT EXPERIMENT
19 EXIT

PH1=B0
PH2=B0 B2 B1 B3
PH3=A0 A0 A0 A0 A0 A0 A0 A0
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2
PH5=A1 A1 A1 A1 A3 A3 A3 A3

PH6=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH7=B1 B1 B1 B1 B2 B2 B2 B2     ;PHASE INCREMENTS SUGGESTED BY
    B3 B3 B3 B3 B0 B0 B0 B0     ;BOLTON
PH8=B0 B0 B0 B0 B1 B1 B1 B1
    B2 B2 B2 B2 B3 B3 B3 B3

;USE NS=4*N
;D1=1-5*T1 FOR 1H
;D3=1/(2J(XH))
;D4=1/(4J(XH)) FOR ALL MULTIPLICITIES
;  =1/(2J(XH)) FOR DOUBLET MULTIP. ONLY
;D0=3E-6
;S1=0H,  S2=DESIRED DEC POWER
;P1,P2 = 90,180 FOR 1H
```

```
;P3,P4 = 90,180 FOR X
;P5 = 240 DEG X PULSE FOR COMPOSITE INVERSION PULSE
;RD=PW=0
;ND0=2
;IN=0.25/SW1,  SW1=0.5*(1H SHIFT RANGE)
```

XHDEDW.AUR

; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From 1H To X-Nuclei And H-H Decoupling.

```
; PHASE-SENSITIVE (TPPI)
; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 144 (83)
; T.T NAKASHIMA ET AL. J.MAGN.RES. 59, 124 (84)

;  1H: D1-90-D0-90-D2-180-D2-90-D0-D2-180-  - P0-  -BB
;   X:                180              90-D2-180-D2-FID

; F2: X NUCLEUS SHIFTS, F1: H SHIFTS WITH J(HH) DECOUPLED
;                              (EXCEPT GEMINAL J(HH) ).

1 ZE
2 D1 S1 DO      ;1H RELAXATION, SET DEC. POWER FOR PULSE
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;SHIFTS AND COUPLINGS EVOLVE
5 P1:D PH7      ;BILINEAR ROTATION SEQUENCE TO REMOVE J(XH)
6 D2                      ;AND J(HH) FROM EVOLUTION.
7 (P2 PH8):D   (P3 PH0 P5 PH9 P3 PH0)   ;COMPOSITE 180-X
8 D2
9 P1:D PH7       ;COMPLETE BILINEAR ROTATION
10 D0            ;CONTINUE EVOLUTION OF H SHIFTS
11 D2            ;CREATE ANTIPARALLEL XH VECTORS
13 (P2 PH2):D  (P3 PH4 D2)  ;180 DEG 1H PULSE TO REFOCUS SHIFTS
                            ;90 DEG X PULSE FOR MQ COHERENCE

14 (P0 PH3):D  (P4 PH5 D2 S2)   ;VARIABLE PULSE FOR 1H TO COMPLETE
                          ;POLARIZATION TRANSFER, 180 X PULSE
                          ;REFOCUSSES X SHIFTS FOR ACQ.
                          ;SET DEC. POWER DURING REFOCUSSING TIME
15 GO=2 PH6 CPD           ;ACQUIRE FID WITH DEC.
16 D2 DO
   IP3           ;INCREMENT PH3 FOR TPPI
17 WR #1
18 IF #1
19 IN=1
   EXIT              ;EXIT WITH DEC. OFF

PH1=B0                         ;DECOUPLER PHASES
PH2=B0 B0 B2 B2
PH3=B1 B3

PH4=A0 A0 A0 A0 A1 A1 A1 A1    ;TRANSMITTER PHASES
    A2 A2 A2 A2 A3 A3 A3 A3
PH5=A0 A0 A2 A2 A1 A1 A3 A3

PH6=R0 R2 R0 R2 R1 R3 R1 R3        ;RECEIVER PHASE
    R2 R0 R2 R0 R3 R1 R3 R1

PH7=B0 B0 B1 B1 B2 B2 B3 B3     ;PHASE INCREMENTS SUGGESTED BY BOLTON
PH8=B1 B1 B2 B2 B3 B3 B0 B0
    B3 B3 B0 B0 B1 B1 B2 B2

PH0=A0 A0 A2 A2
PH9=A1 A1 A3 A3

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;  NB: D2 OCCURS 5 TIMES IN SEQUENCE, IE. NOT SUITABLE FOR
;  SYSTEMS WITH SHORT RELAXATION TIMES (USE XHCORRD.AU).
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
```

```
;P5 = 240 DEG X PULSE FOR COMPOSITE INVERSION PULSE
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.
;P9 = 90 DEG DEC PULSE FOR CPD

;NS=2*N
;RD=PW=0

;NB: ND0=4 FOR TPPI
;MC2=W,  REV=Y
```

XHDEPT.AUR

```
; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From 1H To X-
  Nuclei (see XHDEPTW.AU)
; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 144 (83)

;  1H: D1-90- D2 -180   -         -P0-   - BB
;   X:             90-D2-D0-180-D0-   D2 - FID

; F2: X NUCLEUS SHIFTS, F1: H SHIFTS AND J(HH)

1 ZE
2 D1 S1  DO                     ;1H RELAXATION, SET DEC. POWER
                                ;FOR PULSING
3 (P1 PH1 D2):D                 ;90 DEG 1H PULSE, SHIFTS AND
                                ;J(XH) EVOLVE
4 (P2 PH2):D  (P3 PH4 D2)       ;180 DEG 1H PULSE TO REFOCUS
                                ;SHIFTS, 90 DEG X PULSE FOR MQ
                                ;COHERENCE
  D0            ;EVOLUTION
  P4 PH5        ;180 X PULSE TO REFOCUS X SHIFTS AND X-H COUPLINGS
  D0                      ;EVOLUTION OF H SHIFTS AND H-H COUPLINGS
5 (P0 PH3):D  (D2 S2)     ;VARIABLE PULSE FOR 1H TO COMPLETE
                          ;POLARIZATION TRANSFER
                          ;SET DEC. POWER DURING REFOCUSSING TIME
6 GO=2 PH6 CPD            ;ACQUIRE FID WITH DEC.
7 D2 DO
  WR #1
  IF #1
  IN=1
8 EXIT              ;EXIT WITH DEC. OFF

PH1=B0                          ;DECOUPLER PHASES
PH2=B0 B0 B0 B0 B2 B2 B2 B2
PH3=B1 B3 B2 B0

PH4=A0 A0 A0 A0 A0 A0 A0 A0     ;TRANSMITTER PHASES
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH5=A0 A0 A0 A0 A2 A2 A2 A2
    A1 A1 A1 A1 A3 A3 A3 A3

PH6=R0 R2 R1 R3 R0 R2 R1 R3         ;RECEIVER PHASE
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.
;P9 = DEC. 90 DEG PULSE FOR CPD

;NS=4*N   ND0=2
;RD=PW=0
;MC2=N CAN BE USED FOR PHASE-SENSITIVE SPECTRA WITH PHASE-TWIST
```

XHDEPTD.AUR

```
; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From 1H To X-
  Nuclei And H-H Decoupling. (see XHDEDW.AU)
; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 144 (83)
; T.T NAKASHIMA ET AL. J.MAGN.RES. 59, 124 (84)

;  1H: D1-90-D0-90-D2-180-D2-90-D0-D2-180-  - P0-  -BB
;   X:                 180              90-D2-180-D2-FID

; F2: X NUCLEUS SHIFTS, F1: H SHIFTS WITH J(HH) DECOUPLED
;                              (EXCEPT GEMINAL J(HH) ).

1 ZE
2 D1 S1 DO      ;1H RELAXATION, SET DEC. POWER FOR PULSE
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;SHIFTS AND COUPLINGS EVOLVE
5 P1:D PH7      ;BILINEAR ROTATION SEQUENCE TO REMOVE J(XH)
6 D2                    ;AND J(HH) FROM EVOLUTION.
7 (P2 PH8):D   (P3 PH0 P5 PH9 P3 PH0)   ;COMPOSITE 180-X
8 D2
9 P1:D PH7       ;COMPLETE BILINEAR ROTATION
10 D0            ;CONTINUE EVOLUTION OF H SHIFTS
11 D2            ;CREATE ANTIPARALLEL XH VECTORS
13 (P2 PH2):D  (P3 PH4 D2)  ;180 DEG 1H PULSE TO REFOCUS SHIFTS
                            ;90 DEG X PULSE FOR MQ COHERENCE

14 (P0 PH3):D  (P4 PH5 D2 S2)   ;VARIABLE PULSE FOR 1H TO COMPLETE
                          ;POLARIZATION TRANSFER, 180 X PULSE
                          ;REFOCUSSES X SHIFTS FOR ACQ.
                          ;SET DEC. POWER DURING REFOCUSSING TIME
15 GO=2 PH6 CPD           ;ACQUIRE FID WITH DEC.
16 D2 DO
17 WR #1
18 IF #1
19 IN=1
   EXIT               ;EXIT WITH DEC. OFF

PH1=B0                          ;DECOUPLER PHASES
PH2=B0 B0 B0 B0 B2 B2 B2 B2
PH3=B1 B3 B0 B2

PH4=A0 A0 A0 A0 A0 A0 A0 A0     ;TRANSMITTER PHASES
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3
PH5=A0 A0 A0 A0 A2 A2 A2 A2
    A1 A1 A1 A1 A3 A3 A3 A3

PH6=R0 R2 R1 R3 R0 R2 R1 R3         ;RECEIVER PHASE
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH7=B0 B0 B0 B0 B1 B1 B1 B1      ;PHASE INCREMENTS SUGGESTED BY BOLTON
    B2 B2 B2 B2 B3 B3 B3 B3
PH8=B1 B1 B1 B1 B2 B2 B2 B2
    B3 B3 B3 B3 B0 B0 B0 B0
    B3 B3 B3 B3 B0 B0 B0 B0
    B1 B1 B1 B1 B2 B2 B2 B2

PH0=A0 A0 A0 A0 A2 A2 A2 A2
PH9=A1 A1 A1 A1 A3 A3 A3 A3

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
```

```
;  NB: D2 OCCURS 5 TIMES IN SEQUENCE, IE. NOT SUITABLE FOR
;  SYSTEMS WITH SHORT RELAXATION TIMES (USE XHCORRD.AU).
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P5 = 240 DEG X PULSE FOR COMPOSITE INVERSION PULSE
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;           = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;           = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.
;P9 = 90 DEG DEC PULSE FOR CPD

;NS=4*N   ND0=2
;RD=PW=0
;MC2=N CAN BE USED FOR PHASE-SENSITIVE SPECTRA WITH PHASE-TWIST
```

XHDEPTW.AUR

```
; X-H Shift Correlation 2-D Nmr Using DEPT Polarization Transfer From 1H To X-
  Nuclei
; M.R.BENDALL & D.T.PEGG, J.MAGN.RES. 53, 144 (83)
; PHASE-SENSITIVE WITH TPPI (MC2=W).

;  1H: D1-90- D2 -180   -         -P0-   - BB
;   X:             90-D2-D0-180-D0-   D2 - FID

; F2: X NUCLEUS SHIFTS, F1: H SHIFTS AND J(HH)

1 ZE
2 D1 S1  DO                     ;1H RELAXATION, SET DEC. POWER
                                ;FOR PULSING
3 (P1 PH1 D2):D                 ;90 DEG 1H PULSE, SHIFTS AND
                                ;J(XH) EVOLVE
4 (P2 PH2):D  (P3 PH4 D2)       ;180 DEG 1H PULSE TO REFOCUS
                                ;SHIFTS, 90 DEG X PULSE FOR MQ
                                ;COHERENCE
  D0            ;EVOLUTION
  P4 PH5        ;180 X PULSE TO REFOCUS X SHIFTS AND X-H COUPLINGS
  D0                      ;EVOLUTION OF H SHIFTS AND H-H COUPLINGS
5 (P0 PH3):D  (D2 S2)     ;VARIABLE PULSE FOR 1H TO COMPLETE
                          ;POLARIZATION TRANSFER
                          ;SET DEC. POWER DURING REFOCUSSING TIME
6 GO=2 PH6 CPD            ;ACQUIRE FID WITH DEC.
7 D2 DO
  IP3           ;INCREMENT PH3 BY 90 DEG FOR TPPI
  WR #1
  IF #1
  IN=1
8 EXIT              ;EXIT WITH DEC. OFF

PH1=B0                          ;DECOUPLER PHASES
PH2=B0 B0 B2 B2
PH3=B1 B3

PH4=A0 A0 A0 A0 A1 A1 A1 A1     ;TRANSMITTER PHASES
    A2 A2 A2 A2 A3 A3 A3 A3

PH5=A0 A0 A2 A2 A1 A1 A3 A3


PH6=R0 R2 R0 R2 R1 R3 R1 R3        ;RECEIVER PHASE
    R2 R0 R2 R0 R3 R1 R3 R1

;D1 = 1-5*T1 FOR 1H
;D2 = 0.5/J(XH) FOR OPTIMUM POLARIZATION
;S1 = 0H FOR MAX. POWER PULSES
;S2 = NORMAL POWER FOR CPD DEC.
;P1,P2 = 90,180 PULSES FOR 1H DEC.
;P3,P4 = 90,180 PULSES FOR X
;P0 IS VARIABLE DEPENDING ON DESIRED MULTIPLICITY SELECTION:
;  E.G. P0 = 45 DEG GIVES XH, XH2, XH3 ALL POSITIVE
;          = 90 DEG GIVES XH ONLY (USE TO CALIBRATE DEC. PULSE)
;          = 135 DEG GIVES XH, XH3 POS. AND XH2 NEG.
;P9 = DEC. 90 DEG PULSE FOR CPD

;NS=2*N
;RD=PW=0

;NB: ND0=4 IN ORDER TO GIVE HALF THE NORMAL 'IN' VALUE
;MC2=W, REV=N
```

XHINEPT.AUR

```
; Heteronuc. Shift-Correlated 2-D Nmr Using Polarization Transfer From 1H To X
  Via J(XH) With Refocussing Of Chem. Shifts
; A.BAX & G.MORRIS, J.MAGN.RES. 42, 501 (81)

;  1H: DO-90-D0-   -D0-D5-180-D5-90-  -180-   BB
;   X: D1      -180-     -180-   90-D6-180-D6-FID

; F2 DOMAIN: BB DEC. X-NUCLEUS SPECTRUM
; F1 DOMAIN: X-NUCLEUS DECOUPLED 1H SPECTRUM WITH J(HH)
; J(XH) MUST BE > 1/T2

1 ZE
2 D1 DO S1      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS
5 P4 PH4        ;180 DEG X PULSE TO DECOUPLE X FROM 1H
6 D0            ;FURTHER EVOLUTION
7 D5            ;WAIT FOR OPTIMUM POLARIZATION OF X-H
                ;  1H DOUBLET
  (P2 PH7 D5):D  P4 PH4
8 P1:D PH2  (P3 PH3 D6)  ;90 DEG 1H PULSE COMPLETES POLAR.
                        ; TRANSFER, 90 DEG X PULSE CREATES
                ;DETECTABLE X,Y-MAGN.
9 P2:D PH7 (P4 PH8 D6 S2) ;WAIT FOR ANTI-PHASE X-NUCLEUS MULTIPLETS
                ;  TO REPHASE
10 GO=2 PH6 CPD ;ACQUIRE BB DEC. X-NUCLEUS FID, MODULATED BY
                ;1H SHIFTS AND J(HH).
11 D4 DO        ;GATE DEC. OFF
12 WR #1        ;STORE FID
13 IF #1        ;INCREMENT FILE NUMBER
14 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
15 EXIT

PH1=B0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=B1 B3 B0 B2                 ;SCANS 3-4 FOR F1 QUAD
PH3=A0 A0 A0 A0 A0 A0 A0 A0     ;SCANS 5-8 REVERSE PHASE OF 180-X
    A1 A1 A1 A1 A1 A1 A1 A1
    A2 A2 A2 A2 A2 A2 A2 A2
    A3 A3 A3 A3 A3 A3 A3 A3

PH4=A0 A0 A0 A0 A2 A2 A2 A2

PH6=R0 R2 R1 R3 R0 R2 R1 R3
    R1 R3 R2 R0 R1 R3 R2 R0
    R2 R0 R3 R1 R2 R0 R3 R1
    R3 R1 R0 R2 R3 R1 R0 R2

PH7=B0
PH8=A0

;NS=4*N

;PROGRAM REQUESTS FID FILENAME WITH .SER EXTENSION
;NE DEFINES THE NUMBER OF EXPERIMENTS =TD1 FOR 1H
;D1 = 1-5*T1 FOR 1H
;S1 = 0H, MAX. POWER FOR PULSING
;S2 = NORMAL POWER FOR CPD DECOUPLING
;D0 = 3E-6 INITIAL DELAY
;P1 = 90 DEG 1H PULSE
;P3,P4 = 90,180 X PULSE
;D5 = 0.5/J(XH) FOR MAX. POLARIZATION TRANSFER
;D6 = 0.25/J(XH) TO OBSERVE ALL MULTIPLICITIES
;   = 0.5/J(XH) TO OBSERVE XH DOUBLET MULTIP. ONLY
;RD=PW=0
```

```
;ND0 = 2
;IN = 0.25/SW1,   SW1=0.5*(1H SHIFT RANGE)
```

XHINW.AU

```
; Heteronuc. Shift-Correlated 2-D Nmr Using Polarization Transfer From 1H To X
  Via J(XH) With Refocussing Of Chem. Shifts
; PHASE-SENSITIVE  TPPI
; A.BAX & G.MORRIS, J.MAGN.RES. 42, 501 (81)

;  1H: DO-90-D0-   -D0-D5-180-D5-90-  -180-   BB
;   X: D1      -180-     -180-   90-D6-180-D6-FID

; F2 DOMAIN: BB DEC. X-NUCLEUS SPECTRUM
; F1 DOMAIN: X-NUCLEUS DECOUPLED 1H SPECTRUM WITH J(HH)
; J(XH) MUST BE > 1/T2

1 ZE
2 D1 DO S1      ;1H RELAXATION, SET DEC. FOR PULSING
3 P1:D PH1      ;90 DEG 1H PULSE
4 D0            ;EVOLUTION OF 1H SHIFTS AND COUPLINGS
5 P4 PH4        ;180 DEG X PULSE TO DECOUPLE X FROM 1H
6 D0            ;FURTHER EVOLUTION
7 D5            ;WAIT FOR OPTIMUM POLARIZATION OF X-H
                ;  1H DOUBLET
  (P2 PH7 D5):D  P4 PH4
8 P1:D PH2  (P3 PH3 D6)  ;90 DEG 1H PULSE COMPLETES POLAR.
                        ; TRANSFER, 90 DEG X PULSE CREATES
                ;DETECTABLE X,Y-MAGN.
9 P2:D PH7 (P4 PH8 D6 S2) ;WAIT FOR ANTI-PHASE X-NUCLEUS MULTIPLETS
                ;  TO REPHASE
10 GO=2 PH6 CPD ;ACQUIRE BB DEC. X-NUCLEUS FID, MODULATED BY
                ;1H SHIFTS AND J(HH).
11 D4 DO        ;GATE DEC. OFF
   IP2         ;INCREMENT PH2 FOR TPPI
12 WR #1        ;STORE FID
13 IF #1        ;INCREMENT FILE NUMBER
14 IN=1         ;INCREMENT D0, LOOP FOR NEXT EXPER.
15 EXIT

PH1=B0                          ;SCANS 1-2 SUPPRESS AXIAL PEAKS
PH2=B1 B3                   ;SCANS 3-4 FOR F1 QUAD
PH3=A0 A0 A0 A0 A1 A1 A1 A1    ;SCANS 5-8 REVERSE PHASE OF 180-X
    A2 A2 A2 A2 A3 A3 A3 A3

PH4=A0 A0 A2 A2

PH6=R0 R2 R0 R2 R1 R3 R1 R3
    R2 R0 R2 R0 R3 R1 R3 R1

PH7=B0 B0 B2 B2
PH8=A0 A0 A2 A2 A1 A1 A3 A3

;NS=2*N

;PROGRAM REQUESTS FID FILENAME WITH .SER EXTENSION
;NE DEFINES THE NUMBER OF EXPERIMENTS =TD1 FOR 1H
;D1 = 1-5*T1 FOR 1H
;S1 = 0H, MAX. POWER FOR PULSING
;S2 = NORMAL POWER FOR CPD DECOUPLING
;D0 = 3E-6 INITIAL DELAY
;P1 = 90 DEG 1H PULSE
;P3,P4 = 90,180 X PULSE
;D5 = 0.5/J(XH) FOR MAX. POLARIZATION TRANSFER
;D6 = 0.25/J(XH) TO OBSERVE ALL MULTIPLICITIES
;   = 0.5/J(XH) TO OBSERVE XH DOUBLET MULTIP. ONLY
;RD=PW=0

;NB:  ND0=4 FOR TPPI
```

```
;MC2=W,  REV=Y
```

XHSEL.AUR

; Observe 1H And Select Molecules Containing A Label Spin (Eg. 13C Or 15N) Using BSV-3 BX Heteronuclear Decoupler And Synthesizer To Pulse X-Nucleus, Gives 1H Spectrum With J(XH)

```
;  1H: D1 - 90 - D2     -180-  - D2 - FID
;   X: DO           - 90-   -90-

1 ZE
2 D1 CW DO      ;RELAX, SET DEC. TO CW STATUS FOR PULSING
3 P1 PH1        ;90 DEG 1H PULSE
4 D2            ;=1(2J(XH))
5 P3:D PH3      ;90 DEG X PULSE WITH BSV-3
6 P2 PH2        ;REFOCUSSING PULSE FOR 1H SHIFTS AND J(XH)
7 P3:D PH4      ;90 DEG X, PHASE ALTERN. TO CANCEL NON-LABEL MOL.
8 D2            ;REFOCUS (ALLOWS FOR DECOUPLING)
9 GO=2 PH5      ;DETECT 1H   WITHOUT  X DEC.
  EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1
PH2=A0 A2 A0 A2 A1 A3 A1 A3
    A1 A3 A1 A3 A2 A0 A2 A0
PH3=B0
PH4=B0 B0 B2 B2
PH5=R0 R0 R2 R2 R1 R1 R3 R3
    R2 R2 R0 R0 R3 R3 R1 R1

;  NS=4*N,  DS=2 OR 4
;RD=PW=0
;PHASE CYCLE REJECTS 1H NOT COUPLED TO X
;D1 > T1(1H) , D2=1/(2J(XH))
;P1,P2 = 90, 180 FOR 1H
;P3=90 FOR X

;TO CALIBRATE X-NUCLEUS 90 DEG PULSE, ADJUST P3 TO NULL
;  XH DOUBLETS (THIS OCCURS WHEN P3 = 45 DEG).
;BSV-3 SHOULD BE SET SO THAT 1H-DECOUPLER IS INACTIVE.
```

XHSELD.AUR

; Observe 1H And Select Molecules Containing A Label Spin (Eg. 13C or 15N), Decouple X During Acq. Using BSV-3 BX Heteronuclear Decoupler And Synthesizer To Pulse X-Nucleus, With Modification For Computer Control Of CW/BB Or Using CPD.

```
;  1H: D1 - 90 - D2     -180-  - D2 - FID
;   X: DO           - 90-   -90-    - CPD

1 ZE
2 D1 CW DO      ;RELAX, SET DEC. TO CW STATUS FOR PULSING
3 P1 PH1        ;90 DEG 1H PULSE
4 D2            ;=1(2J(XH))
5 P3:D PH3      ;90 DEG X PULSE WITH BSV-3
6 P2 PH2        ;REFOCUSSING PULSE FOR 1H SHIFTS AND J(XH)
7 P3:D PH4      ;90 DEG X, PHASE ALTERN. TO CANCEL NON-LABEL MOL.
8 D2            ;REFOCUS (ALLOWS FOR DECOUPLING)
9 GO=2 PH5 CPD  ;DETECT 1H   WITH  X DEC.
  EXIT

PH1=A0 A0 A0 A0 A1 A1 A1 A1
PH2=A0 A2 A0 A2 A1 A3 A1 A3
    A1 A3 A1 A3 A2 A0 A2 A0
PH3=B0
PH4=B0 B0 B2 B2
PH5=R0 R0 R2 R2 R1 R1 R3 R3
    R2 R2 R0 R0 R3 R3 R1 R1

;  NS=4*N,  DS=2 OR 4
;RD=PW=0
;PHASE CYCLE REJECTS 1H NOT COUPLED TO X
;D1 > T1(1H) , D2=1/(2J(XH))
;P1,P2 = 90, 180 FOR 1H
;P3=90 FOR X

;TO CALIBRATE X-NUCLEUS 90 DEG PULSE (AT POWER USED FOR BB)
;  ADJUST P3 TO NULL
;  XH DOUBLETS (THIS OCCURS WHEN P3 = 45 DEG).
;  WHEN USING CPD1, SET P9=P3=90 DEG PULSE.
;BSV-3 SHOULD BE SET SO THAT 1H-DECOUPLER IS INACTIVE.
```